

[BasesFuncionales.mw](#)

>

Bases para un espacio de funciones

Luis A. Núñez

Escuela de Física

Universidad Industrial de Santander

Bucaramanga Colombia

Actualizado Noviembre 2020

Expansión por series de Taylor. Una base de monomios en un espacio de polinomios

```
>restart: with(plots):
```


Consideremos una función sencilla

```
>f := x -> sqrt(1-x^2)*sin(3*x);
```

$$f := x \mapsto \sqrt{1-x^2} \sin(3x)$$

(1)

```
>plot(f(x),x=-1..1, legend=["f"]);
```



y recordemos la aproximación por serie de Taylor. Fijemos el mayor orden del monomio en la aproximación

```
> n:=7;
```

```
n := 7
```

(2)

```
> Taylor[n] := taylor(f(x), x=0, n);
```

$$Taylor_7 := 3x - 6x^3 + \frac{39}{10}x^5 + O(x^7)$$


(3)

Nótese que fijamos el orden de la aproximación es $n-1$

```
> fAproxTaylor[n] := convert(Taylor[n], polynom);
```

La serie de Taylor no es más que una expansión en una base de monomios en la cual conocemos los coeficientes y sabemos como generarlos. Si graficamos la función y su aproximación vemos, que mejora cuando aumentamos el orden del polinomio

```
> plot([f(x), fAproxTaylor[n]], x=-1..1, legend=["f", "fAproxTaylor"]);
```




```
> nn:=9; Taylor2[nn] := taylor(f(x), x=0, nn); fAproxTaylor2[nn] := convert(Taylor2[nn], polynom);
```

$$m := 9$$

$$Taylor2_9 := 3x - 6x^3 + \frac{39}{10}x^5 - \frac{15}{14}x^7 + O(x^9)$$

(4)

```
> plot([f(x), fAproxTaylor[n], fAproxTaylor2[nn]], x=-1..1, legend=["f", "fAproxTaylor", "fAproxTaylor2"]);
```



Como siempre el error de la aproximación será la norma de la resta de los dos vectores: la función original y la función aproximada por series de Taylor. Para el producto interno $\langle f | g \rangle = \int_{-1}^1 f(x) g(x) dx$ el error será

$$\langle f - f_{\text{Aprox}} | f - f_{\text{Aprox}} \rangle = \int_{-1}^1 (f - f_{\text{Aprox}})^2 dx$$

Esto es para el caso de $n=$

> n;

7

(5)

> ErrorTaylor := sqrt(int((f(x)-fAproxTaylor[n])^2,x=-1..1)); evalf(%);

$$\begin{aligned} \text{ErrorTaylor} := & \frac{1}{6930} \left(8004150 \sqrt{\pi} \left(\frac{4}{\sqrt{\pi}} + \frac{\cos(6)}{3\sqrt{\pi}} - \frac{\sin(6)}{18\sqrt{\pi}} \right) - 96049800 \pi \left(-\text{BesselJ}(0,3) + \frac{2 \text{BesselJ}(1,3)}{3} \right) + 85377600 \pi \left(\frac{3 \text{BesselJ}(0,3)}{4} \right. \right. \\ & \left. \left. + \frac{7 \text{BesselJ}(1,3)}{4} \right) - 24664640 \pi \left(-\frac{45 \text{BesselJ}(0,3)}{8} - 3 \text{BesselJ}(1,3) \right) + 44993718 \right)^{1/2} \\ & 0.3330194739 \end{aligned}$$

(6)

y para el caso de n

```
> nn;
```

9

(7)

```
> ErrorTaylor2 := sqrt(int((f(x)-fAproxTaylor2[nn])^2,x=-1..1)); evalf(%);
```

$$\begin{aligned} \text{ErrorTaylor2} := & \frac{1}{20790} \left(308314512 + 72037350 \sqrt{\pi} \left(\frac{4}{\sqrt{\pi}} + \frac{\cos(6)}{3\sqrt{\pi}} - \frac{\sin(6)}{18\sqrt{\pi}} \right) - 864448200 \pi \left(-\text{BesselJ}(0,3) + \frac{2 \text{BesselJ}(1,3)}{3} \right) \right. \\ & + 768398400 \pi \left(\frac{3 \text{BesselJ}(0,3)}{4} + \frac{7 \text{BesselJ}(1,3)}{4} \right) - 221981760 \pi \left(-\frac{45 \text{BesselJ}(0,3)}{8} - 3 \text{BesselJ}(1,3) \right) + 27104000 \pi \left(\frac{1233 \text{BesselJ}(0,3)}{16} \right. \\ & \left. \left. + \frac{1947 \text{BesselJ}(1,3)}{32} \right) \right)^{1/2} \end{aligned}$$

0.05870966633

(8)

Claramente, para este caso el error disminuye un orden de magnitud al aumetar un orden en la aproximación. En ambos casos las desviaciones ocurren en los extremos de la función

Taylor (la base de monomios) no es ortogonal

Para el producto interno $\langle f | g \rangle = \int_{-1}^1 f(x) g(x) dx$ esta base de monomios no es ortogonal

```
> Int(x^l*x^m,x=-1..1)=int(x^l*x^m,x=-1..1);
```

$$\int_{-1}^1 x^l x^m dx = \frac{(-1)^{l+m} + 1}{l+m+1}$$

(9)

Miren el caso particular para $l = 14$ y $m = 28$

```
> Int(x^l*x^m,x=-1..1)=int(x^l*x^m,x=-1..1);
Prodintern:=int(x^l*x^m,x=-1..1);subs({l=14,m=28},Prodintern);
```

$$\int_{-1}^1 x^l x^m dx = \frac{(-1)^{l+m} + 1}{l+m+1}$$

$$Prodintern := \frac{(-1)^{l+m} + 1}{l+m+1}$$

$$\frac{2}{43}$$

(10)

Ortogonalización de la base de monomios

Supongamos que ahora ortogonalizamos esa base de monomios $\{1, x, x^2, x^3, x^4, x^5, x^6 \dots\}$ mediante del procedimiento de Gram Schmidt. Podemos construir los coeficientes a partir de la definición de producto interno $\langle f | g \rangle = \int_{-1}^1 f(x) g(x) dx$

Para ello almacenamos en un arreglo la base de monomios que es, en general, la base "oblicua"

```
> KK := n-2;
  for k from 0 to KK do BaseOblic[k] := x^k end do;
```

```
KK := 5
BaseOblic0 := 1
BaseOblic1 := x
BaseOblic2 := x2
BaseOblic3 := x3
BaseOblic4 := x4
BaseOblic5 := x5
```

(11)

y luego ortogonalizamos mediante el procedimiento de Gram-Schmidt para obtener

```
> BaseOrtog[0] := BaseOblic[0];
BaseOrtog[1] := BaseOblic[1] - (simplify(int(BaseOrtog[0]*BaseOblic[1], x=-1..1)/int(BaseOrtog[0]^2, x=-1..1))
)*BaseOblic[0];
BaseOrtog[2] := BaseOblic[2] - (simplify(int(BaseOrtog[1]*BaseOblic[2], x=-1..1)/int(BaseOrtog[1]^2, x=-1..1))
)*BaseOblic[1]
```

```

- (simplify (int (BaseOrtog[0] * BaseOblic[2], x=-1..1) / int (BaseOrtog[0]^2, x=-1..1)
)) * BaseOrtog[0];
BaseOrtog[3] := BaseOblic[3] - (simplify (int (BaseOrtog[2] * BaseOblic[3], x=-1..1) / int (BaseOrtog[2]^2, x=-1..1)
)) * BaseOrtog[2]
- (simplify (int (BaseOrtog[1] * BaseOblic[3], x=-1..1) / int (BaseOrtog[1]^2, x=-1..1)
)) * BaseOrtog[1]
- (simplify (int (BaseOrtog[0] * BaseOblic[3], x=-1..1) / int (BaseOrtog[0]^2, x=-1..1)
)) * BaseOrtog[0];

```

$$\begin{aligned}
 \text{BaseOrtog}_0 &:= 1 \\
 \text{BaseOrtog}_1 &:= x \\
 \text{BaseOrtog}_2 &:= x^2 - \frac{1}{3} \\
 \text{BaseOrtog}_3 &:= x^3 - \frac{3}{5}x
 \end{aligned}$$

(12)

Pero la gracia del método de Gram-Schmidt es que es algorítmico y se puede programar. Por lo tanto es más inteligente hacer

```

> BaseOrtog[0] := BaseOblic[0];
for u from 1 to KK do
  BaseOrtog[u] := BaseOblic[u] - add( (simplify (int (BaseOrtog[v-1] * BaseOblic[u], x=-1..1) / int (BaseOrtog[v-1]^2, x=-1..1)
)) * BaseOrtog[v-1], v=1..u)
end do;

```

$$\begin{aligned}
 \text{BaseOrtog}_0 &:= 1 \\
 \text{BaseOrtog}_1 &:= x \\
 \text{BaseOrtog}_2 &:= x^2 - \frac{1}{3} \\
 \text{BaseOrtog}_3 &:= x^3 - \frac{3}{5}x
 \end{aligned}$$

$$\begin{aligned} \text{BaseOrtog}_4 &:= x^4 + \frac{3}{35} - \frac{6}{7} x^2 \\ \text{BaseOrtog}_5 &:= x^5 + \frac{5}{21} x - \frac{10}{9} x^3 \end{aligned} \quad (13)$$

Por construcción serán ortogonales. Miren

```
> int (BaseOrtog[0]*BaseOrtog[1], x=-1..1);
  int (BaseOrtog[1]*BaseOrtog[2], x=-1..1);
  int (BaseOrtog[1]*BaseOrtog[3], x=-1..1);
```

$$\begin{matrix} 0 \\ 0 \\ 0 \end{matrix} \quad (14)$$

Son verdaderamente ortogonales

y podemos expresar la función f como combinación lineal de esta base ortogonal $|f\rangle = C_0|P_0\rangle + C_1|P_1\rangle + C_2|P_2\rangle + C_3|P_3\rangle + C_4|P_4\rangle + \dots$ Donde, gracias a los $|P_n\rangle$ son ortogonales, los coeficientes se pueden calcular facilmente y uno por uno como $C_n = \frac{\langle f | P_n \rangle}{\langle P_n | P_n \rangle}$ Por lo tanto las proyecciones a lo largo de cada uno de los vectores base serán

```
> BaseOrtog[0]; C[0]:=Int(f(x)*BaseOrtog[0], x=-1..1)/Int(BaseOrtog[0]^2, x=-1..1) =
  int(f(x)*BaseOrtog[0], x=-1..1)/int(BaseOrtog[0]^2, x=-1..1);
```

$$C_0 := \frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) \, dx}{\int_{-1}^1 1 \, dx} = 0 \quad (15)$$

```
> BaseOrtog[1]; C[1]:=Int(f(x)*BaseOrtog[1], x=-1..1)/Int(BaseOrtog[1]^2, x=-1..1)
  =int(f(x)*BaseOrtog[1], x=-1..1)/int(BaseOrtog[1]^2, x=-1..1); evalf(rhs(%));
```

$$C_1 := \frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) x \, dx}{\int_{-1}^1 x^2 \, dx} = \frac{\pi \left(-\text{BesselJ}(0, 3) + \frac{2 \text{BesselJ}(1, 3)}{3} \right)}{2}$$

0.7635503665

(16)

```
> BaseOrtog[2];
```

```
C[2] := Int(f(x)*BaseOrtog[2], x=-1..1)/Int(BaseOrtog[2]^2, x=-1..1) = int(f(x)*BaseOrtog[2], x=-1..1)/int(BaseOrtog[2]^2, x=-1..1)
```

$$C_2 := \frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) \left(x^2 - \frac{1}{3}\right) dx}{\int_{-1}^1 \left(x^2 - \frac{1}{3}\right)^2 dx} = 0$$

(17)

Otra vez la hacemos de forma algorítmica y para ello almacenamos los coeficientes en un arreglo y los calculamos

```
> for u from 0 to KK do
```

```
  C[u] := evalf(simplify( int(f(x)*BaseOrtog[u], x=-1..1)/int(BaseOrtog[u]^2, x=-1..1) ))
```

```
end do;
```

$$\begin{aligned} C_0 &:= 0. \\ C_1 &:= 0.7635503670 \\ C_2 &:= 0. \\ C_3 &:= -2.625793952 \\ C_4 &:= 0. \end{aligned}$$

$$C_5 := 2.521874335$$

(18)

y la combinación lineal será $|f\rangle = C_0|P_0\rangle + C_1|P_1\rangle + C_2|P_2\rangle + C_3|P_3\rangle + C_4|P_4\rangle + \dots$


```
> fAproxOrto1 := add(evalf(C[u])*BaseOrtog[u], u=0..KK);
```

$$fAproxOrto1 := 2.939473008x - 5.427876546x^3 + 2.521874335x^5$$

(19)

y graficamos esta aproximación con la función original

```
> plot([f(x), fAproxTaylor[n], simplify(fAproxOrto1)], x=-1..1, legend=["f", "fAproxTaylor", "fAproxOrtog"]);
```



y los errores serán

En Taylor

```
> "n"=n;"error Taylor "=evalf(ErrorTaylor);
```

"n"=7

```
"error Taylor " = 0.3330194739
```

(20)

```
> "n"=nn; "error Taylor " = evalf(ErrorTaylor2) ;
```

```
"n" = 9  
"error Taylor " = 0.05870966633
```

(21)

En los polinomios de Legendre (polinomios ortogonalizados)

```
> "n"=KK; ErrorOrto1 := evalf(sqrt(int((f(x)-simplify(fAproxOrto1))^2,x=-1..1)));
```

```
"n" = 5  
ErrorOrto1 := 0.009348849336
```

(22)

la expansión en término de funciones ortogonales es mucho mas precisa. En particular 10 veces mas precisa.

¿La expansión en Taylor es la única expansión en una base oblicua de monomios?

Retomamos la base oblicua de monomio y hacemos la expansión de la función como combinación lineal en esa base. Es decir, no utilizamos la receta de Taylor sino que buscamos los coeficientes a partir de proyecciones y resolver un sistema de ecuaciones

```
> fAproxObli := add(Cmono[u]*BaseOblic[u], u=0..KK);
```

$$fAproxObli := Cmono_5 x^5 + Cmono_4 x^4 + Cmono_3 x^3 + Cmono_2 x^2 + Cmono_1 x + Cmono_0$$

(23)

y proyectando con cada uno de los elementos de la base se obtiene el sistema de ecuaciones algebraicas, donde los coeficientes serán las incógnitas

```
> for u from 0 to KK do  
  EcuacMono[u] := (int(f(x)*BaseOblic[u],x=-1..1,numeric)) =  
  add(Cmono[v]*evalf(int(BaseOblic[v]*BaseOblic[u],x=-1..1,numeric)), v=0..KK);  
end do;
```

$$\begin{aligned}
 EcuacMono_0 &:= 0. = 2. \cdot Cmono_0 + 0.6666666667 \cdot Cmono_2 + 0.4000000000 \cdot Cmono_4 \\
 EcuacMono_1 &:= 0.5090335777 = 0.6666666667 \cdot Cmono_0 + 0.4000000000 \cdot Cmono_3 + 0.2857142857 \cdot Cmono_5 \\
 EcuacMono_2 &:= 0. = 0.6666666667 \cdot Cmono_0 + 0.4000000000 \cdot Cmono_2 + 0.2857142857 \cdot Cmono_4 \\
 EcuacMono_3 &:= 0.1853838518 = 0.4000000000 \cdot Cmono_0 + 0.2857142857 \cdot Cmono_3 + 0.2222222222 \cdot Cmono_5 \\
 EcuacMono_4 &:= 0. = 0.4000000000 \cdot Cmono_0 + 0.2857142857 \cdot Cmono_2 + 0.2222222222 \cdot Cmono_4 \\
 EcuacMono_5 &:= 0.09217724939 = 0.2857142857 \cdot Cmono_1 + 0.2222222222 \cdot Cmono_3 + 0.1818181818 \cdot Cmono_5
 \end{aligned} \tag{24}$$

Construimos una lista con las ecuaciones

```
> ListEcuacMono := {seq(EcuacMono[w], w=0..KK)};
```

$$\begin{aligned}
 ListEcuacMono := \{ & 0. = 0.4000000000 \cdot Cmono_0 + 0.2857142857 \cdot Cmono_2 + 0.2222222222 \cdot Cmono_4, 0. = 0.6666666667 \cdot Cmono_0 + 0.4000000000 \cdot Cmono_2 \\
 & + 0.2857142857 \cdot Cmono_4, 0. = 2. \cdot Cmono_0 + 0.6666666667 \cdot Cmono_2 + 0.4000000000 \cdot Cmono_4, 0.09217724939 = 0.2857142857 \cdot Cmono_1 \\
 & + 0.2222222222 \cdot Cmono_3 + 0.1818181818 \cdot Cmono_5, 0.1853838518 = 0.4000000000 \cdot Cmono_0 + 0.2857142857 \cdot Cmono_3 + 0.2222222222 \cdot Cmono_5, \\
 & 0.5090335777 = 0.6666666667 \cdot Cmono_0 + 0.4000000000 \cdot Cmono_3 + 0.2857142857 \cdot Cmono_5 \}
 \end{aligned} \tag{25}$$

```
> ListIncogMono := {seq(Cmono[w], w=0..KK)};
```

$$ListIncogMono := \{Cmono_0, Cmono_1, Cmono_2, Cmono_3, Cmono_4, Cmono_5\} \tag{26}$$

Resolvemos el sistema ecuaciones

```
> soluc:= fsolve(ListEcuacMono,ListIncogMono);assign(soluc);
```

$$soluc := \{Cmono_0 = 0., Cmono_1 = 2.939473037, Cmono_2 = 0., Cmono_3 = -5.427876682, Cmono_4 = -0., Cmono_5 = 2.521874458\} \tag{27}$$

y la función aproximada será

```
> fAproxObli;
```

$$2.521874458 x^5 - 5.427876682 x^3 + 2.939473037 x \tag{28}$$

y si la comparamos con la expansión en Taylor

```
> fAproxTaylor[n];
```

$$3x - 6x^3 + \frac{39}{10}x^5 \quad (29)$$

Nótese la diferencia entre la expansión por Taylor y la expansión por monomios todas truncadas en el mismo punto.


Por su parte la aproximación por funciones ortogonales (polinomios de Legendre)

```
> fAproxOrto1;
```

$$2.939473008x - 5.427876546x^3 + 2.521874335x^5 \quad (30)$$

y la gráfica será

```
> plot([f(x), fAproxTaylor[n], fAproxObli, fAproxOrto1], x=-1..1, legend=["f", "fAproxTaylor", "fAproxMono", "fAproxLegendre"]);
```



Calculamos entonces los errores correspondientes

Error de Taylor

```
>evalf(ErrorTaylor) ;
```

0.3330194739

(31)

Error de la aproximación de monomios

```
>ErrorMono := evalf(sqrt(int((f(x)-fAproxObli)^2,x=-1..1)));
```

ErrorMono := 0.009348849453

(32)

Error de Legendre

```
>ErrorOrtol;
```

0.009348849336

(33)

Taylor resulta un fiasco y la proyección sobre los monomios casi es igual que la expansión por polinomios ortogonales. Como demostramos la aproximación en polinomios ortogonales es mejor, pero solo un poco

Base de funciones exponenciales

Igual nos hubiera pasado con otro conjunto de funciones linealmente independiente

```
>exp0 := 1;exp1 := exp(x);exp2 := exp(2*x);exp3 := exp(3*x); exp4 := exp(4*x);exp5 := exp(5*x);exp6 := exp(6*x);  
exp7 := exp(7*x);exp8 := exp(8*x);exp9 := exp(9*x);
```

$$\begin{aligned}
 \exp0 &:= 1 \\
 \exp1 &:= e^x \\
 \exp2 &:= e^{2x} \\
 \exp3 &:= e^{3x} \\
 \exp4 &:= e^{4x} \\
 \exp5 &:= e^{5x} \\
 \exp6 &:= e^{6x} \\
 \exp7 &:= e^{7x} \\
 \exp8 &:= e^{8x} \\
 \exp9 &:= e^{9x}
 \end{aligned}$$

(34)

las cuales claramente no son ortogonales

> Int (exp0*exp1,x=-1..1)=int (exp0*exp1,x=-1..1) ; Int (exp0*exp2,x=-1..1)=int (exp0*exp1,x=-1..1) ; Int (exp0*exp3,x=-1..1)=int (e

$$\begin{aligned}
 \int_{-1}^1 e^x dx &= -e^{-1} + e \\
 \int_{-1}^1 e^{2x} dx &= -e^{-1} + e \\
 \int_{-1}^1 e^{3x} dx &= -e^{-1} + e
 \end{aligned}$$

(35)

> Int (exp1*exp2,x=-1..1)=int (exp1*exp2,x=-1..1) ; Int (exp1*exp3,x=-1..1)=int (exp1*exp3,x=-1..1) ;

$$\int_{-1}^1 e^x e^{2x} dx = -\frac{e^{-3}}{3} + \frac{e^3}{3}$$

$$\int_{-1}^1 e^x e^{3x} dx = -\frac{e^{-4}}{4} + \frac{e^4}{4}$$

(36)

Angulos entre las funciones exponenciales

```
> cosangulo01 := int(exp0*exp1, x=-1..1)/(sqrt(int(exp0*exp0, x=-1..1))*sqrt(int(exp1*exp1, x=-1..1)));evalf(%);
```

$$\text{cosangulo01} := \frac{(-e^{-1} + e) \sqrt{2}}{\sqrt{-2e^{-2} + 2e^2}}$$

$$0.8726936206$$

(37)

el ángulo entre exp0 y exp1

```
> arccos(%);
```

$$0.5101042292$$

(38)

```
> cosangulo02 := int(exp0*exp2, x=-1..1)/(sqrt(int(exp0*exp0, x=-1..1))*sqrt(int(exp2*exp2, x=-1..1)));evalf(%);
```

$$\text{cosangulo02} := \frac{\left(-\frac{e^{-2}}{2} + \frac{e^2}{2}\right) \sqrt{2}}{\sqrt{-e^{-4} + e^4}}$$

$$0.6942721294$$

(39)

el ángulo entre exp0 y exp2

```
> arccos(%);
```

$$0.8033882551$$

(40)

```
> cosangulo03 := int(exp0*exp3, x=-1..1)/(sqrt(int(exp0*exp0, x=-1..1))*sqrt(int(exp3*exp3, x=-1..1)));evalf(%);
```

$$\text{cosangulo03} := \frac{3 \left(-\frac{e^{-3}}{3} + \frac{e^3}{3} \right) \sqrt{2}}{\sqrt{-6 e^{-6} + 6 e^6}} = 0.5759209299 \quad (41)$$

el ángulo entre exp0 y exp3

```
> arccos(%);
```

$$0.9570661128 \quad (42)$$

Expansión en la base oblicua de funciones exponenciales

Como las funciones exponenciales son linealmente independientes podemos plantear nos una expansión del tipo $f = C_0 \cdot 1 + C_1 \cdot \exp(x) + C_2 \cdot \exp(2x) + C_3 \cdot \exp(3x) + \dots$. Pero como la base $\{1, \exp(x), \exp(2x), \exp(3x), \dots\}$ no es ortogonal la expansión de una función nos genera un sistema de ecuaciones para despejar los coeficientes C .

Construimos entonces la base oblicua de funciones exponenciales

```
> for k from 0 to KK do BaseOblic3[k] := exp(k*x); end do;
```

$$\begin{aligned} \text{BaseOblic3}_0 &:= 1 \\ \text{BaseOblic3}_1 &:= e^x \\ \text{BaseOblic3}_2 &:= e^{2x} \\ \text{BaseOblic3}_3 &:= e^{3x} \\ \text{BaseOblic3}_4 &:= e^{4x} \\ \text{BaseOblic3}_5 &:= e^{5x} \end{aligned} \quad (43)$$

y la expansión en esta base de funciones oblicuas

```
> fAproxObli3 := add(CexpOblic[u]*BaseOblic3[u], u=0..KK);
```

$$fAproxObli3 := CexpOblic_0 + CexpOblic_1 e^x + CexpOblic_2 e^{2x} + CexpOblic_3 e^{3x} + CexpOblic_4 e^{4x} + CexpOblic_5 e^{5x} \quad (44)$$

Al igual que hicimos arriba podemos expandir nuestra función en término de esa base oblicua

```
> for u from 0 to KK do
  EcuacExpOblic[u] := (int(f(x)*BaseOblic3[u], x=-1..1, numeric)) =
  add(CexpOblic[v]*evalf(int(BaseOblic3[v]*BaseOblic3[u], x=-1..1, numeric)), v=0..KK);
end do;
```

$$\begin{aligned} EcuacExpOblic_0 &:= 0. = 2. CexpOblic_0 + 2.350402387 CexpOblic_1 + 3.626860408 CexpOblic_2 + 6.678583285 CexpOblic_3 + 13.64495860 CexpOblic_4 \\ &+ 29.68128423 CexpOblic_5 \\ EcuacExpOblic_1 &:= 0.5407098789 = 2.350402387 CexpOblic_0 + 3.626860408 CexpOblic_1 + 6.678583285 CexpOblic_2 + 13.64495860 CexpOblic_3 \\ &+ 29.68128423 CexpOblic_4 + 67.23771912 CexpOblic_5 \\ EcuacExpOblic_2 &:= 1.291253516 = 3.626860408 CexpOblic_0 + 6.678583285 CexpOblic_1 + 13.64495860 CexpOblic_2 + 29.68128423 CexpOblic_3 \\ &+ 67.23771912 CexpOblic_4 + 156.6617495 CexpOblic_5 \\ EcuacExpOblic_3 &:= 2.573532407 = 6.678583285 CexpOblic_0 + 13.64495860 CexpOblic_1 + 29.68128423 CexpOblic_2 + 67.23771912 CexpOblic_3 \\ &+ 156.6617495 CexpOblic_4 + 372.6197064 CexpOblic_5 \\ EcuacExpOblic_4 &:= 5.004651834 = 13.64495860 CexpOblic_0 + 29.68128423 CexpOblic_1 + 67.23771912 CexpOblic_2 + 156.6617495 CexpOblic_3 \\ &+ 372.6197064 CexpOblic_4 + 900.3426449 CexpOblic_5 \\ EcuacExpOblic_5 &:= 9.872377974 = 29.68128423 CexpOblic_0 + 67.23771912 CexpOblic_1 + 156.6617495 CexpOblic_2 + 372.6197064 CexpOblic_3 \\ &+ 900.3426449 CexpOblic_4 + 2202.646575 CexpOblic_5 \end{aligned} \quad (45)$$

Una vez mas, construimos una lista con las ecuaciones

```
> ListEcuacExpOblic := {seq(EcuacExpOblic[w], w=0..KK)};
```

$$\begin{aligned}
 \text{ListEcuacExpOblic} := \{ & 0. = 2. \text{CexpOblic}_0 + 2.350402387 \text{CexpOblic}_1 + 3.626860408 \text{CexpOblic}_2 + 6.678583285 \text{CexpOblic}_3 + 13.64495860 \text{CexpOblic}_4 \\
 & + 29.68128423 \text{CexpOblic}_5, 0.5407098789 = 2.350402387 \text{CexpOblic}_0 + 3.626860408 \text{CexpOblic}_1 + 6.678583285 \text{CexpOblic}_2 \\
 & + 13.64495860 \text{CexpOblic}_3 + 29.68128423 \text{CexpOblic}_4 + 67.23771912 \text{CexpOblic}_5, 1.291253516 = 3.626860408 \text{CexpOblic}_0 + 6.678583285 \text{CexpOblic}_1 \\
 & + 13.64495860 \text{CexpOblic}_2 + 29.68128423 \text{CexpOblic}_3 + 67.23771912 \text{CexpOblic}_4 + 156.6617495 \text{CexpOblic}_5, 2.573532407 = 6.678583285 \text{CexpOblic}_0 \\
 & + 13.64495860 \text{CexpOblic}_1 + 29.68128423 \text{CexpOblic}_2 + 67.23771912 \text{CexpOblic}_3 + 156.6617495 \text{CexpOblic}_4 + 372.6197064 \text{CexpOblic}_5, \\
 & 5.004651834 = 13.64495860 \text{CexpOblic}_0 + 29.68128423 \text{CexpOblic}_1 + 67.23771912 \text{CexpOblic}_2 + 156.6617495 \text{CexpOblic}_3 + 372.6197064 \text{CexpOblic}_4 \\
 & + 900.3426449 \text{CexpOblic}_5, 9.872377974 = 29.68128423 \text{CexpOblic}_0 + 67.23771912 \text{CexpOblic}_1 + 156.6617495 \text{CexpOblic}_2 + 372.6197064 \text{CexpOblic}_3 \\
 & + 900.3426449 \text{CexpOblic}_4 + 2202.646575 \text{CexpOblic}_5 \}
 \end{aligned}$$

(46)

```
> ListIncogExpOblic := {seq(CexpOblic[w], w=0..KK )};
```

$$\text{ListIncogExpOblic} := \{ \text{CexpOblic}_0, \text{CexpOblic}_1, \text{CexpOblic}_2, \text{CexpOblic}_3, \text{CexpOblic}_4, \text{CexpOblic}_5 \}$$

(47)

```
> solucExpOblic:= fsolve(ListEcuacExpOblic,ListIncogExpOblic);assign(solucExpOblic);
```

$$\begin{aligned}
 \text{solucExpOblic} := \{ & \text{CexpOblic}_0 = 7.783863816, \text{CexpOblic}_1 = -37.29378628, \text{CexpOblic}_2 = 56.78012295, \text{CexpOblic}_3 = -37.13200709, \text{CexpOblic}_4 \\
 & = 11.07660084, \text{CexpOblic}_5 = -1.246457563 \}
 \end{aligned}$$

(48)

finalmente la expansión en la base oblicua de funciones exponenciales será

```
> fAproxObli3;
```

$$7.783863816 - 37.29378628 e^x + 56.78012295 e^{2x} - 37.13200709 e^{3x} + 11.07660084 e^{4x} - 1.246457563 e^{5x}$$

(49)

Bases Exponenciales ortogonales

Igual las podemos ortogonalizar, y para empezar almacenamos las funciones de la base oblicuas en un arreglo y ortogonalizamos con Gram-Schmidt

```
> BaseOrtog3[0] := BaseOblic3[0];
  for u from 1 to KK do
    BaseOrtog3[u] := (BaseOblic3[u] - add( (simplify(int(BaseOrtog3[v-1]*BaseOblic3[u], x=-1..1)/int(BaseOrtog3[v-1]^2, x=-1..1)) * BaseOrtog3[v-1], v=1..u))
  end do;
```

$$\begin{aligned}
& \text{BaseOrtog}_{3_0} := 1 \\
& \text{BaseOrtog}_{3_1} := e^x + \frac{e^{-1}}{2} - \frac{e}{2} \\
& \text{BaseOrtog}_{3_2} := e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \\
& \text{BaseOrtog}_{3_3} := e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} - \frac{e^{-2}(e^6 + 3e^4 + 3e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \\
& \quad - \frac{e^{-1}(5e^6 - 39e^4 - 57e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5e^4 - 40e^2 - 25} \\
& \text{BaseOrtog}_{3_4} := e^{4x} + \frac{e^{-4}}{8} - \frac{e^4}{8} - \frac{e^{-3}(3e^8 + 8e^6 + 8e^4 + 8e^2 + 13) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{40} \\
& \quad - \frac{(9e^6 - 72e^4 - 57e^2 - 120e^2 - 120) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{10e^4 - 80e^2 - 50} \\
& \quad - \frac{1}{-35e^6 + 189e^4 + 567e^2 + 119} \left(-57e^7 + 276e^5 + 1080e^3 + 201e^{-1} + 1020e \right) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} \right. \\
& \quad \left. - \frac{e^{-2}(e^6 + 3e^4 + 3e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right. \\
& \quad \left. \left. - \frac{e^{-1}(5e^6 - 39e^4 - 57e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5e^4 - 40e^2 - 25} \right) \right) \right) \\
& \text{BaseOrtog}_{3_5} := e^{5x} + \frac{e^{-5}}{10} - \frac{e^5}{10} - \frac{e^{-4}(2e^{10} + 5e^8 + 5e^6 + 5e^4 + 5e^2 + 8) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{30} \\
& \quad + \frac{2e^{-3}(-14e^{10} + 115e^8 + 205e^6 + 235e^4 + 205e^2 + 94) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{35e^4 - 280e^2 - 175} \\
& \quad - \frac{1}{-35e^6 + 189e^4 + 567e^2 + 119} \left(-69e^8 + 315e^6 + 1380e^4 + 1860e^2 + 249e^{-2} + 1305 \right) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} \right. \\
& \quad \left. - \frac{e^{-2}(e^6 + 3e^4 + 3e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right. \\
& \quad \left. \left. - \frac{e^{-1}(5e^6 - 39e^4 - 57e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1}(e^4 + 4e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5e^4 - 40e^2 - 25} \right) \right) \right) \\
& \quad - \frac{1}{-117e^8 + 288e^6 + 3888e^4 + 3168e^2 + 333} \left(-254e^9 + 490e^7 + 8440e^5 + 13640e^3 + 734e^{-1} + 7190e \right) \left(e^{4x} + \frac{e^{-4}}{8} - \frac{e^4}{8} \right.
\end{aligned}$$

(50)

$$\begin{aligned}
& - \frac{e^{-3} (3 e^8 + 8 e^6 + 8 e^4 + 8 e^2 + 13) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{40} \\
& - \frac{(9 e^6 - 72 e^4 - 57 e^{-2} - 120 e^2 - 120) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{10 e^4 - 80 e^2 - 50} \\
& - \frac{1}{-35 e^6 + 189 e^4 + 567 e^2 + 119} \left(-57 e^7 + 276 e^5 + 1080 e^3 + 201 e^{-1} + 1020 e \right) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} \right) \\
& - \frac{e^{-2} (e^6 + 3 e^4 + 3 e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \\
& - \frac{e^{-1} (5 e^6 - 39 e^4 - 57 e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5 e^4 - 40 e^2 - 25} \right) \right) \right) \right)
\end{aligned}$$

```

>int(BaseOrtog3[2]*BaseOrtog3[3],x=-1..1);
int(BaseOrtog3[1]*BaseOrtog3[3],x=-1..1);
int(BaseOrtog3[4]*BaseOrtog3[3],x=-1..1);

```

0
0
0

(51)

y los nuevos coeficientes serán

```

>for u from 0 to KK do
  CexpOrt[u] := (simplify( int(f(x)*BaseOrtog3[u],x=-1..1,numeric)/int(BaseOrtog3[u]^2,x=-1..1,numeric) ))
end do;

```

$$\begin{aligned}
CexpOrt_0 &:= 0. \\
CexpOrt_1 &:= 0.6253405145 \\
CexpOrt_2 &:= -0.6962843792 \\
CexpOrt_3 &:= -0.8541834883 \\
CexpOrt_4 &:= 1.802274230 \\
CexpOrt_5 &:= -1.248527815
\end{aligned} \tag{52}$$


y la combinación lineal

> fAproxOrto3 := add(CexpOrt[u]*BaseOrtog3[u], u=0..KK);

$$\begin{aligned}
fAproxOrto3 &:= 1.802274230 e^{4x} - 0.6962843792 e^{2x} - 0.3126702572 e - 0.8541834883 e^{3x} + 0.3126702572 e^{-1} - 1.248527815 e^{5x} \\
&- \frac{1.802274230 (9 e^6 - 72 e^4 - 57 e^{-2} - 120 e^2 - 120) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{10 e^4 - 80 e^2 - 50} - 0.1248527815 e^{-5} \\
&+ 0.07118195736 e^{-2} (e^6 + 3 e^4 + 3 e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right) + 0.05802369827 e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right) - 0.04505685575 e^{-3} (3 e^8 \\
&+ 8 e^6 + 8 e^4 + 8 e^2 + 13) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right) + 0.6253405145 e^x + 0.1248527815 e^5 - 0.1423639147 e^{-3} + 0.1423639147 e^3 \\
&+ \frac{0.8541834883 e^{-1} (5 e^6 - 39 e^4 - 57 e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5 e^4 - 40 e^2 - 25} + 0.2252842788 e^{-4} \\
&- 0.2252842788 e^4 - 0.1740710948 e^{-2} + 0.1740710948 e^2 - \frac{1}{-35 e^6 + 189 e^4 + 567 e^2 + 119} \left(1.802274230 (-57 e^7 + 276 e^5 + 1080 e^3 \right. \\
&+ 201 e^{-1} + 1020 e) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} - \frac{e^{-2} (e^6 + 3 e^4 + 3 e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right. \\
&\left. \left. - \frac{e^{-1} (5 e^6 - 39 e^4 - 57 e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5 e^4 - 40 e^2 - 25} \right) \right) \\
&+ \frac{1}{-35 e^6 + 189 e^4 + 567 e^2 + 119} \left(1.248527815 (-69 e^8 + 315 e^6 + 1380 e^4 + 1860 e^2 + 249 e^{-2} + 1305) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} \right. \right. \\
&\left. \left. - \frac{e^{-2} (e^6 + 3 e^4 + 3 e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right) \right. \\
&\left. \left. - \frac{e^{-1} (5 e^6 - 39 e^4 - 57 e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5 e^4 - 40 e^2 - 25} \right) \right) + 0.04161759383 e^{-4} (2 e^{10} + 5 e^8 + 5 e^6 \\
&+ 5 e^4 + 5 e^2 + 8) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right) \\
&- \frac{2.497055630 e^{-3} (-14 e^{10} + 115 e^8 + 205 e^6 + 235 e^4 + 205 e^2 + 94) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{35 e^4 - 280 e^2 - 175}
\end{aligned} \tag{53}$$

$$\begin{aligned}
& + \frac{1}{-117 e^8 + 288 e^6 + 3888 e^4 + 3168 e^2 + 333} \left(1.248527815 \left(-254 e^9 + 490 e^7 + 8440 e^5 + 13640 e^3 + 734 e^{-1} + 7190 e \right) \left(e^{4x} + \frac{e^{-4}}{8} - \frac{e^4}{8} \right) \right. \\
& - \frac{e^{-3} (3 e^8 + 8 e^6 + 8 e^4 + 8 e^2 + 13) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{40} \\
& - \frac{(9 e^6 - 72 e^4 - 57 e^{-2} - 120 e^2 - 120) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{10 e^4 - 80 e^2 - 50} \\
& - \frac{1}{-35 e^6 + 189 e^4 + 567 e^2 + 119} \left((-57 e^7 + 276 e^5 + 1080 e^3 + 201 e^{-1} + 1020 e) \left(e^{3x} + \frac{e^{-3}}{6} - \frac{e^3}{6} \right) \right. \\
& - \frac{e^{-2} (e^6 + 3 e^4 + 3 e^2 + 5) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \\
& - \left. \left. \left. \frac{e^{-1} (5 e^6 - 39 e^4 - 57 e^2 - 29) \left(e^{2x} + \frac{e^{-2}}{4} - \frac{e^2}{4} - \frac{e^{-1} (e^4 + 4 e^2 + 7) \left(e^x + \frac{e^{-1}}{2} - \frac{e}{2} \right)}{12} \right)}{5 e^4 - 40 e^2 - 25} \right) \right) \right) \right)
\end{aligned}$$

```
> plot([f(x), fAproxTaylor[n], fAproxOrto1, fAproxObli3, fAproxOrto3], x=-1..1, legend=
["f", "fAproxTaylor", "fAproxLegendre", "fAproxExpObli", "fAproxExpOrto"] );
```



file:///Users/luisnunez/Documents/GitHub/Cursos/MisMateriales/ProgramasScripts/BasesFuncionales/BasesFuncionales.html

El de Taylor

```
>evalf(ErrorTaylor) ;
```

0.3330194739

(54)

El de Legendre

```
>evalf(ErrorOrto1) ;
```

0.009348849336

(55)

El error de la base de funciones exponenciales oblicuas

```
>ErrorOrto3 := (sqrt(int((f(x)-fAproxObli3)^2,x=-1..1,numeric)));
```

ErrorOrto3 := 0.03914518801

(56)

El error de la base de funciones exponenciales ortogonales

```
>ErrorOrto3 := (sqrt(int((f(x)-fAproxOrto3)^2,x=-1..1,numeric)));
```

ErrorOrto3 := 0.03914443568

(57)

Es claro que la expansión en una base de funciones ortogonales, no necesariamente es más precisa. Cuando comparamos polinomios, claramente si lo es.

Polinomios ortogonales predefinidos

En MAPLE los polinomios ortogonales están predefinidos (en una biblioteca orthopoly) y se denotan como $P(n,x)$ donde n es el orden del polinomio y x la variable. Para detalles pueden consultar la hoja de ayuda

```
>with(orthopoly) ;
```

[G, H, L, P, T, U]

(58)

> ?P

> P(0,x);P(1,x);P(2,x);P(3,x);P(4,x);P(5,x);

$$\begin{aligned}
 &1 \\
 &x \\
 &-\frac{1}{2} + \frac{3x^2}{2} \\
 &\frac{5}{2}x^3 - \frac{3}{2}x \\
 &\frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2 \\
 &\frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x
 \end{aligned}$$

(59)

Los Polinomios de Legendre son ortogonales bajo un producto interno definido por $\langle P_n | P_m \rangle = \int_{-1}^1 P_n P_m dx$

> int(P(z,x)*P(m,x),x=-1..1);

$$\int_{-1}^1 P(z,x) P(m,x) dx$$

(60)

> int(P(14,x)*P(28,x),x=-1..1);

0

(61)

> int(P(77,x)*P(77,x),x=-1..1);

$$\frac{2}{155}$$

(62)

Por lo tanto los coeficientes de la expansión serán $\langle f | P_m \rangle = \int_{-1}^1 f(x) P_m(x) dx$

m=0

```
> Int(f(x)*P(0,x),x=-1..1)/Int(P(0,x)*P(0,x),x=-1..1)=int(f(x)*P(0,x),x=-1..1)/int(P(0,x)*P(0,x),x=-1..1);
```

$$\frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) dx}{\int_{-1}^1 1 dx} = 0$$

(63)

m=1

```
> Int(f(x)*P(1,x),x=-1..1)/Int(P(1,x)*P(1,x),x=-1..1)=int(f(x)*P(1,x),x=-1..1)/int(P(1,x)*P(1,x),x=-1..1);
```

$$\frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) x dx}{\int_{-1}^1 x^2 dx} = \frac{\pi \left(-\text{BesselJ}(0, 3) + \frac{2 \text{BesselJ}(1, 3)}{3} \right)}{2}$$

(64)

```
> evalf(%);
```

$$0.7635503665 = 0.7635503665$$

(65)

m=2

```
> Int(f(x)*P(2,x),x=-1..1)/Int(P(2,x)*P(2,x),x=-1..1)=int(f(x)*P(2,x),x=-1..1)/int(P(2,x)*P(2,x),x=-1..1);
```

$$\frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) \left(-\frac{1}{2} + \frac{3x^2}{2}\right) dx}{\int_{-1}^1 \left(-\frac{1}{2} + \frac{3x^2}{2}\right)^2 dx} = 0 \quad (66)$$

m=3

> Int(f(x)*P(3,x),x=-1..1)/Int(P(3,x)*P(3,x),x=-1..1)=int(f(x)*P(3,x),x=-1..1)/int(P(3,x)*P(3,x),x=-1..1);

$$\frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) \left(\frac{5}{2}x^3 - \frac{3}{2}x\right) dx}{\int_{-1}^1 \left(\frac{5}{2}x^3 - \frac{3}{2}x\right)^2 dx} = \frac{35\pi \left(\frac{3 \text{BesselJ}(0,3)}{4} + \frac{7 \text{BesselJ}(1,3)}{4}\right)}{27} - \frac{7\pi \left(-\text{BesselJ}(0,3) + \frac{2 \text{BesselJ}(1,3)}{3}\right)}{4} \quad (67)$$

> evalf(%);

$$-1.050317580 = -1.050317580 \quad (68)$$

m=4

> Int(f(x)*P(4,x),x=-1..1)/Int(P(4,x)*P(4,x),x=-1..1)=int(f(x)*P(4,x),x=-1..1)/int(P(4,x)*P(4,x),x=-1..1);

$$\frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) \left(\frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2\right) dx}{\int_{-1}^1 \left(\frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2\right)^2 dx} = 0 \quad (69)$$

en general los coeficientes de la expansión se expresan

> CoefLegendre[z] := int(f(x)*P(z,x),x=-1..1)/int(P(z,x)*P(z,x),x=-1..1);

$$\text{CoefLegendre}_z := \frac{\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) P(z,x) dx}{\int_{-1}^1 P(z,x)^2 dx} \quad (70)$$

y la aproximación de la función

```
> f(x)=Sum(CoefLegendre[z]*P(z,x),z=0..nn);
```

$$\sqrt{-x^2+1} \sin(3x) = \sum_{z=0}^9 \frac{\left(\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) P(z,x) dx \right) P(z,x)}{\int_{-1}^1 P(z,x)^2 dx} \quad (71)$$

```
> h:=5;
```

$$h := 5 \quad (72)$$

```
> f(x)=Sum(CoefLegendre[z]*P(z,x),z=0..h);
```

$$\sqrt{-x^2+1} \sin(3x) = \sum_{z=0}^5 \frac{\left(\int_{-1}^1 \sqrt{-x^2+1} \sin(3x) P(z,x) dx \right) P(z,x)}{\int_{-1}^1 P(z,x)^2 dx} \quad (73)$$

```
> AproxLegendre[h] := sum(CoefLegendre[z]*P(z,x),z=0..h);
```

$$\begin{aligned} \text{AproxLegendre}_5 := & \frac{\pi \left(-\text{BesselJ}(0,3) + \frac{2 \text{BesselJ}(1,3)}{3} \right) x}{2} \\ & + \frac{7 \left(\frac{10 \pi \left(\frac{3 \text{BesselJ}(0,3)}{4} + \frac{7 \text{BesselJ}(1,3)}{4} \right)}{27} - \frac{\pi \left(-\text{BesselJ}(0,3) + \frac{2 \text{BesselJ}(1,3)}{3} \right)}{2} \right) \left(\frac{5}{2} x^3 - \frac{3}{2} x \right)}{2} \end{aligned} \quad (74)$$

```
> ?BesselJ
```

```
> simplify(AproxLegendre[h]);
```

(75)

```
>plot([f(x),AproxLegendre[h]],x=-1..1, legend=["f","AproxLegendre"]);
```

Otra vez, el error de la aproximación será la norma de la distancia entre los dos vectores. Esto es

```
>ErrorLegendre := sqrt(int((f(x)-AproxLegendre[h])^2,x=-1..1));
```

(76)

es decir

```
>evalf(%);
```

(77)

Otro conjunto de polinomios ortogonales son los polinomios de Tchevichev

```
>T(0,x);T(1,x);T(2,x);T(3,x);T(4,x);T(5,x);
```

(78)

Que se parecen a los de Legendre, pero no son

```
> P(0,x); P(1,x); P(2,x); P(3,x); P(4,x); P(5,x);
```

(79)

Porque el producto interno con el cual ortogonalizamos la base de monomios

es distinto

```
> pesoChevichev := 1/sqrt(1-x^2);
   CoefChevi[m] := int(f(x)*T(m,x)*pesoChevichev,x=-1..1)/int(T(m,x)*T(m,x)*pesoChevichev,x=-1..1);
```

(80)

```
> AproxChevi[h] := sum(CoefChevi[m]*T(m,x),m=0..h);
```

(81)

```
> Taylor[h+2] := taylor(f(x),x=0,h+2); AproxTaylor[h+2] := convert(Taylor[h+2],polynom);
```

(82)

```
> plot([f(x),AproxTaylor[h+2],AproxLegendre[h],AproxChevi[h]],x=-1..1);
```

```
>ErrorChevi := sqrt(int((f(x)-AproxChevi[h])^2*pesoChevichev,x=-1..1));
```

(83)

```
>evalf(%);
```

(84)

Si seleccionamos una función un poco mas compleja

```
>restart: with(plots): with(orthopoly);
```

(85)

```
>f := x -> sqrt(x^4+x^2+1)*sin(3*Pi*x);plot(f(x), x=-1..1);
```

Probamos con Taylor

```
> n:=10;
```

(86)

```
> Taylor[n] := taylor(f(x), x=0, n);
```

(87)

```
> AproxTaylor[n] := convert(Taylor[n], polynom);
```

(88)

```
>plot([f(x),AproxTaylor[n]],x=-1..1,y=-2..2);
```

y con Legendre

```
>CoefLegendre[k]:=int(f(x)*P(k,x),x=-1..1)/int(P(k,x)*P(k,x),x=-1..1);
```

(89)

```
>AproxLegendre[n]:=sum(CoefLegendre[k]*P(k,x),k=0..n);
```

(90)

```
> simplify(%);
```

(91)

```
> AproxLegendreDefinitiva:=evalf(%);
```

(92)

```
> plot([f(x),AproxTaylor[n],AproxLegendreDefinitiva],x=-1..1,y=-2..2,legend=["func","AproxTaylor","AproxLegendre"]);
```

Los errores

```
>ErrorTaylor := sqrt(int((f(x)-AproxTaylor[n])^2,x=-1..1));
```

(93)

```
>evalf(%);
```

(94)

```
>ErrorLegendre := sqrt(int((f(x)-AproxLegendreDefinitiva)^2,x=-1..1));
```

(95)

```
>evalf(%);
```

>

>

>