Richard H. Enns
Computer Algebra Recipes for Mathematical Physics
Brickhauser Boston 2005

Chapter 4

Linear PDEs of Physics

In this chapter, we will solve a wide variety of physical problems involving linear PDEs, first in Cartesian coordinates, then in other curvilinear coordinate systems. The common linear PDEs of physics include:

(1) The *wave equation* (with ψ a continuous scalar function of position $\vec{r}$ and time t, and c the wave velocity),

$$\nabla^2\psi = \frac{1}{c^2}\frac{\partial^2\psi}{\partial t^2}, \tag{4.1}$$

for vibrating elastic strings and membranes, sound waves in fluids, elastic waves in solids, electromagnetic waves (with ψ a vector), etc.

(2) The *diffusion equation* (with d a positive diffusion constant),

$$\frac{\partial\psi}{\partial t} = d\,\nabla^2\psi, \tag{4.2}$$

which applies to time-dependent heat flow, mixing of fluids, neutron diffusion in nuclear reactors, diffusion of impurities in solids, and so on.

(3) *Laplace's equation*,

$$\nabla^2\psi = 0, \tag{4.3}$$

which applies to steady-state heat flow, irrotational flow of an incompressible fluid, and electro(magneto)statics in charge(current)-free regions.

(4) *Poisson's equation* (with S a source term)

$$\nabla^2\psi = S(\vec{r}, t), \tag{4.4}$$

which applies, for example, to the electrostatic potential due to a charge distribution and to the steady-state temperature with a heat source present.

(5) The time-independent *Schrödinger equation* (with $\hbar = h/2\pi$ (h being Planck's constant), m the mass, V the potential, and E the total energy),

$$-\frac{\hbar^2}{2m}\nabla^2\psi + V\,\psi = E\,\psi, \tag{4.5}$$

which describes stationary states in quantum mechanics.

4.1 Three Cheers for the String

When I introduce my physics and engineering students to the PDEs of physics, I always begin with the transverse vibrations of the humble string. The reason is quite simple. The vibrating string is familiar and easy to visualize, the basic underlying equation of motion (the 1-dimensional wave equation) easy to derive, and a wide variety of important methods and ideas can be introduced. So this section illustrates a gourmet selection of string recipes. Lest you get ideas of stringing me up after munching on some of the stringy concoctions, I will show you that there is life beyond the string in the second part of this chapter.

4.1.1 Jennifer Finds the General Solution

Friends are like fiddle strings, they must not be screwed too tight.
English Proverb. Collected in: H. G. Bohn, *A Handbook of Proverbs* (1855)

Consider a light, uniform, stretched string of linear density (mass per unit length) ϵ which is horizontal in equilibrium. The goal of this recipe, provided by Jennifer the MIT mathematician, is to obtain the equation of motion for the transverse (vertical) oscillations of the string and then the general solution.

The PDEtools package is loaded, because it contains the `declare` and `dchange` commands. Jennifer needs the former to introduce subscript notation, favored by mathematicians, for the resulting 1-dimensional wave equation and the latter to transform the variables in arriving at the general solution. The plots package is included because a typical solution will be animated.

```
> restart: with(PDEtools): with(plots):
```

Let the vertical displacement of a point x on the string from equilibrium at time t be $\psi(x,t)$. Consider an infinitesimal element of string of arclength $ds = \sqrt{(dx)^2 + (d\psi)^2} = \sqrt{1 + (\partial\psi/\partial x)^2}\, dx$, located between x and $x + dx$. Since the string is only to move vertically, the horizontal component T of the tension in the string is constant along the string. The vertical component of the tension, which is given by $T\,\partial\psi/\partial x$, will vary along the string. The net vertical force F on the infinitesimal element is equal to the difference between the vertical forces at its ends. This force is entered.

```
> F:=T*Diff(psi(x+dx,t),x)-T*Diff(psi(x,t),x);
```

$$F := T\,(\frac{\partial}{\partial x}\,\psi(x+dx,\,t)) - T\,(\frac{\partial}{\partial x}\,\psi(x,\,t))$$

Since dx is small, F is taylor expanded in powers of dx to order 2 and the order of term removed with the `convert( ,polynom)` command.

```
> F:=convert(taylor(F,dx=0,2),polynom);
```

$$F := T\,\mathrm{D}_1(\psi)(x,\,t) - T\,(\frac{\partial}{\partial x}\,\psi(x,\,t)) + T\,\mathrm{D}_{1,\,1}(\psi)(x,\,t)\,dx$$

Newton's second law is applied to the string element in the vertical direction. The net vertical force F is converted to differential form on the lhs of *ode*. This must be equal to the mass of the string element times the acceleration, i.e., to $\epsilon\, ds\, (\partial^2\psi/\partial t^2)$. Assuming that the vibrations are such that $\partial\psi/\partial x \ll 1$, then $ds \approx dx$. This *linear approximation* is used on the rhs of *ode*.

```
> ode:=convert(F,diff)=epsilon*dx*diff(psi(x,t),t,t);
```

$$ode := T\,(\frac{\partial^2}{\partial x^2}\,\psi(x,\,t))\,dx = \varepsilon\,dx\,(\frac{\partial^2}{\partial t^2}\,\psi(x,\,t))$$

Dividing *ode* by $(dx\,T)$ and substituting $\epsilon = T/c^2$, yields the 1-dimensional wave equation *WE*, with c the wave speed. By using the `declare` command, Jennifer has introduced subscript notation for the derivatives in the wave equation.

```
> declare(psi(x,t)): WE:=subs(epsilon=T/c^2,ode/(dx*T));
```

$$\psi(x,\,t)\ \textit{will now be displayed as}\ \psi$$

$$WE := \psi_{x,\,x} = \frac{\psi_{t,\,t}}{c^2}$$

If proceeding by hand, the general solution of *WE* can be obtained by introducing two new independent variables u and v related to x and t by the transformation $u = x + c\,t$, $v = x - c\,t$. Then one would apply the chain rule of differentiation, e.g., calculating,

$$\frac{\partial\psi}{\partial x} = \frac{\partial\psi}{\partial u}\frac{\partial u}{\partial x} + \frac{\partial\psi}{\partial v}\frac{\partial v}{\partial x} = \frac{\partial\psi}{\partial u} + \frac{\partial\psi}{\partial v},\ \text{and then}\ \frac{\partial^2\psi}{\partial x^2} = \frac{\partial^2\psi}{\partial u^2} + 2\,\frac{\partial^2\psi}{\partial v\partial u} + \frac{\partial^2\psi}{\partial v^2}$$

The second time derivative would be similarly calculated. This hand transformation of the wave equation *WE* can be easily accomplished with Maple as follows. Jennifer enters the transformation *tr* and solves for x and t, thus obtaining the inverse transformation *itr*.

```
> tr:={u=x+c*t,v=x-c*t}: itr:=solve(tr,{x,t});
```

$$itr := \{t = -\frac{v-u}{2\,c},\ x = \frac{u}{2} + \frac{v}{2}\}$$

Then applying the inverse transformation to the wave equation with the `dchange` command, and simplifying, yields the transformed wave equation *WE2*. Note that c is indicated as a parameter (`params=c`) in the command.

```
> WE2:=simplify(dchange(itr,WE,[u,v],params=c));
```

$$WE2 := \psi_{u,\,u} + 2\,\psi_{u,\,v} + \psi_{v,\,v} = \psi_{u,\,u} - 2\,\psi_{u,\,v} + \psi_{v,\,v}$$

The cross-derivative term is isolated in *WE2*, yielding the greatly simplified wave equation *WE3*.

```
> WE3:=isolate(WE2,diff(psi(u,v),u,v));
```

$$WE3 := \psi_{u,\,v} = 0$$

Clearly, the solution of *WE3* is a linear combination of an arbitrary function of u and an arbitrary function of v. This general solution can be obtained by applying the partial differential equation solve command, `pdsolve`, to *WE3*.

```
> sol:=pdsolve(WE3);
```

$$sol := \psi(u,\,v) = \_F2(u) + \_F1(v)$$

The quantities *_F1* and *_F2* denote arbitrary functions. The general solution to the original wave equation follows on substituting the transformation *tr* into the right-hand side of the solution *sol*.

```
> sol2:=psi(x,t)=subs(tr,rhs(sol));
```

$$sol2 := \psi = \_F2(x + c\,t) + \_F1(x - c\,t)$$

Of course, since Jennifer is using Maple, it was not really necessary to mimic the hand calculation. Applying the `pdsolve` command directly to the original wave equation *WE* yields a similar general solution.

```
> sol3:=pdsolve(WE);
```

$$sol3 := \psi = \_F1(x + c\,t) + \_F2(c\,t - x)$$

The solution *sol3* is a *linear superposition* of a general wave form traveling in the negative-x direction (the first term) with speed c and a wave form (the second term) traveling in the positive-x direction with the same speed. The wave forms propagate unchanged in shape since dissipative forces have not been included. As a representative example, Jennifer considers the following specific wave form ψ which is a linear combination of two Gaussian profiles.

```
> psi:=exp(-(x-t)^2)+exp(-(x+t)^2);
```

$$\psi := e^{(-(x-t)^2)} + e^{(-(x+t)^2)}$$

On animating ψ with the following command, the initial frame shows a single pulse of amplitude 2. As the animation progresses, this pulse splits into two pulses of amplitude 1 propagating in opposite directions with speed $c=1$.

```
> animate(psi,x=-10..10,t=0..7,frames=50,numpoints=500);
```

4.1.2 Daniel Separates Strings: I Separate Variables

There are strings in the human heart that had better not be vibrated.
Charles Dickens, English novelist, *Barnaby Rudge* (1841)

In contrast to Dickens's quote, there are strings in my heart which vibrate with pleasure as I watch my young grandson, Daniel, explore his new world. However, given a multi-stranded string, he would likely try to separate it into its constituent strands, rather than carrying out the following scenario. But, who knows? Perhaps, some day, he will separate variables instead of strings.

Young Daniel is playing with a light horizontal string of length L fixed at $x = 0$ and L, i.e., $\psi(0,t) = \psi(L,t) = 0$. Suppose that he cleverly plucks the string (which is initially at rest) in such a way that it has an initial profile $\psi(x,0) = f(x) = h\,x^3\,(L-x)/L^4$. Our task is to use the method of separation of variables to mathematically determine the subsequent motion of the string and then animate the string vibrations, taking $L = 1$ m, $h = 5$ m, and $c = 5$ m/s.

After loading the plots package, needed for the animation,

```
> restart: with(plots):
```

the 1-dimensional wave equation is entered in *pde*. Unlike Jennifer, I will not bother with the subscript notation.

```
> pde:=diff(psi(x,t),x,x)=(1/c^2)*diff(psi(x,t),t,t);
```

$$pde := \frac{\partial^2}{\partial x^2}\,\psi(x,\,t) = \frac{\frac{\partial^2}{\partial t^2}\,\psi(x,\,t)}{c^2}$$

The separation of variables method assumes that the solution can be written as a product of unknown functions, each function depending on only one of the independent variables. For the present problem, this assumption takes the form $\psi(x,t) = X(x)\,T(t)$. Mentally substituting $\psi(x,t)$ into *pde* and dividing by $\psi(x,t)$ would yield $(d^2X(x)/dx^2)/X(x) = (d^2T(t)/dt^2)/(c^2\,T(t))$. The only way this equality can hold is if both sides of the equation are equal to a constant, called the *separation constant*. This then would yield two ODEs, one for determining $X(x)$ and a second for finding $T(t)$. These steps may be achieved by applying the `pdsolve` command to *pde*, but now including the product assumption as a `HINT`.

```
> pdsolve(pde,psi(x,t),HINT=X(x)*T(t));
```

$$(\psi(x,\,t) = X(x)\,T(t))\ \&\text{where}\,[\{\frac{d^2}{dt^2}\,T(t) = \_c_1\,T(t)\,c^2,\ \frac{d^2}{dx^2}\,X(x) = \_c_1\,X(x)\}]$$

The resulting ODEs are shown in the above output, with $\_c_1$ the separation constant. The next step is to solve the ODEs for X and T. If, in addition to supplying the `HINT`, the `INTEGRATE` option is included, the separated ODEs are readily solved with the `pdsolve` command.

```
> pdsolve(pde,psi(x,t),HINT=X(x)*T(t),INTEGRATE);
```

$$(\psi(x,\,t) = X(x)\,T(t))\ \&\text{where}[\{\{T(t) = \_C3\,e^{(\sqrt{\_c_1}\,c\,t)} + \_C4\,e^{(-\sqrt{\_c_1}\,c\,t)}\},$$
$$\{X(x) = \_C1\,e^{(\sqrt{\_c_1}\,x)} + \_C2\,e^{(-\sqrt{\_c_1}\,x)}\}\}]$$

Finally, the product of X and T must be formed to give $\psi(x,t)$. Including the `build` option in the `pdsolve` command accomplishes this step. In the recipes which follow in this and the following chapter, all three options will be included whenever we wish to separate variables. You will begin to really appreciate the power of the `pdsolve` command when we tackle problems in non-Cartesian coordinate systems.

```
> sol:=pdsolve(pde,psi(x,t),HINT=X(x)*T(t),INTEGRATE,build);
```

$$sol := \psi(x,\,t) = e^{(\sqrt{\_c_1}\,c\,t)}\,\_C3\,\_C1\,e^{(\sqrt{\_c_1}\,x)} + \frac{e^{(\sqrt{\_c_1}\,c\,t)}\,\_C3\,\_C2}{e^{(\sqrt{\_c_1}\,x)}}$$
$$+ \frac{\_C4\,\_C1\,e^{(\sqrt{\_c_1}\,x)}}{e^{(\sqrt{\_c_1}\,c\,t)}} + \frac{\_C4\,\_C2}{e^{(\sqrt{\_c_1}\,c\,t)}\,e^{(\sqrt{\_c_1}\,x)}}$$

To determine the four constants *_C1*, etc., the two boundary conditions and the two initial conditions must be applied. In order to do this, the form of the solution will be simplified somewhat. First, let's make the substitution $\_c_1 = -k^2$ on the rhs of *sol* and assume that $k > 0$.

```
> psi:=simplify(subs(_c[1]=-k^2,rhs(sol))) assuming k>0;
```

$$\psi := \_C3\ \_C1\ e^{(k\,(c\,t+x)\,I)} + \_C3\ \_C2\ e^{(k\,(c\,t-x)\,I)} + \_C4\ \_C1\ e^{(-I\,k\,(c\,t-x))} + \_C4\ \_C2\ e^{(-I\,k\,(c\,t+x))}$$

Then ψ is converted to trigonometric form and the result expanded.

```
> psi2:=expand(convert(psi,trig));
```

The lengthy output, which has been suppressed here in the text, involves the terms $\cos(k\,x)$, $\sin(k\,x)$, $\cos(c\,k\,t)$, and $\sin(c\,k\,t)$. To satisfy the boundary condition at $x = 0$, the $\cos(k\,x)$ terms are removed from $\psi 2$ by substituting $\cos(k\,x)=0$. At $x=L$, the boundary condition yields $\sin(k\,L)=0$ or $k=n\,\pi/L$, with $n = 1, 2, 3, \ldots$. This is also substituted. Since the transverse velocity of the string is initially zero, we must also set $\sin(c\,k, t)=0$.

```
> psi3:=subs({cos(k*x)=0,sin(c*k*t)=0,k=n*Pi/L},psi2);
```

On then factoring $\psi 3$, the result shown in $\psi 4$ results.

```
> psi4:=factor(psi3);
```

$$\psi 4 := (-\_C2 + \_C1)\,(\_C3 + \_C4)\cos(\frac{n\,\pi\,c\,t}{L})\sin(\frac{n\,\pi\,x}{L})\,I$$

The following `select` command is used to replace the "ugly" coefficient combination in $\psi 4$ with the symbol A_n. The resulting term, labeled $\psi 5$ here, is the nth term in the infinite series which will represent $\psi(x, t)$.

```
> psi5:=A[n]*select(has,psi4,{sin,cos});
```

$$\psi 5 := A_n \cos(\frac{n\,\pi\,c\,t}{L})\sin(\frac{n\,\pi\,x}{L})$$

The initial shape of the string is entered.

```
> f:=h*x^3*(L-x)/L^4:
```

Using linear superposition, $\psi(x, t) = \sum_{n=1}^{\infty} A_n \sin(n\,\pi\,x/L)\cos(n\,\pi\,c\,t/L)$. At $t = 0$, one has $f = \sum_{n=1}^{\infty} A_n \sin(n\,\pi\,x/L)$. Multiplying this equation by $\sin(m\,\pi\,x/L)$ and integrating from $x=0$ to L, the only term which will survive in the series due to orthogonality of the sine functions is the term corresponding to $n=m$. This procedure is carried out in the following equation.

```
> eq:=int(f*sin(n*Pi*x/L),x=0..L)
      =int(subs(t=0,psi5)*sin(n*Pi*x/L),x=0..L):
```

The equation is then solved for A_n, assuming that n is an integer.

```
> A[n]:=solve(eq,A[n]) assuming n::integer;
```

$$A_n := -\frac{12\,h\,(4 - 4\,(-1)^n + n^2\,\pi^2\,(-1)^n)}{n^5\,\pi^5}$$

The parameter values $L = 1$, $h = 5$, and $c = 5$ are entered and a formal sum performed with the first 25 terms in the series being retained.

```
> L:=1: h:=5: c:=5:
```

```
> psi6:=Sum(psi5,n=1..25);
```

$$\psi 6 := \sum_{n=1}^{25} \left(-\frac{60\,(4 - 4\,(-1)^n + n^2\,\pi^2\,(-1)^n)\cos(5\,n\,\pi\,t)\sin(n\,\pi\,x)}{n^5\,\pi^5}\right)$$

Applying the `value` command to $\psi 6$, the motion of the string is now animated over the time range $t = 0$ to 160 seconds. The opening frame of the animation is shown in Figure 4.1. By executing the command line, you will be able to observe the vibrations of the string.

```
> animate(value(psi6),x=0..L,t=0..160,frames=100,
  thickness=2,scaling=constrained,tickmarks=[4,3]);
```



Figure 4.1: The initial shape of the plucked string.

4.1.3 Daniel Strikes Again: Mr. Fourier Reappears

We are all instruments endowed with feeling and memory. Our senses are so many strings that are struck by surrounding objects and that also frequently strike themselves.
Denis Diderot, French philosopher, (1713–84)

If Daniel didn't separate the strands or otherwise destroy the string, another possible scenario is that he would strike it in some manner. Consider a string fixed at $x = 0$ and L to be initially horizontal, but is struck in such a way that it is given the initial piecewise velocity profile $g(x) = 4\,v\,x/L$ for $x \leq L/4$, $g(x) = 4\,(v/L)\,(L/2 - x)$ for $L/4 \leq x \leq L/2$, and $g(x) = 0$ for $x \geq L/2$. Using the Fourier series approach, determine the subsequent motion of the string and animate it for $L = 20$ cm, $v = 5$ cm/s, and $c = 1$ cm/s.

The plots package is needed for the animation. To perform the integration involving the piece-wise velocity profile, it is necessary to assume that both v and L are positive.

```
> restart: with(plots): assume(v>0,L>0):
```

A functional operator ψ is introduced to generate the necessary Fourier series. From the previous recipe, it is clear that the spatial part should be built up of terms of the form $\sin(n\,\pi\,x/L)$, with $n = 1, 2, 3, ...$, in order to satisfy the boundary conditions at $x=0$ and L. The time part is left quite general, so that the recipe will run even if the initial conditions are changed.

```
> psi:=N->Sum(sin(n*Pi*x/L)*(a[n]*cos(n*Pi*c*t/L)
      +b[n]*sin(n*Pi*c*t/L)),n=1..N);
```

$$\psi := N \to \sum_{n=1}^{N} \sin(\frac{n\,\pi\,x}{L})\,(a_n \cos(\frac{n\,\pi\,c\,t}{L}) + b_n \sin(\frac{n\,\pi\,c\,t}{L}))$$

The given initial spatial $(f(x))$ and velocity $(g(x))$ profiles are entered.

```
> f(x):=0:
  g(x):=piecewise(x<L/4,4*v*x/L,x<L/2,(4*v/L)*(L/2-x),x<L,0);
```

$$g(x) := \begin{cases} \frac{4\,v\,x}{L} & x < \frac{L}{4} \\ \frac{4\,v\,(\frac{L}{2} - x)}{L} & x < \frac{L}{2} \\ 0 & x < L \end{cases}$$

The coefficients a_n and b_n are explicitly evaluated.

```
> a[n]:=(2/L)*int(f(x)*sin(n*Pi*x/L),x=0..L);
```

$$a_n := 0$$

```
> b[n]:=simplify((2/(n*Pi*c))*int(g(x)*sin(n*Pi*x/L),x=0..L));
```

$$b_n := -\frac{8\,v\,L\,(\sin(\frac{n\,\pi}{2}) - 2\sin(\frac{n\,\pi}{4}))}{n^3\,\pi^3\,c}$$

The first 25 terms in the Fourier series are calculated, being left as formal sum.

```
> sol:=psi(25);
```

$$sol := \sum_{n=1}^{25} \left(-\frac{8\sin(\frac{n\,\pi\,x}{L})\,v\,L\,(\sin(\frac{n\,\pi}{2}) - 2\sin(\frac{n\,\pi}{4}))\sin(\frac{n\,\pi\,c\,t}{L})}{n^3\,\pi^3\,c} \right)$$

For animation purposes, the parameter values are substituted into the solution. The transverse velocity of the string is then calculated so that we can decide if 25 terms is sufficient to fit the initial velocity profile and thus ensure an accurate animation of the string.

```
> sol2:=subs({L=20,v=5,c=1},sol): vel:=diff(sol2,t):
```

Substituting the parameter values into $g(x)$ and evaluating the velocity at $t=0$, $g(x)$ and *vel* are now plotted in the same figure, the former curve being colored blue, the latter colored red. The scaling is constrained. The resulting picture is reproduced in Figure 4.2.

```
> plot([subs({L=20,v=5,c=1},g(x)),eval(vel,t=0)],x=0..20,color
  =[blue,red],thickness=2,scaling=constrained,tickmarks=[4,4]);
```



Figure 4.2: Fourier series superimposed on initial velocity profile.

Clearly, 25 terms produces a very good fit to the initial velocity profile. The motion of the string, described by *sol2*, is then animated.

```
> animate(sol2,x=0..20,t=0..50,frames=100,thickness=2,
  scaling=constrained,tickmarks=[4,3]);
```

What do you think happens? You will have to execute the recipe to find out.

4.1.4 The 3-Piece String

What is called an acute knowledge of human nature is mostly nothing but the observer's own weaknesses reflected back from others.
G. C. Lichtenberg, German physicist, philosopher, (1742–99)

In this recipe, we will derive the energy reflection and transmission coefficients for a monochromatic (frequency ω) plane wave traveling from $x=-\infty$ in an infinitely long string which has a different constant linear density in the region $x=0$ to L than in the rest of the string. If the linear density in region 1 ($x<0$) is ϵ, in region 2 ($0<x<L$) is ϵ_2, and in region 3 ($x>L$) is ϵ again, the corresponding wave numbers are $k_1=k=\omega\sqrt{\epsilon/T}$, $k_2=\omega\sqrt{\epsilon_2/T}=r\,k$ with $r\equiv\sqrt{(\epsilon_2/\epsilon)}$, and $k_3=k$, respectively, where T is the tension in the string.

Using complex notation, in region 1 the wave form is $\psi 1=e^{i(kx-\omega t)}+b1\,e^{-i(kx+\omega t)}$. (The physical solution is the real part of $\psi 1$.) The spatial part is now entered, $e^{-i\omega t}$ being omitted since it will cancel.

```
> restart:
> psi||1:=exp(I*k*x)+b1*exp(-I*k*x);
```

$$\psi 1 := e^{(k\,x\,I)} + b1\,e^{(-I\,k\,x)}$$

The first term is the "incident" wave with unit amplitude[1] traveling in the

[1] Since the reflection and transmission coefficients involve ratios, the incident wave amplitude can be chosen to be 1 without any loss of generality.

positive x-direction, the second the "reflected" wave with amplitude $b1$ traveling in the negative x-direction. The energy reflection coefficient, which measures the fraction of the incident wave energy in region 1 which is reflected, then is $R = |b1|^2 = (b1)\,(b1^*)$, where the star denotes complex conjugate.

The spatial part of the solution in region 2, which is a linear combination of plane waves traveling in the positive and negative x directions, is entered. The amplitudes are labeled *a2* and *b2*.

```
> psi||2:=a2*exp(I*r*k*x)+b2*exp(-I*r*k*x);
```

$$\psi 2 := a2\, e^{(r\,k\,x\,I)} + b2\, e^{(-I\,r\,k\,x)}$$

In region 3, there will only be a plane wave traveling in the positive x-direction.

```
> psi||3:=a3*exp(I*k*x);
```

$$\psi 3 := a3\, e^{(k\,x\,I)}$$

The energy transmission coefficient, which measures the fraction of the incident energy in region 1 which is transmitted into region 3, will be given by $T = |a3|^2$. Clearly, since energy must be conserved, one must have $R + T = 1$. To determine R and T, the four unknown coefficients must be determined. The string is continuous, so $\psi 1 = \psi 2$ at $x = 0$ and $\psi 2 = \psi 3$ at $x = L$. These boundary conditions are entered in *eq1* and *eq2*.

```
> eq||1:=eval(psi||1=psi||2,x=0);
```

$$eq1 := 1 + b1 = a2 + b2$$

```
> eq||2:=eval(psi||2=psi||3,x=L);
```

$$eq2 := a2\, e^{(r\,k\,L\,I)} + b2\, e^{(-I\,r\,k\,L)} = a3\, e^{(k\,L\,I)}$$

The continuity of the slopes at $x = 0$ and L is entered in *eq3* and *eq4*.

```
> eq||3:=eval(diff(psi||1=psi||2,x),x=0);
```

$$eq3 := k\,I - b1\,k\,I = a2\,r\,k\,I - b2\,r\,k\,I$$

```
> eq||4:=eval(diff(psi||2=psi||3,x),x=L);
```

$$eq4 := a2\,r\,k\,e^{(r\,k\,L\,I)}\,I - b2\,r\,k\,e^{(-I\,r\,k\,L)}\,I = a3\,k\,e^{(k\,L\,I)}\,I$$

The sequence of four equations is solved for *b1*, *a2*, *b2*, and *a3*.

```
> sol:=solve({seq(eq||i,i=1..4)},{b1,a2,b2,a3}); assign(sol):
```

On assigning the solution, R is calculated by multiplying *b1* by its complex conjugate, employing the complex evaluation command, and simplifying.

```
> R:=simplify(evalc(b||1*conjugate(b||1)));
```

$$R := \frac{(-1 + \cos(r\,k\,L)^2)\,(r^2 - 1)^2}{-2\,r^2\cos(r\,k\,L)^2 - r^4 + r^4\cos(r\,k\,L)^2 - 2\,r^2 - 1 + \cos(r\,k\,L)^2}$$

The energy transmission coefficient $T = (a3)\,(a3^*)$ is similarly calculated.

```
> T:=simplify(evalc(a||3*conjugate(a||3)));
```

$$T := -\frac{4\,r^2}{-1 - 2\,r^2 + \cos(r\,k\,L)^2 - 2\,r^2\cos(r\,k\,L)^2 + r^4\cos(r\,k\,L)^2 - r^4}$$

As a check, let's confirm that the sum $S \equiv R + T = 1$.

```
> S:=simplify(R+T);
```

$$S := 1$$

To get a feeling for the behavior of R and T, let's choose, say, $r=2$, i.e., region 2 has a density 4 times that of regions 1 and 3. Setting $k\,L=z$ in R and T, the reflection and energy coefficients are given by $T1$ and $R1$,

```
> r:=2: T1:=algsubs(k*L=z,T); R1:=algsubs(k*L=z,R);
```

$$T1 := -\frac{16}{-25+9\cos(2\,z)^2} \qquad R1 := \frac{-9+9\cos(2\,z)^2}{-25+9\cos(2\,z)^2}$$

which are then plotted along with $S=1$ over the range $z=0$ to 6.

```
> plot([R1,T1,S],z=0..6,color=[red,blue,green],thickness=2,
  labels=["kL","R,T"],numpoints=100);
```



Figure 4.3: R and T for the 3-piece string as a function of $k\,L$.

The bottom oscillatory curve is R, the top oscillatory curve is T, and the horizontal line is the sum S of the two. Whenever $r\,z \equiv r\,k\,L = n\,\pi$, with $n=0$, 1, 2, ..., the reflection coefficient R is 0 and the transmission coefficient T is 1 (100% transmission).

4.1.5 Encore?

Applause is a receipt, not a bill.
Artur Schnabel, American pianist, on why he didn't give encores, (1882–1951)

Early in my academic career, I was assigned the task of teaching the freshman physics class of several hundred in a huge tiered lecture hall. Being a theoretician, I approached this task with some trepidation, knowing that in order to demonstrate some basic ideas of physics, and to keep the class awake, I would have to intersperse the lecture material with demonstrations. In one demo, a heavy iron ball was suspended from a long vertical rope attached to a hidden catwalk far above the lecture podium. Standing with my back against a 4×8 plywood sheet, I would pull the heavy ball away from the vertical a sizeable distance, bring it up to my chin, and release it. I would stand there bravely, lecturing on the conservation of energy, as the ball completed a few oscillations without, of course, hitting my chin.

This seemed a bit tame, so I decided to add a humorous twist. A departmental assistant was placed out of sight on the catwalk with instructions to give the rope a good heave on the third swing. I would step away from the plywood sheet after the second swing and the iron ball would crash into the plywood. I would then express my relief at having avoided injury by a clear violation of energy conservation. Unfortunately, either the assistant couldn't count or was paid off by the students. He pushed on the second swing! As the ball hurtled towards my chin at an alarming speed, I knew that something was wrong, but I reacted slowly. I got my hands up in time to avoid serious injury, but the momentum of the ball knocked me against the plywood sheet which toppled with a crash to the floor. The students loved it, whistling and cheering and demanding an encore. Somewhat dazed, I declined!

However, I did repeat the demo the next year with a better trained assistant, replacing the plywood with a large plate of old glass painted black so the students didn't know it was glass. This time it was a dazzling success as I stepped away in time and the ball shattered the glass. The down side was that I had to sweep up all the glass, as the janitorial staff refused to do so.

Although the following recipe is not identical to the situation described above, it is inspired by that early exciting demo. A small iron ball of mass M is attached to the lower end of a long vertical rope of length L which has a uniform linear density ϵ. Derive the equation of motion for small transverse oscillations of the rope. Solve the equation of motion for the normal modes of oscillation if the initially vertical rope is given a non-zero transverse velocity. Taking $M = 10$ kg, $L = 10$ m, $\epsilon = 0.1$ kg/m, and the gravitational acceleration $g = 10$ m/s^2, animate the normal mode with the lowest frequency.

The plots library package is needed for the animation and the plottools package required in order to rotate the animation by 90°.

```
> restart: with(plots): with(plottools):
```

Taking the origin at the top of the rope and measuring the vertical distance y downwards, the tension T in the rope will be given by $T = M\,g + \epsilon\,(L - y)\,g$. At the bottom of the rope ($y = L$), the upward tension has to only balance the weight of the ball, but at the top of the rope ($y = 0$) it has to balance both the weight of the ball and the weight of the entire rope.

```
> T:=M*g+epsilon*(L-y)*g;
```

$$T := M\,g + \varepsilon\,(L - y)\,g$$

The transverse (horizontal) oscillations of the rope will be described by the 1-dimensional wave equation $\partial(T\,\partial\psi/\partial y)/\partial y = \epsilon\,\partial^2\psi/\partial t^2$. On entering this PDE, the tension is automatically substituted, yielding the equation of motion *pde*.

```
> pde:=diff(T*diff(psi(y,t),y),y)=epsilon*diff(psi(y,t),t,t);
```

$$pde := -\varepsilon\, g\,(\frac{\partial}{\partial y}\,\psi(y,\,t)) + (M\,g + \varepsilon\,(L-y)\,g)\,(\frac{\partial^2}{\partial y^2}\,\psi(y,\,t)) = \varepsilon\,(\frac{\partial^2}{\partial t^2}\,\psi(y,\,t))$$

Since the rope initially has zero displacement, but a non-zero velocity, a normal-mode solution of the form $\psi(y,t) = X(y)\,\sin(\omega\,t)$ is sought. The `pdsolve` command is applied, with the assumptions that ω, ϵ, M, L, and g are positive, and $y < L$.

```
> sol:=pdsolve(pde,psi(y,t),HINT=X(y)*sin(omega*t),INTEGRATE,
  build) assuming omega::positive, epsilon::positive,
  M::positive,L::positive, g::positive, y<L;
```

$$sol := \psi(y,\,t) = \sin(\omega\,t)\,\_C1\,\mathrm{BesselJ}(0,\,\frac{2\,\omega\,\sqrt{M+\varepsilon\,L-\varepsilon\,y}}{\sqrt{\varepsilon}\,\sqrt{g}}) + \sin(\omega\,t)\,\_C2\,\mathrm{BesselY}(0,\,\frac{2\,\omega\,\sqrt{M+\varepsilon\,L-\varepsilon\,y}}{\sqrt{\varepsilon}\,\sqrt{g}})$$

The spatial part of the solution is a linear combination of zeroth-order Bessel functions of the first and second kinds. The spatial part X is now extracted.

```
> X:=simplify(rhs(sol)/sin(omega*t),symbolic);
```

$$X := \_C1\,\mathrm{BesselJ}(0,\,\frac{2\,\omega\,\sqrt{M+\varepsilon\,L-\varepsilon\,y}}{\sqrt{\varepsilon}\,\sqrt{g}}) + \_C2\,\mathrm{BesselY}(0,\,\frac{2\,\omega\,\sqrt{M+\varepsilon\,L-\varepsilon\,y}}{\sqrt{\varepsilon}\,\sqrt{g}})$$

Assuming that the free-end boundary condition $(dX/dy)|_{y=L} = 0$ applies, we solve for $\_C1$.

```
> _C1:=solve(subs(y=L,diff(X,y))=0,_C1);
```

$$\_C1 := -\frac{\_C2\,\mathrm{BesselY}(1,\,\frac{2\,\omega\,\sqrt{M}}{\sqrt{\varepsilon}\,\sqrt{g}})}{\mathrm{BesselJ}(1,\,\frac{2\,\omega\,\sqrt{M}}{\sqrt{\varepsilon}\,\sqrt{g}})}$$

The parameter values are entered and X divided by the arbitrary constant $\_C2$ and expanded to yield $X2$.

```
> L:=10: M:=10: g:=10: epsilon:=1/10: X2:=expand(X/_C2);
```

$$X2 := -\frac{\mathrm{BesselY}(1,\,2\,\omega\,\sqrt{10})\,\mathrm{BesselJ}(0,\,2\,\omega\,\sqrt{-\frac{y}{10}+11})}{\mathrm{BesselJ}(1,\,2\,\omega\,\sqrt{10})} + \mathrm{BesselY}(0,\,2\,\omega\,\sqrt{-\frac{y}{10}+11})$$

The fixed-end boundary condition at $y=0$ is applied to the numerator of $X2$.

```
> bc:=eval(numer(X2),y=0)=0;
```

$$bc := -\mathrm{BesselY}(1,\ 2\,\omega\,\sqrt{10})\,\mathrm{BesselJ}(0,\ 2\,\omega\,\sqrt{11}) \\ + \mathrm{BesselY}(0,\ 2\,\omega\,\sqrt{11})\,\mathrm{BesselJ}(1,\ 2\,\omega\,\sqrt{10}) = 0$$

The left-hand side of bc is plotted in Figure 4.4 over the frequency range $\omega = 0.1$ to 20 and the vertical view limited so as to be able to clearly see the zeros.

```
> plot(lhs(bc),omega=0.1..20,view=[0..20,-0.2..0.1]);
```



Figure 4.4: Plot of lhs of the boundary condition at $y = 0$ versus frequency.

Guided by the plot, the lowest zero is numerically sought between $\omega = 4$ and 6.

```
> omega:=fsolve(bc,omega=4..6);
```

$$\omega := 5.137764180$$

The lowest possible frequency for a normal mode is $\omega = 5.14$ Hz. The mode with the lowest frequency then is given by ψ. Higher frequency normal modes can be obtained by selecting other zeros in the fixed-end boundary condition.

```
> psi:=X2*sin(omega*t);
```

$$\psi := \left(-\frac{\mathrm{BesselY}(1,\ 10.27552836\,\sqrt{10})\,\mathrm{BesselJ}(0,\ 10.27552836\,\sqrt{-\frac{y}{10}+11})}{\mathrm{BesselJ}(1,\ 10.27552836\,\sqrt{10})} \right. \\ \left. + \mathrm{BesselY}(0,\ 10.27552836\,\sqrt{-\frac{y}{10}+11}) \right) \sin(5.137764180\,t)$$

The solution ψ is animated, the scaling being left unconstrained for better viewing. Using the `rotate` command, the figure is rotated by $-90°$ so that the string is initially vertical and the subsequent time-dependent displacement is horizontal.

```
> rotate(animate(psi,y=0..L,t=0..5,color=red,
  thickness=2,frames=50,numpoints=100),-Pi/2);
```

You will have to execute the worksheet on your computer to see the animation.

4.2 Beyond the String

Many more string examples are given in the **Supplementary Recipes**, but it's time to move on to other physical examples and also look at two- and three-dimensional situations. All the recipes in this section involve Cartesian coordinates. The next section is devoted entirely to non-Cartesian examples.

4.2.1 Heaviside's Telegraph Equation

Get your facts first,
and then you can distort them as much as you please.
Mark Twain, American author on propaganda, (1835–1910)

Consider an electrical transmission line carrying a current $I(x,t)$ that has a uniform inductance L, capacitance C to ground (zero potential), and resistance R (all per unit length) and has a leakage (leakage coefficient G per unit length) proportional to the potential V from the line to ground.

(a) Show that $V(x,t)$ satisfies the *telegraph equation*,

$$\frac{\partial^2 V}{\partial x^2} = L\,C\,\frac{\partial^2 V}{\partial t^2} + (R\,C + L\,G)\frac{\partial V}{\partial t} + R\,G\,V. \tag{4.6}$$

(b) Under what conditions does the telegraph equation reduce to a wave equation? a diffusion equation? Identify the wave velocity for the former, the diffusion coefficient for the latter.

(c) Show that a general solution of the form $V(x,t)=e^{-k\,t}f(\alpha\,x+\beta\,t)$, where f is an arbitrary function, will satisfy (4.6), provided that $R\,C=L\,G$ and k, α, and β take on certain forms which are to be determined. Briefly discuss what this solution means.

This problem was first solved by the English electrical engineer Oliver Heaviside in 1887. Heaviside also reduced Maxwell's equations into the simpler form that we now know, invented the differential operator (D) notation for solving differential equations, predicted the existence of an ionized reflective layer (the Heaviside layer) in the ionosphere which bounces radio signals back to earth, and predicted that the mass of an electric charge would increase with velocity.

Now let's answer the question. In order to use the symbol I for the current, the imaginary unit $\sqrt{-1}$ is set equal to j with the following interface command.

```
> restart: interface(imaginaryunit=j):
```

The rate of change of the voltage V with distance x is given by *eq1*. The first term $-R\,I$ on the rhs is the potential drop per unit length due to the resistance, which is given by Ohm's law. The second term $-L\,\partial I/\partial t$ is the back emf per unit length due to the inductance.

```
> eq1:=diff(V(x,t),x)=-R*I(x,t)-L*diff(I(x,t),t);
```

$$eq1 := \frac{\partial}{\partial x}V(x,\,t) = -R\,I(x,\,t) - L\,(\frac{\partial}{\partial t}\,I(x,\,t))$$

The rate of change of the current I with x is given by *eq2*. The first term on the rhs is the leakage of current per unit length from the line to ground, the second term the capacitive loss per unit length to ground.

```
> eq2:=diff(I(x,t),x)=-G*V(x,t)-C*diff(V(x,t),t);
```

$$eq2 := \frac{\partial}{\partial x} I(x, t) = -G\, V(x, t) - C\, (\frac{\partial}{\partial t} V(x, t))$$

eq1 and *eq2* are differentiated with respect to x, t, respectively, in *eq3* and *eq4*.

```
> eq3:=diff(eq1,x); eq4:=diff(eq2,t);
```

$$eq3 := \frac{\partial^2}{\partial x^2} V(x, t) = -R\, (\frac{\partial}{\partial x} I(x, t)) - L\, (\frac{\partial^2}{\partial x\, \partial t} I(x, t))$$

$$eq4 := \frac{\partial^2}{\partial x\, \partial t} I(x, t) = -G\, (\frac{\partial}{\partial t} V(x, t)) - C\, (\frac{\partial^2}{\partial t^2} V(x, t))$$

A PDE for $V(x,t)$ alone is obtained by substituting *eq2* and *eq4* into *eq3*. Collecting the $\partial V/\partial t$ terms then yields the telegraph equation *TE*.

```
> eq5:=subs({eq2,eq4},eq3);
```

```
> TE:=collect(eq5,diff(V(x,t),t));
```

$$TE := \frac{\partial^2}{\partial x^2} V(x, t) = (R\, C + L\, G)\, (\frac{\partial}{\partial t} V(x, t)) + R\, G\, V(x, t) + L\, C\, (\frac{\partial^2}{\partial t^2} V(x, t))$$

The wave equation *WE* follows on setting $R=0$ and $G=0$ in *TE*, the wave velocity being $c=1/\sqrt{L\,C}$.

```
> WE:=eval(TE,{R=0,G=0});
```

$$WE := \frac{\partial^2}{\partial x^2} V(x, t) = L\, C\, (\frac{\partial^2}{\partial t^2} V(x, t))$$

The diffusion equation *DE* results on setting $L = 0$ and $G = 0$ in *TE*, the diffusion coefficient being $d= 1/(R\,C)$.

```
> DE:=eval(TE,{L=0,G=0});
```

$$DE := \frac{\partial^2}{\partial x^2} V(x, t) = R\, C\, (\frac{\partial}{\partial t} V(x, t))$$

To answer the last part of the question, the given ansatz is entered.

```
> ansatz:=V(x,t)=exp(-k*t)*f(alpha*x+beta*t);
```

$$ansatz := V(x, t) = e^{(-k\, t)} f(\alpha\, x + \beta\, t)$$

The `pdetest` command is used to test whether the proposed solution satisfies the telegraph equation *TE*. If it does, the answer would be zero, but instead it yields an algebraic equation, appropriately in terms of differential operators, the notation introduced by Heaviside.

```
> eq6:=pdetest(ansatz,TE);
```

$$\begin{aligned} eq6 := -e^{(-k\,t)}(&-(\mathrm{D}^{(2)})(f)(\alpha\,x+\beta\,t)\,\alpha^2 - R\,C\,k\,f(\alpha\,x+\beta\,t)\\ &+ R\,C\,\mathrm{D}(f)(\alpha\,x+\beta\,t)\,\beta - L\,G\,k\,f(\alpha\,x+\beta\,t) + L\,G\,\mathrm{D}(f)(\alpha\,x+\beta\,t)\,\beta\\ &+ R\,G\,f(\alpha\,x+\beta\,t) + L\,C\,k^2\,f(\alpha\,x+\beta\,t) - 2\,L\,C\,k\,\mathrm{D}(f)(\alpha\,x+\beta\,t)\,\beta\\ &+ L\,C\,(\mathrm{D}^{(2)})(f)(\alpha\,x+\beta\,t)\,\beta^2) \end{aligned}$$

By inspection of the output in *eq6*, the second-order differential terms will cancel if we choose $\beta = \alpha/\sqrt{L\,C}$. Assuming that $G = R\,C/L$, all remaining terms will cancel provided that $k = R/L$. To confirm that this is so, these relations are entered and the new ansatz, *ansatz2* displayed, and entered as the argument in `pdetest`.

```
> beta:=alpha/sqrt(L*C): G:=R*C/L: k:=R/L: ansatz2:=ansatz;
```

$$ansatz2 := V(x,\,t) = e^{\left(-\frac{R\,t}{L}\right)}\,f(\alpha\,x + \frac{\alpha\,t}{\sqrt{L\,C}})$$

```
> pdetest(ansatz2,TE);
```

$$0$$

The answer is zero, indicating that a solution to the telegraph equation exists for $G{=}R\,C/L$ of the form $V{=}e^{-R\,t/L}f(\alpha(x+t/\sqrt{L\,C}))$. This solution implies "distortionless" propagation of the input wave form with velocity $1/\sqrt{L\,C}$, the wave form decreasing in amplitude with time.

4.2.2 Spiegel's Diffusion Problems

Any solution to a problem changes the problem.
R. W. Johnson, American journalist, *Washingtonian, Nov. 1979*

Murray Spiegel's *Advanced Mathematics* [Spi71] is probably the classic source book of solved problems relevant to mathematical physics. The following recipe is based on two diffusion examples (Problems **12.16** and **12.17**) taken from that text. Of course, we shall use computer algebra here, instead of performing a hand calculation. I shall take each example one step further by animating the solutions. Simply staring at the series solutions is no substitute for observing what actually happens as time progresses. So here are the problems.

(a) Using the method of separation of variables, solve the 1-dimensional heat diffusion equation in a thin bar 3 m in length whose surface is insulated and has a diffusion coefficient of 2 m^2/s. Its ends are kept at 0 °C ($T(0,t) = T(3,t) = 0$) and it has the initial temperature profile $T(x,0){=}5\,\sin(4\,\pi\,x) - 3\,\sin(8\,\pi\,x) + 2\,\sin(10\,\pi\,x)$. Animate the temperature profile over the time interval $t{=}0$ to 0.025 seconds.

(b) If the bar in part (a) has an initial internal constant temperature of 25 °C with the ends held at 0 °C, use the Fourier series method to derive the temperature distribution for arbitrary time $t > 0$. Animate the solution over the time interval $t{=}0$ to 2 seconds, keeping 20 terms in the series.

After loading the plot package, needed for the animations, the 1-dimensional heat diffusion equation for the bar of length $L=3$ and diffusion coefficient $d=2$ is entered in *pde*.

```
> restart: with(plots): d:=2: L:=3:
> pde:=diff(T(x,t),t)=d*diff(T(x,t),x,x);
```

$$pde := \frac{\partial}{\partial t}\, T(x,\, t) = 2\,(\frac{\partial^2}{\partial x^2}\, T(x,\, t))$$

The `pdsolve` command is applied to *pde*. Since the separation of variables method is requested, the hint that $T(x,t)=X(x)\,Y(t)$ is provided. The separated ODEs are integrated and the product solution built in *sol*.

```
> sol:=pdsolve(pde,HINT=X(x)*Y(t),INTEGRATE,build);
```

$$sol := T(x,\, t) = \_C3\,(e^{(\_c_1\, t)})^2\,\_C1\, e^{(\sqrt{\_c_1}\, x)} + \frac{\_C3\,(e^{(\_c_1\, t)})^2\,\_C2}{e^{(\sqrt{\_c_1}\, x)}}$$

Without loss of generality the integration constant $\_C3$ can be set equal to 1 on the rhs of *sol*. For later convenience, $\_C1$ and $\_C2$ are relabeled as A and B, respectively, and the separation constant $\_c_1=-k^2$.

```
> T:=subs({_C3=1,_C1=A,_C2=B,_c[1]=-k^2},rhs(sol));
```

$$T := (e^{(-k^2\, t)})^2\, A\, e^{(\sqrt{-k^2}\, x)} + \frac{(e^{(-k^2\, t)})^2\, B}{e^{(\sqrt{-k^2}\, x)}}$$

T is simplified with the symbolic option in *T2*.

```
> T2:=simplify(T,symbolic);
```

$$T2 := A\, e^{(-k\,(2\,k\,t - x\, I))} + B\, e^{(-k\,(2\,k\,t + x\, I))}$$

To satisfy the boundary condition $T(0,t) = 0$, we must set $B = -A$ in *T2*. The complex evaluation command is then applied so as to convert the complex exponential into a sine function.

```
> T3:=evalc(subs(B=-A,T2));
```

$$T3 := 2\, I\, A\, e^{(-2\,k^2\, t)}\, \sin(k\, x)$$

To satisfy the boundary condition $T(3,t)=0$, we must have $\sin(3\,k)=0$, so $k= m\,\pi/3$, with $m=1$, 2, 3, ... The coefficient is removed from *T3* by substituting $A=1/(2\,I)$.

```
> k:=m*Pi/L: T4:=subs(A=1/(2*I),T3);
```

$$T4 := e^{(-\frac{2\,m^2\,\pi^2\,t}{9})}\,\sin(\frac{m\,\pi\,x}{3})$$

The general solution is of the form $T(x,t)=\sum_{m=1}^{\infty} C_m\, e^{(-\frac{2\,m^2\,\pi^2\,t}{9})}\,\sin(\frac{m\,\pi\,x}{3})$. The initial temperature profile will be satisfied by retaining the three terms corresponding to $m = 12$, 24, and 30, with $C_{12} = 5$, $C_{24} = -3$, and $C_{30} = 2$. Using *T4*, this is done in *T5*.

```
> T5:=5*eval(T4,m=12)-3*eval(T4,m=24)+2*eval(T4,m=30);
```

$$T5 := 5\, e^{(-32\,\pi^2\,t)}\,\sin(4\,\pi\,x) - 3\, e^{(-128\,\pi^2\,t)}\,\sin(8\,\pi\,x) + 2\, e^{(-200\,\pi^2\,t)}\,\sin(10\,\pi\,x)$$

The temperature profile given in *T5* is animated over the time interval $t=0$ to 0.025 s, the opening frame of the animation (initial temperature profile) being shown in Figure 4.5. As t increases, $T(x,t)$ decays to zero inside the bar.

```
> animate(T5,x=0..L,t=0..0.025,frames=100,numpoints=500,
  thickness=2,labels=["x","T"]);
```



Figure 4.5: The initial temperature profile in the bar for part **(a)**.

Now let's tackle part **(b)** of the problem. The initial temperature profile is a constant 25 °C everywhere inside the bar. Using the Fourier series approach, the general Fourier coefficient is given by $C_m = (2/L)\int_0^L 25\,\sin(m\,\pi\,x/L)\,dx$ with $L=3$ and m a positive integer. This coefficient is now calculated.

```
> C[m]:=(2/L)*int(25*sin(m*Pi*x/L),x=0..L) assuming m::integer;
```

$$C_m := -\frac{50\,(-1+(-1)^m)}{m\,\pi}$$

The formal Fourier series representation (out to $m=20$) of the temperature is displayed in *Temp*, and then explicitly calculated with the `value` command.

```
> Temp:=Sum(C[m]*T4,m=1..20);
```

$$Temp := \sum_{m=1}^{20}\left(-\frac{50\,(-1+(-1)^m)\,e^{\left(-\frac{2\,m^2\,\pi^2\,t}{9}\right)}\sin\left(\frac{m\,\pi\,x}{3}\right)}{m\,\pi}\right)$$

```
> Temp:=value(Temp);
```

$$Temp := \frac{100\,e^{\left(-\frac{2\,\pi^2 t}{9}\right)}\sin\left(\frac{\pi x}{3}\right)}{\pi} + \frac{100}{3}\,\frac{e^{(-2\,\pi^2 t)}\sin(\pi x)}{\pi}$$

$$+\frac{20\,e^{\left(-\frac{50\,\pi^2 t}{9}\right)}\sin\left(\frac{5\pi x}{3}\right)}{\pi} + \frac{100}{7}\,\frac{e^{\left(-\frac{98\,\pi^2 t}{9}\right)}\sin\left(\frac{7\pi x}{3}\right)}{\pi} + \frac{100}{9}\,\frac{e^{(-18\,\pi^2\,t)}\sin(3\,\pi\,x)}{\pi}$$

$$+\frac{100}{11}\,\frac{e^{\left(-\frac{242\,\pi^2 t}{9}\right)}\sin\left(\frac{11\pi x}{3}\right)}{\pi} + \frac{100}{13}\,\frac{e^{\left(-\frac{338\,\pi^2 t}{9}\right)}\sin\left(\frac{13\pi x}{3}\right)}{\pi} + \frac{20}{3}\,\frac{e^{(-50\,\pi^2 t)}\sin(5\pi x)}{\pi}$$

$$+\frac{100}{17}\,\frac{e^{\left(-\frac{578\,\pi^2 t}{9}\right)}\sin\left(\frac{17\pi x}{3}\right)}{\pi}+\frac{100}{19}\,\frac{e^{\left(-\frac{722\,\pi^2 t}{9}\right)}\sin\left(\frac{19\pi x}{3}\right)}{\pi}$$

The temperature distribution is now animated over the time interval $t=0$ to 2 seconds, the initial frame in the animation being shown in Figure 4.6.

```
> animate(Temp,x=0..L,t=0..2,frames=50,thickness=2,
  numpoints=500,labels=["x","T"]);
```



Figure 4.6: The initial temperature profile for part **(b)** with 20 terms.

Because of the step function nature of the initial temperature profile at the ends of the bar, Gibb's oscillations appear in the initial series representation, oscillating around the exact profile. The oscillations quickly disappear as the animation progresses and once again $T(x,t)$ decreases to zero inside the bar.

4.2.3 Introducing Laplace's Equation

An editor is someone who separates the wheat from the chaff and then prints the chaff.
Adlai Stevenson, American politician, referring to news reporting (1900–65)

In electrostatics, the electric field $\vec{E}$ is related to the electric potential V by $\vec{E}=-\nabla V$. In SI units, Maxwell's equations are $\nabla\times\vec{E}=0$ and $\nabla\cdot\vec{E}=\rho/\epsilon_0$, where ρ is the electric charge density and ϵ_0 the permittivity of free space. Substituting $\vec{E}$ into the first relation yields the vector identity $\nabla\times\nabla V=0$, while the second relation produces Poisson's equation, $\nabla^2 V=-\rho/\epsilon_0$. In charge-free regions of space, $\rho=0$ and Poisson's equation reduces to Laplace's equation $\nabla^2 V=0$. This recipe involves solving a two-dimensional electrostatic boundary-value problem, using Laplace's equation in Cartesian coordinates.

Determine the potential $V(x, y)$ in the charge-free region $x \geq 0$, $0 \leq y \leq a$, given that the boundary at $x=0$ is held at the constant potential V_0, while the boundaries at $y=0$ and a are held at zero potential. The asymptotic boundary condition is $V \to 0$ for $x \to \infty$. Taking $a=1$ meter and $V_0=2$ volts, produce two- and three-dimensional contour plots of $V(x, y)$ for $x \geq 0$, $0 \leq y \leq a$.

After loading the plots library package, needed for the contour plots, the two-dimensional form of Laplace's equation is entered for the potential $V(x, y)$.

```
> restart: with(plots):
> pde:=diff(V(x,y),x,x)+diff(V(x,y),y,y)=0;
```

$$pde := (\frac{\partial^2}{\partial x^2} V(x, y)) + (\frac{\partial^2}{\partial y^2} V(x, y)) = 0$$

pde is analytically solved, assuming that $V(x, y) = X(x)\, Y(y)$.

```
> sol:=pdsolve(pde,HINT=X(x)*Y(y),INTEGRATE,build);
```

$$sol := V(x, y) = \_C3 \sin(\sqrt{\_c_1}\, y)\, \_C1\, e^{(\sqrt{\_c_1}\, x)} + \frac{\_C3 \sin(\sqrt{\_c_1}\, y)\, \_C2}{e^{(\sqrt{\_c_1}\, x)}} + \_C4 \cos(\sqrt{\_c_1}\, y)\, \_C1\, e^{(\sqrt{\_c_1}\, x)} + \frac{\_C4 \cos(\sqrt{\_c_1}\, y)\, \_C2}{e^{(\sqrt{\_c_1}\, x)}}$$

The substitution $\_c_1 = k$ is made on the rhs of *sol* and the result simplified.

```
> sol:=simplify(subs(sqrt(_c[1])=k,rhs(sol)));
```

$$sol := (\_C3 \sin(k\, y)\, \_C1\, e^{(2\, k\, x)} + \_C3 \sin(k\, y)\, \_C2 + \_C4 \cos(k\, y)\, \_C1\, e^{(2\, k\, x)} + \_C4 \cos(k\, y)\, \_C2\,)\, e^{(-k\, x)}$$

To satisfy the boundary condition $V=0$ at $y=0$, the $\cos(k\, y)$ term is removed from *sol* by setting it equal to zero. To satisfy $V = 0$ at $y = a$ for arbitrary $x > 0$, we must have $\sin(k\, a) = 0$, so $k = n\, \pi / a$, where n is a positive integer. The term $e^{2\, k\, x}$ must also be removed so that $V \to 0$ as $x \to \infty$.

```
> sol2:=subs({cos(k*y)=0,exp(2*k*x)=0,k=n*Pi/a},sol);
```

$$sol2 := \_C3 \sin(\frac{n\, \pi\, y}{a})\, \_C2\, e^{(-\frac{n\, \pi\, x}{a})}$$

Using the operand command, `op`, the coefficient combination is replaced with A. The arguments may have to be changed if the terms are ordered differently.

```
> sol3:=A*op(2,sol2)*op(4,sol2);
```

$$sol3 := A \sin(\frac{n\, \pi\, y}{a})\, e^{(-\frac{n\, \pi\, x}{a})}$$

Making use of orthogonality of the sine functions over the interval $y=0$ to a, the coefficient A is calculated, assuming that n is an integer.

```
> A:=(2/a)*int(V0*sin(n*Pi*y/a),y=0..a) assuming n::integer;
```

$$A := -\frac{2\, V0\, (-1 + (-1)^n)}{n\, \pi}$$

Entering the given values of the parameters, the result in *sol3* is summed. A large number of terms (100) is kept to obtain smooth contour plots.

```
> a:=1: V0:=2: V:=sum(sol3,n=1..100):
```

Using the `contourplot` command, V is plotted over the interval $x = 0$ to 2 and $y = 0$ to $a = 2$. Constant potential curves (equipotentials) are plotted for 0.2, 0.4,...,1.6, 1.8 volts. The grid is taken to be 50 × 50 in order to obtain reasonably smooth curves. The resulting picture is shown in Figure 4.7, the 0.2 volt curve being furthest to the right.

```
> contourplot(V,x=0..2,y=0..a,contours=[seq(i*V0/10,i=1..9)],
  thickness=2,grid=[50,50]);
```



Figure 4.7: Equipotentials corresponding to 0.2 (far right), 0.4,...,1.8 volts.

The `contourplot3d` command is used to produce a three-dimensional contour plot, the equipotential curves again being spaced 0.2 volts apart.

```
> contourplot3d(V,x=0..1,y=0..a,contours=[seq(i*V0/10,i=1..9)],
  grid=[60,60],shading=zhue,filled=true,axes=boxed,
  tickmarks=[2,2,2]);
```

The resulting picture can be viewed by executing the recipe on your computer.

4.2.4 Grandpa's "Trampoline"

In the first place God made idiots. This was for practice. Then He made School Boards.
Mark Twain, *Following the Equator, (1897)*

From his sundeck, my grandson Daniel can peek over the fence and see his neighbor's children happily bouncing up and down on their trampoline after coming home from school. He is too young to join in the fun, so must be content (not very!) to watch. On the other hand, I am getting too old to engage in trampoline antics, being more content to substitute the following recipe instead.

A trampoline, consisting of a square horizontal elastic membrane fixed on its four edges ($x\!=\!0$, b and $y\!=\!0$, b), has an initial transverse profile $\psi(x,y,0) \equiv f = A\,x^2\,y\,(b-x)\,(b-y)^3$ and is released from rest. Determine $\psi(x,y,t)$ for arbitrary time $t > 0$. Taking $A\!=\!1/5$ m^{-7}, $b\!=\!2$ m, and the wave speed $c\!=\!1$ m/s, animate the trampoline motion.

The plots package is loaded, because we shall be animating the motion of the trampoline. A functional operator F is created to calculate the second derivative of $\psi(x,y,t)$ with respect to an arbitrary variable v.

```
> restart: with(plots): F:=v->diff(psi(x,y,t),v,v):
```

The transverse vibrations of the trampoline are governed by the two-dimensional wave equation which is entered using F.

```
> pde:=F(x)+F(y)=F(t)/c^2;
```

$$pde := (\frac{\partial^2}{\partial x^2}\,\psi(x,\,y,\,t)) + (\frac{\partial^2}{\partial y^2}\,\psi(x,\,y,\,t)) = \frac{\frac{\partial^2}{\partial t^2}\,\psi(x,\,y,\,t)}{c^2}$$

The wave equation is solved by the method of separation of variables, the assumed form $\psi(x,y,t)\!=\!X(x)\,Y(y)\,T(t)$ being supplied as a hint.

```
> sol:=pdsolve(pde,HINT=X(x)*Y(y)*T(t),INTEGRATE,build);
```

The output (not displayed here) has two separation constants, $\_c_1$ and $\_c_2$. To simplify the form of the solution, the substitution $\_c_1 = -\alpha^2$ and $\_c_2 = -\beta^2$ is made on the rhs of *sol* and the result simplified with the symbolic option.

```
> psi1:=simplify(subs({_c[1]=-alpha^2,_c[2]=-beta^2},
          rhs(sol)),symbolic);
```

$$\psi 1 := \_C5\,\sin(c\,\sqrt{\alpha^2+\beta^2}\,t)\,\_C3\,\_C1\,e^{((\beta\,y+\alpha\,x)\,I)} + \cdots$$
$$+\,\_C6\,\cos(c\,\sqrt{\alpha^2+\beta^2}\,t)\,\_C3\,\_C1\,e^{((\beta\,y+\alpha\,x)\,I)} + \cdots$$

The initial transverse velocity is zero, so the time-dependent sine terms must be removed from $\psi 1$.

```
> psi2:=remove(has,psi1,sin);
```

The exponential terms in $\psi 2$ are converted to a trig form in $\psi 3$ and the result expanded. The terms $\cos(\beta\,y)$, $\sin(\beta\,y)$, $\cos(\alpha\,x)$, and $\sin(\alpha\,x)$ appear.

```
> psi3:=expand(convert(psi2,trig));
```

To satisfy the fixed edge conditions along $y\!=\!0$ and $x\!=\!0$, the cosine terms are removed from the solution by setting them equal to zero, while one must have $\alpha\!=\!m\,\pi/b$ and $\beta\!=\!n\,\pi/b$, with $m,n\!=\!1$, 2, 3, ..., to satisfy the conditions along $x\!=\!b$ and $y\!=\!b$. Making these substitutions and factoring yields $\psi 4$.

```
> psi4:=factor(subs({cos(beta*y)=0,cos(alpha*x)=0,
          alpha=m*Pi/b,beta=n*Pi/b},psi3));
```

$$\psi 4 := -\cos(c\,\pi\sqrt{\frac{m^2+n^2}{b^2}}\,t)\,\sin(\frac{n\,\pi\,y}{b})\,\sin(\frac{m\,\pi\,x}{b})\,\_C6\,(\_C1 - \_C2)(\_C3 - \_C4)$$

The following select command is used to replace the cumbersome coefficient combination in $\psi 4$ with the symbol $B_{m,n}$.

```
> psi5:=B[m,n]*select(has,psi4,{sin,cos});
```

$$\psi 5 := B_{m,n} \cos(c\,\pi \sqrt{\frac{m^2+n^2}{b^2}}\, t) \sin(\frac{n\,\pi\,y}{b}) \sin(\frac{m\,\pi\,x}{b})$$

The initial transverse profile of the trampoline is entered.

```
> f:=A*x^2*y*(b-x)*(b-y)^3;
```

$$f := A\,x^2\,y\,(b-x)\,(b-y)^3$$

Using orthogonality of the sine functions, the coefficient $B_{m,n}$ is evaluated by performing the double integration

$$B_{m,n} = (2/b)^2 \int_0^b \int_0^b f\ \sin(m\,\pi\,x/b)\ \sin(n\,\pi\,y/b)\,dx\,dy,$$

and assuming that m and n are integers.

```
> B[m,n]:=(2/b)^2*int(int(f*sin(m*Pi*x/b)*sin(n*Pi*y/b),
          x=0..b),y=0..b) assuming m::integer,n::integer;
```

$$B_{m,\,n} := -48\,b^7\,A(-4+8\,(-1)^{(1+m)}+n^2\,\pi^2+2\,(-1)^m\,\pi^2\,n^2+4\,(-1)^n$$
$$+8\,(-1)^{(n+m)})\Big/(m^3\,\pi^8\,n^5)$$

The solution then will be of the form

$$\psi(x,y,t) = \sum_{m=1}^{\infty}\sum_{n=1}^{\infty} B_{m,n}\ \sin(m\,\pi\,x/b)\ \sin(n\,\pi\,y/b)\ \cos(c\,\pi\,\sqrt{m^2+n^2}\,t/b).$$

A functional operator `G` is formed, using $\psi 5$, for explicitly calculating this double Fourier series out to N terms in each sum,

```
> G:=N->sum(sum(psi5,m=1..N),n=1..N):
```

The given parameters are entered, and the double series calculated for $N = 10$. You will have to execute the recipe on your own computer to see the $10 \times 10 = 100$ terms in ψ.

```
> A:=1/5: b:=2: c:=1: N:=10: psi:=G(N);
```

Using the `animate` command, the answer ψ is animated over the time interval $t = 0$ to 5 seconds, 50 frames being used. Execute the recipe and enjoy!

```
> animate(plot3d,[psi,x=0..b,y=0..b],t=0..5,frames=50,
  axes=boxed,shading=zhue,tickmarks=[3,3,3]);
```

4.2.5 Irma Insect's Isotherm

A man thinks he amounts to a great deal but to a flea or a mosquito a human being is merely something good to eat.
Don Marquis, American humorist, (1878–1937)

In the Erehwon zoo, Irma insect lives inside a rectangular enclosure with walls at $x = 0$, a and $y = 0$, b, floor at $z = 0$ and ceiling at $z = c$. In the winter, the floor and walls have a temperature of zero degrees while the ceiling has a steady temperature profile $T(x, y, c) = 1600\, x\, (a - x)\, y\, (b - y)$ degrees. Determine the temperature inside Irma's enclosure. Irma is most comfortable when the temperature is $20\,°$. If $a = b = c = 1$, plot the 3-dimensional isothermal surface corresponding to this temperature. What is T at the center of the enclosure?

The steady-state temperature inside the enclosure is given by the solution of Laplace's equation, $\nabla^2 T(x, y, z) = 0$. For variety, let's load the VectorCalculus package and use the `Laplacian` command in Cartesian coordinates.

```
> restart: with(plots): with(VectorCalculus):
> pde:=Laplacian(T(x,y,z),'cartesian'[x,y,z])=0;
```

$$pde := \left(\frac{\partial^2}{\partial x^2}\, T(x, y, z)\right) + \left(\frac{\partial^2}{\partial y^2}\, T(x, y, z)\right) + \left(\frac{\partial^2}{\partial z^2}\, T(x, y, z)\right) = 0$$

The temperature of the walls at $x = 0$ and a is held at zero degrees. On separating variables, the x part of the solution will be a linear combination of a sine and a cosine. To satisfy $T = 0$ at $x = 0$ and a, this part of the solution must involve $\sin(m\,\pi\, x/a)$, with m a positive integer. Similarly, the y part of the solution must be $\sin(n\,\pi\, y/b)$, with n a positive integer. So, let's apply the `pdsolve` command with the hint that the solution is of the form $\sin(m\,\pi\, x/a)\, \sin(n\,\pi\, y/b)\, Z(z)$, with the structure of $Z(z)$ to be determined.

```
> sol:=pdsolve(pde,T(x,y,z),HINT=sin(m*Pi*x/a)*
      sin(n*Pi*y/b)*Z(z),INTEGRATE,build);
```

$$sol := T(x, y, z) = \sin\left(\frac{m\,\pi\, x}{a}\right) \sin\left(\frac{n\,\pi\, y}{b}\right) \_C1 \sin\left(\frac{\pi\sqrt{-m^2\, b^2 - n^2\, a^2}\, z}{a\, b}\right) + \sin\left(\frac{m\,\pi\, x}{a}\right) \sin\left(\frac{n\,\pi\, y}{b}\right) \_C2 \cos\left(\frac{\pi\sqrt{-m^2\, b^2 - n^2\, a^2}\, z}{a\, b}\right)$$

Since the floor at $z = 0$ is held at zero degrees, the cosine term must be removed from the rhs of the solution *sol*. The result is the general term $T_{m,\,n}$ in a double Fourier series, with the coefficient $\_C1$ to be determined.

```
> T[m,n]:=remove(has,rhs(sol),cos);
```

$$T_{m,\,n} := \sin\left(\frac{m\,\pi\, x}{a}\right) \sin\left(\frac{n\,\pi\, y}{b}\right) \_C1 \sin\left(\frac{\pi\sqrt{-m^2\, b^2 - n^2\, a^2}\, z}{a\, b}\right)$$

The temperature profile at $z = c$ is entered in f. To determine the form of $\_C1$, the combination $\sin(m\,\pi\, x/a)\, \sin(n\,\pi\, y/b)$ is first entered in g.

```
> f:=1600*x*(a-x)*y*(b-y): g:=sin(m*Pi*x/a)*sin(n*Pi*y/b):
```

Making use of orthogonality of the sine functions, f is multiplied by g and the double integral over x and y carried out. This must be equal to $T_{m,\,n}$, evaluated at $z = c$, multiplied by g, with the same double integration performed.

```
> eq:=int(int(f*g,x=0..a),y=0..b)
    =int(int(eval(T[m,n],z=c)*g,x=0..a),y=0..b):
```

Then eq is solved for $\_C1$, and simplified assuming that m and n are integers.

```
> _C1:=simplify(solve(eq,_C1)) assuming m::integer,n::integer;
```

$$\_C1 := \frac{25600\,a^2\,b^2\,(1-(-1)^m+(-1)^m\,(-1)^n-(-1)^n)}{\sin(\frac{\pi\,\sqrt{-m^2\,b^2-n^2\,a^2}\,c}{a\,b})\,m^3\,\pi^6\,n^3}$$

The dimensions of the enclosure are entered, and the series representation of the temperature T calculated, keeping $50 \times 50 = 2500$ terms to ensure accurate numerical results.

```
> a:=1.0: b:=1.0: c:=1.0: T:=sum(sum(T[m,n],m=1..50),n=1..50):
```

Setting $T = 20$, the `implicitplot3d` command is used to plot the 20 degree isotherm, the result being the bowl-shaped surface shown in Figure 4.8.

```
> implicitplot3d(T=20,x=0..a,y=0..b,z=0..c,axes=box,
  orientation=[10,70],style=PATCHCONTOUR,tickmarks=[2,2,2]);
```



Figure 4.8: The 20 °C isotherm.

The temperature is now evaluated at the center of the enclosure,

```
> T2:=evalf(eval(T,{x=a/2,y=b/2,z=c/2}));
```

$$T2 := 11.36308962$$

and found to be about 11 degrees.

4.2.6 Daniel Hits Middle C

The notes I handle no better than many pianists. But the pauses between the notes – ah, that is where the art resides.
Artur Schnabel, American pianist, *Chicago Daily News, 11 June 1958*

A horizontal bar of nickel-iron (Young's modulus, $Y = 2.1 \times 10^{11}$ N/m^2, and density, $\rho = 7.8 \times 10^3$ kg/m^3) of length $L = 1$ m, whose cross-section is rectangular with width $W = 0.1$ m and height $H = 0.049$ m, is clamped (i.e., $\psi = \psi' = 0$) at $x = 0$ and L. Daniel strikes the bar sharply at the midpoint of one of its wider sides in such a way that the initial transverse velocity is (approximately) $v(x,0) = v_0\,\delta(x - L/2)$, with $v_0 = 1$ m^2/s and δ the Dirac delta function. Determine the transverse displacement $\psi(x,t)$ of the bar at arbitrary time $t > 0$. Show that the fundamental frequency corresponds to approximately middle C (f=261.63 Hz). Animate the motion of the bar over the interval $t=0$ to $10\,T$, where $T = 1/f$ is the fundamental period.

After loading the plots library package, needed for the animation,

```
> restart: with(plots):
```

the equation of motion for transverse oscillations of the bar is entered.

```
> pde:=a^4*diff(psi(x,t),x$4)+diff(psi(x,t),t,t)=0;
```

$$pde := a^4\,(\frac{\partial^4}{\partial x^4}\,\psi(x,\,t)) + (\frac{\partial^2}{\partial t^2}\,\psi(x,\,t)) = 0$$

The parameter $a = (S\,\kappa^2\,Y/\epsilon)^{1/4}$, where $S = W H$ is the cross-sectional area of the bar, $\kappa = H/\sqrt{12}$ is the radius of gyration about the horizontal midplane through the bar, and $\epsilon = \rho\,S$ is the linear density.

Using pdsolve, *pde* is solved with the hint $\psi(x,t) = X(x)\,T(t)$.

```
> sol:=pdsolve(pde,HINT=X(x)*T(t),INTEGRATE,build);
```

Setting the separation constant $\_c_1 = k^4$ in *sol*, the frequency is $\omega = a^2\,k^2$.

```
> sol:=simplify(subs(_c[1]=k^4,sol),symbolic);
```

$$sol := \psi(x,\,t) = e^{(-I\,k\,x)}\,\_C5\,\sin(a^2\,k^2\,t)\,\_C1 + \cdots + \_C6\,\cos(a^2\,k^2\,t)\,\_C4\,e^{(k\,x)}$$

Since the bar is initially horizontal, i.e., $\psi(x,0) = 0$, the cosine term must be removed from the rhs of *sol*. The result is then factored.

```
> psi:=factor(remove(has,rhs(sol),cos));
```

$$\psi := \_C5\,\sin(a^2\,k^2\,t)(\_C1\,e^{(-I\,k\,x)} + e^{(-k\,x)}\,\_C2 + \_C3\,e^{(k\,x\,I)} + e^{(k\,x)}\,\_C4)$$

To apply the boundary conditions at the end of the bar, the spatial part is extracted with the select command. The time part is also selected.

```
> X:=select(has,psi,x); T:=select(has,psi,t);
```

$$X := \_C1\,e^{(-I\,k\,x)} + e^{(-k\,x)}\,\_C2 + \_C3\,e^{(k\,x\,I)} + e^{(k\,x)}\,\_C4$$

$$T := \sin(a^2\,k^2\,t)$$

X is converted to trig form, and the cosine, sine, cosh, and sinh terms collected.

```
> X:=collect(convert(X,trig),[cos,sin,cosh,sinh]);
```

$$X := (\_C3 + \_C1)\cos(k\,x) + (-\_C1\,I + \_C3\,I)\sin(k\,x) \\ + (\_C4 + \_C2)\cosh(k\,x) + (\_C4 - \_C2)\sinh(k\,x)$$

Using the op command to extract the relevant operands, new coefficients *A1*, etc, are introduced into X in the following line.

```
> X:=add(A||i*op([i,2],X),i=1..4);
```

$$X := A1\,\cos(k\,x) + A2\,\sin(k\,x) + A3\,\cosh(k\,x) + A4\,\sinh(k\,x)$$

The clamped-end boundary conditions are applied at $x=0$ in *bc1* and *bc2* and at $x=L$ in *bc3* and *bc4*.

```
> bc1:=eval(X,x=0)=0; bc2:=eval(diff(X,x),x=0)=0;
```

$$bc1 := A1 + A3 = 0 \qquad bc2 := A2\,k + A4\,k = 0$$

```
> bc3:=eval(X,x=L)=0; bc4:=eval(diff(X,x),x=L)=0;
```

$$bc3 := A1\cos(k\,L) + A2\sin(k\,L) + A3\cosh(k\,L) + A4\sinh(k\,L) = 0$$

$$bc4 := -A1\sin(k\,L)\,k + A2\cos(k\,L)\,k + A3\sinh(k\,L)\,k + A4\cosh(k\,L)\,k = 0$$

On attempting to solve the four boundary conditions for the four unknown coefficients, one of the coefficients will be undetermined and a transcendental equation for k will result. Let's choose the undetermined coefficient to be $A4$. Then, $bc1$, $bc2$, and $bc3$ are solved for $A1$, $A2$, and $A3$ and $sol2$ assigned.

```
> sol2:=solve({bc1,bc2,bc3},{A||1,A||2,A||3}); assign(sol2):
```

The coefficient $A4$ is temporarily set equal to 1 and the fourth boundary condition divided by k^3 and simplified with the symbolic option.

```
> A||4:=1: bc4:=simplify(bc4/k^3,symbolic);
```

$$bc4 := \frac{2\,(-1 + \cos(k\,L)\cosh(k\,L))}{k^2\,(\cos(k\,L) - \cosh(k\,L))} = 0$$

We set $k\,L = K$ in $bc4$ and isolate the $\cos(K)$ term to the left of the equation.

```
> eq:=isolate(subs(k*L=K,bc4),cos(K));
```

$$eq := \cos(K) = \frac{1}{\cosh(K)}$$

On comparing the transcendental equation eq with the corresponding equation for the clamped-free end situation in Recipe **01-1-3**, we see that they differ by a minus sign on the right-hand side. As before, eq must be solved numerically for the allowed K values. For large K, $\cosh(K)$ becomes very large and $\cos K$ approaches zero. The allowed K values, therefore, are approximately the zeros of the cosine function. A functional operator f is introduced to determine the zeros in the range $3(n-1)$ to $3\,n$, where n will take on the values 1, 2, etc.

```
> f:=n->fsolve(eq,K,3*(n-1)..3*n):
```

Then forming `f(n)`, dividing by L, and using the sequence command, the first four zeros of k are given in $sol3$ which is assigned.

```
> sol3:=seq(k||n=f(n)/L,n=1..4); assign(sol3):
```

$$sol3 := k1 = 0.,\ k2 = \frac{4.730040745}{L},\ k3 = \frac{7.853204624}{L},\ k4 = \frac{10.99560784}{L}$$

The first zero, $k1$, must be rejected, because for $k = 0$ the lhs of $bc4$ is finite. The parameters L, v_0, W, H, Y, and ρ are entered, and the radius of gyration κ, cross-sectional area S, linear density ϵ, and the parameter a calculated.

```
> L:=1: v0:=1: W:=0.1: H:=0.049: Y:=2.1*10^11: rho:=7.8*10^3:
> kappa:=evalf(H/sqrt(12)); S:=W*H; epsilon:=rho*S;
  a:=(S*kappa^2*Y/epsilon)^(1/4);
```

$$\kappa := 0.01414508160 \quad S := 0.0049 \quad \varepsilon := 38.22000 \quad a := 8.567101289$$

The coefficient $A4$, which had been temporarily set equal to 1, is relabeled

B, and will now be calculated. The initial transverse velocity is of the form $v(x,0)=v_0\,\delta(x-L/2)=\sum_{n=1}^{\infty} a^2\,k_n^2\,B_n\,X_n$, where X_n is the nth spatial mode. Multiplying $v(x,0)$ by X_n, integrating x over the range 0 to L, and using the orthogonality of the spatial modes, yields $v_0\,X_n(x=L/2)=B_n\,a^2\,k_n^2\,\int_0^L X_n^2\,dx$, which is easily solved for B_n. The coefficient B is evaluated in the following line for a general k value.

```
> B:=v0*eval(X,x=L/2)/(a^2*k^2*int(X^2,x=0..L)):
```

The product $B\,X\,T$ is evaluated at $k=kn$ and the first three terms of the complete solution ψ added. Only the fundamental ($n=2$) contribution is shown here, the remaining terms having rapidly decreasing amplitudes.

```
> psi:=simplify(add(eval(B*X*T,k=k||n),n=2..4));
```

$$\begin{aligned}\psi := &-0.0009671479296\sin(1642.092308\,t)\cos(4.730040745\,x)\\ &+0.0009502249823\sin(1642.092308\,t)\sin(4.730040745\,x)\\ &+0.0009671479296\sin(1642.092308\,t)\cosh(4.730040745\,x)\\ &-0.0009502249823\sin(1642.092308\,t)\sinh(4.730040745\,x)+\cdots\end{aligned}$$

The fundamental frequency $\omega=a^2\,k2^2$ is 1642 rads/s or $f=\omega/(2\,\pi)\approx 261.3$ Hz, which is close to middle C. The fundamental period $T=1/f$ is also calculated.

```
> omega:=a^2*k||2^2; f:=evalf(omega/(2*Pi)); T:=1/f;
```

$$\omega := 1642.092308 \quad f := 261.3471078 \quad T := 0.003818536004$$

Finally, the vibrations of the bar are animated over the interval $t=0$ to $10\,T$.

```
> animate(psi,x=0..L,t=0..10*T,frames=100,thickness=2);
```

Execute the recipe on your computer and click on the plot and on the start arrow to see the vibrations.

4.2.7 A Poisson Recipe

Life is good for only two things, discovering mathematics and teaching mathematics

Simon Poisson, French mathematician, (1781–1840)

In this day and age, being overweight and having high blood pressure and/or high cholesterol cause great concern among an aging population. To address these issues, all types of diets are proposed and cookbooks written. Amongst the latter are the *HeartSmart* [Ste94] cook books, sponsored by the Heart and Stroke Foundation of Canada. Each book presents "over 200 healthful & delicious recipes", with a section devoted to fish and seafood. So, in the spirit of keeping us intellectually healthy, here's a delicious Poisson (equation) recipe.

The simplest non-trivial Poisson equation problems involve point or line charges in the vicinity of one or more "grounded" (zero potential) conducting surfaces. Since the charge densities for these sources may be characterized by Dirac δ-functions, the solutions of these problems are electrostatic Green's functions. As a simple 2-dimensional example, let us consider an infinitely long line charge characterized by the charge density $\rho = 4\,\pi\,\epsilon_0\,\delta(x-a)\,\delta(y-b)$, with ϵ_0 the permittivity of free space. The line charge is located between two infinite grounded conducting plates at $y=0$ and at $L > b > 0$. To firmly establish the geometry in our minds, a finite portion of the two infinite conducting planes is plotted in the x-y plane as green lines in gr1, with $L=1$.

```
> restart: with(plots):
> gr1:=plot([[[-1/2,0],[3/2,0]],[[-1/2,1],[3/2,1]]],color=
      green,thickness=2,tickmarks=[4,3],labels=["x","y"]):
```

Taking $a = 0.5$ and $b = 0.8$, a point is plotted in gr2, representing an end-on view of the line charge. The geometry for the Green's function problem is then displayed by superimposing the two graphs in Figure 4.9.

```
> gr2:=pointplot([[0.5,0.8]],symbol=circle,symbolsize=12):
> display({gr1,gr2},scaling=constrained);
```



Figure 4.9: End-on view of line charge between two grounded conducting plates.

The relevant Poisson equation for the Green's function potential G is

$$\nabla^2 G = \frac{\partial^2 G}{\partial x^2} + \frac{\partial^2 G}{\partial y^2} = -\,\frac{\rho}{\epsilon_0} = -\,4\,\pi\,\delta(x-a)\,\delta(y-b). \tag{4.7}$$

For the region outside the line charge, Equation (4.7) becomes homogeneous and a solution is easily constructed. In the y direction G is zero at $y=0$ and L, so G must include terms of the structure $\sin(n\,\pi\,y/L)$, with $n=1, 2, 3, \ldots$. As $x \to \pm\infty$, we must have $G \to 0$. For $x < a$, the Green's function GL to the "left" of the line charge will be built up of terms of the structure $e^{n\,\pi\,(x-a)/L}$. We have used the fact that our final result should ultimately depend only on the difference $x-a$. The factor $n\,\pi/L$ is included so that the homogeneous form of Poisson's equation will be satisfied. The nth term in the infinite series then takes the following form for GL, the coefficients A_n still to be determined.

```
> GL:=A[n]*sin(n*Pi*y/L)*exp(n*Pi*(x-a)/L);
```

$$GL := A_n \sin(\frac{n\,\pi\,y}{L})\, e^{(\frac{n\,\pi\,(x-a)}{L})}$$

For $x > a$, the Green's function GR to the "right"of the line charge is built up of terms of the structure $e^{-n\,\pi\,(x-a)/L}$. The nth term in the infinite series then takes the following form for GR, the coefficients B_n still to be determined.

```
> GR:=B[n]*sin(n*Pi*y/L)*exp(-n*Pi*(x-a)/L);
```

$$GR := B_n \sin(\frac{n\,\pi\,y}{L})\,e^{(-\frac{n\,\pi\,(x-a)}{L})}$$

As a check on, e.g., GL, we confirm that $\partial^2(GL)/\partial x^2 + \partial^2(GL)/\partial y^2 = 0$.

```
>   check:=diff(GL,x,x)+diff(GL,y,y);
```

$$check := 0$$

The Green's function must be continuous, i.e., $GL = GR$, at $x = a$ for arbitrary y between 0 and L. The continuity condition is implemented in *eq1*.

```
>   eq1:=expand(eval(GL=GR,x=a)/sin(n*Pi*y/L));
```

$$eq1 := A_n = B_n$$

To determine the discontinuity in $\partial G/\partial x$ at $x = a$, we appeal to the divergence (Gauss's) theorem, viz., $\oint_V (\nabla \cdot \vec{A})\,dv = \oint_S (\hat{n} \cdot \vec{A})\,ds$ for a vector field $\vec{A}$. Here $\hat{n}$ is the outward unit normal to the closed surface S, enclosing a volume V. Let's take V to be a thin (thickness 2ϵ, where ultimately $\epsilon \to 0$) slice between the plates of unit length in the z direction, with faces at $x - \epsilon$ and $x + \epsilon$, and $\vec{A} = \nabla G$. Then, making use of Poisson's equation, the divergence theorem yields

$$\int\!\!\int \nabla^2 G\,dx\,dy = \int (\hat{n} \cdot \nabla G)\,d\ell = \int \frac{\partial G}{\partial n}\,d\ell = -4\pi \int\!\!\int \delta(x-a)\,\delta(y-b)\,dx\,dy = -4\,\pi.$$

Here $\partial G/\partial n$ is the normal derivative of G and the line integral ($\int d\ell$) is carried out along the perimeter of the slice in the x-y plane. As $\epsilon \to 0$, the "ends" of the slice at $y = 0$ and L will make no contribution to the line integral, so that $\lim_{\epsilon \to 0}[\int_{\text{along } x=a+\epsilon} (\partial G/\partial x)\,dy - \int_{\text{along } x=a-\epsilon} (\partial G/\partial x)\,dy] = -4\pi$, which implies (in the limit $\epsilon \to 0$) that $(\partial G/\partial x)_{a+\epsilon} - (\partial G/\partial x)_{a-\epsilon} = -4\pi\delta(y-b)$. Multiplying this result by $\sin(n\pi y/L)$, integrating from $y = 0$ to L, and using orthogonality, yields the desired second relation for the coefficients. This calculation is implemented in *eq2*, the assumptions that $b > 0$, $L > b$, and n is an integer being included to accomplish the integration over the δ-function.

```
>   eq2:=int(eval(diff(GL-GR,x),x=a)*sin(n*Pi*y/L),y=0..L)
        =4*Pi*int(Dirac(y-b)*sin(n*Pi*y/L),y=0..L)
         assuming b>0,L>b,n::integer;
```

$$eq2 := \frac{1}{2}\,n\,\pi\,A_n + \frac{1}{2}\,n\,\pi\,B_n = 4\,\pi \sin(\frac{n\,\pi\,b}{L})$$

eq1 and *eq2* are solved for A_n and B_n, and the solution assigned.

```
>   sol:=solve({eq1,eq2},{A[n],B[n]}): assign(sol):
```

Keeping 30 terms in the sum, the complete Green's function to the left and right of $x = a$ are given in *GL2* and *GR2*, respectively.

```
>   GL2:=Sum(GL,n=1..30); GR2:=Sum(GR,n=1..30);
```

$$GL2 := \sum_{n=1}^{30} \left(\frac{4 \sin(\frac{n\,\pi\,b}{L}) \sin(\frac{n\,\pi\,y}{L})\,e^{(\frac{n\,\pi\,(x-a)}{L})}}{n} \right)$$

$$GR2 := \sum_{n=1}^{30} \left(\frac{4 \sin(\frac{n\,\pi\,b}{L}) \sin(\frac{n\,\pi\,y}{L})\, e^{(-\frac{n\,\pi\,(x-a)}{L})}}{n} \right)$$

The parameter values $a = 0.5$, $b = 0.8$, and $L = 1$ are entered, and the piecewise Green's function, $G = GL2$ for $x < a$ and $GR2$ for $x > a$, formed.

```
> a:=0.5: b:=0.8: L:=1:
> G:=piecewise(x<a,value(GL2),x>a,value(GR2)):
```

Using `contourplot`, equipotentials are drawn in `cp` for $G = 1/8$, $1/4$, $1/2$, 1, 2, 3 and then superimposed on `gr1` and `gr2`, the result being shown in Fig. 4.10. The curve closest to the line charge is for $G = 3$, the furthest for $G = 1/8$.

```
> cp:=contourplot(G,x=-2..2,y=0..L,contours=[1/8,1/4,1/2,
      1,2,3],grid=[70,70],color=blue,thickness=2):
> display({gr1,gr2,cp},scaling=constrained);
```



Figure 4.10: Equipotentials around line charge between grounded plates.

4.3 Beyond Cartesian Coordinates

Some representative non-Cartesian examples are now presented. Many more are included in the **Supplementary Recipes** at the end of the chapter.

4.3.1 Is It Separable?

Alliance. In international politics, the union of two thieves who have their hands so deeply inserted in each other's pockets that they cannot separately plunder a third.
Ambrose Bierce, American author, *The Devil's Dictionary (1881–1906)*

The method of separation of variables can be applied to other orthogonal curvilinear coordinate systems besides the Cartesian system. I have found that, at first, some students are surprised that the variable separation method works at all. After a while, they assume that it always works. It turns out that the *scalar Helmholtz equation*, $\nabla^2 S + k^2\, S = 0$, which is the spatial part of either the wave or diffusion equations with k a constant, is separable [MF53] in 11, and only 11, 3-dimensional orthogonal curvilinear coordinate systems. Fortunately, these include spherical polar and cylindrical coordinates, which are the two most commonly used non-Cartesian systems. An example of a 3-dimensional coordinate system for which the Helmholtz equation is not separable are the *bispherical* coordinates u, v, w, which are related to x, y, z by

$$x = \frac{a\,\sin u\,\cos v}{\cosh w - \cos u}, \quad y = \frac{a\,\sin u\,\sin v}{\cosh w - \cos u}, \quad z = \frac{a\,\sinh w}{\cosh w - \cos u}. \tag{4.8}$$

Here a is a scale factor and $0 \le u < \pi$, $0 \le v \le 2\,\pi$, $-\infty < w < \infty$.

As the following recipe illustrates, Laplace's equation is not separable in bispherical coordinates either, but can be separated into three ODEs by a modified separation assumption. This is useful, e.g., in determining the potential outside two spheres of equal diameters, held at different potentials, and with their centers separated by a distance greater than the sphere diameter.[2]

Although, the bispherical system is known (with $a = 1$) to Maple, it is instructive to tackle the following problem from first principles.

- **(a)** Plot the contours in the x-z plane corresponding to holding u and w fixed. What surfaces are generated if v is constant?
- **(b)** Calculate the scale factors and the Laplacian operator.
- **(c)** Show that Laplace's equation is not completely separable if one makes the "standard" ansatz, $S(u,v,w) = U(u)\,V(v)\,W(w)$.
- **(d)** Show that Laplace's equation is completely separable if one assumes that $S(u,v,w) = \sqrt{(\cosh w - \cos u)}\,U(u)\,V(v)\,W(w)$. Assuming that $\cosh w > \cos u$, identify any special functions which occur in the separated ODEs.

It is assumed that $u \ge 0$, $u < \pi$, $v \ge 0$, $v \le 2\pi$, and $\cosh w > \cos u$. The coordinate relations are then entered.

```
> restart: with(plots): assume(u>=0,u<Pi,v>=0,v<=2*Pi,
  cosh(w)>cos(u)):
> x:=a*sin(u)*cos(v)/(cosh(w)-cos(u));
> y:=a*sin(u)*sin(v)/(cosh(w)-cos(u));
> z:=a*sinh(w)/(cosh(w)-cos(u));
```

To plot the surfaces corresponding to holding w fixed, let's form $X^2 + Y^2 + (Z - a\,\coth w)^2 = x^2 + y^2 + (z - a\,\coth w)^2$ and simplify the right-hand side.

```
> eq1:=X^2+Y^2+(Z-a*coth(w))^2
      =simplify(x^2+y^2+(z-a*coth(w))^2);
```

$$eq1 := X^2 + Y^2 + (Z - a\coth(w))^2 = \frac{a^2}{\sinh(w)^2}$$

The result is the equation of a sphere of radius $a/\sinh w$ centered at $X = 0$,

[2] An excellent source of electrostatic problems in bispherical and other coordinate systems is *Problems in Mathematical Physics* by Lebedev, Skal'skaya, and Uflyand (Pergamon, 1966).

$Y = 0$, and $Z = a \coth w$. Different choices of w will generate different size spheres with centers located at different Z values.

Similarly, $(\sqrt{X^2+Y^2} - a\,\cot u)^2 + Z^2 = (\sqrt{x^2+y^2} - a\,\cot u)^2 + z^2$ is entered in *eq2* and simplified.

```
>  eq2:=(sqrt(X^2+Y^2)-a*cot(u))^2+Z^2
       =simplify((sqrt(x^2+y^2)-a*cot(u))^2+z^2,symbolic);
```

$$eq2 := (\sqrt{X^2+Y^2} - a\cot(u))^2 + Z^2 = \frac{a^2}{\sin(u)^2}$$

To see what type of surface is generated by holding u fixed, let's set $a=1$ and $Y=0$ so as to generate plots in the X-Z plane. The `unapply` command is used to free up w and u in *eq1* and *eq2* for plotting purposes.

```
>  a:=1: Y:=0: A:=unapply(eq1,w): B:=unapply(eq2,u):
```

Using A and B and the `implicitplot` command, representative contours are drawn for fixed w (solid circles) in `gr1` and fixed u (dashed circles) in `gr2`

```
>  gr1:=implicitplot({seq(A(n/2),n=-4..-1),seq(A(n/2),n=1..4)},
   X=-5*a..5*a,Z=-5*a..5*a,grid=[70,70],color=red,thickness=2):
```

```
>  gr2:=implicitplot({seq(B(0.2*Pi*n),n=1..4)},X=-5*a..5*a,
   Z=-5*a..5*a,grid=[70,70],color=blue,thickness=2,linestyle=2):
```

and superimposed to produce Fig. 4.11.

```
>  display({gr1,gr2},scaling=constrained);
```



Figure 4.11: Contours for fixed w (solid circles) and fixed u (dashed).

The corresponding 3-dimensional surfaces result on rotating the figure about the vertical (Z) axis, the circles becoming spheres, hence the name "bispherical" for the coordinate system. From the defining relations between the bispherical and Cartesian systems, one has $y/x = \sin v / \cos v = \tan v$, so holding v fixed will produce half-planes passing through the z-axis.

A functional operator L is created for generating the Laplacian of $S(u, v, w)$ on specifying the coordinates u, v, w and the scale factors h_u, h_v, h_w.

```
> L:=(u,v,w,hu,hv,hw)->(diff(hv*hw*diff(S(u,v,w),u)/hu,u)
    +diff(hu*hw*diff(S(u,v,w),v)/hv,v)
    +diff(hu*hv*diff(S(u,v,w),w)/hw,w))/(hu*hv*hw):
```

An operator H is formed for producing and simplifying the scale factors.

```
> H:=u->simplify(sqrt(diff(x,u)^2+diff(y,u)^2+diff(z,u)^2),
        symbolic):
```

Using H the three scale factors are explicitly calculated.

```
> h[u]:=H(u); h[v]:=H(v); h[w]:=H(w);
```

$$h_u := \frac{1}{\cosh(w) - \cos(u)} \quad h_v := \frac{\sin(u)}{\cosh(w) - \cos(u)} \quad h_w := \frac{1}{\cosh(w) - \cos(u)}$$

Then employing L, Laplace's equation, $\nabla^2 S(u, v, w) = 0$, is generated.

```
> Lap:=L(u,v,w,h[u],h[v],h[w])=0;
```

$$Lap := \left(\frac{\cos(u)(\frac{\partial}{\partial u} S(u,v,w))}{\cosh(w) - \cos(u)} + \frac{\sin(u)^2(\frac{\partial}{\partial u} S(u,v,w))}{(\cosh(w) - \cos(u))^2} + \frac{\sin(u)(\frac{\partial^2}{\partial u^2} S(u,v,w))}{\cosh(w) - \cos(u)} \right.$$

$$+ \frac{\frac{\partial^2}{\partial v^2} S(u,v,w)}{(\cosh(w) - \cos(u))\sin(u)} + \frac{\sin(u)\,(\frac{\partial}{\partial w}\, S(u,v,w))\sinh(w)}{(\cosh(w) - \cos(u))^2}$$

$$\left. + \frac{\sin(u)\,(\frac{\partial^2}{\partial w^2} S(u,v,w))}{\cosh(w) - \cos(u)} \right) (\cosh(w) - \cos(u))^3 / \sin(u) = 0$$

An unsuccessful attempt is made to separate Laplace's equation by assuming that $S(u, v, w) = U(u)\, V(v)\, W(w)$.

```
> pdsolve(Lap,HINT=U(u)*V(v)*W(w));
```

Warning : Incomplete separation.

$$(S(u,\, v,\, w) = V(v)\, \_F1(u,\, w))\,\&\text{where}[\{\frac{d^2}{dv^2}\, V(v) = \_c_2\, V(v), \cdots\cdots$$

The separation is incomplete, an ODE resulting for $V(v)$, but the u and w dependence remaining coupled in a PDE, which is not displayed here in the text. Supplying the modified separation assumption as a hint, Laplace's equation is now completely separated.

```
> pdsolve(Lap,HINT=sqrt((cosh(w)-cos(u)))*U(u)*V(v)*W(w),
  INTEGRATE);
```

$$(S(u, v, w) = \sqrt{\cosh(w) - \cos(u)}\, U(u)\, V(v)\, W(w))\ \&\text{where}$$
$$[\{\{W(w) = \_C5\, e^{(\sqrt{\_c_3}\, w)} + \_C6\, e^{(-\sqrt{\_c_3}\, w)}\},$$
$$\{V(v) = \_C3\, e^{(\sqrt{\_c_2}\, v)} + \_C4\, e^{(-\sqrt{\_c_2}\, v)}\},$$
$$\{U(u) = \_C1\, (\frac{1}{2}\cos(2\,u) - \frac{1}{2})^{(1/2\, I\, \sqrt{\_c_2})} \sin(2\,u) \text{hypergeom}([\frac{1}{2}\, I\, \sqrt{\_c_2}$$
$$+\frac{1}{2}\sqrt{\_c_3} + \frac{3}{4}, \frac{1}{2}\, I\, \sqrt{\_c_2} - \frac{1}{2}\sqrt{\_c_3} + \frac{3}{4}], [\frac{3}{2}], \frac{1}{2}\cos(2\,u) + \frac{1}{2})\,/$$
$$\sqrt{1 - \cos(2\,u)} + \cdots\}\}]$$

$W(w)$ and $V(v)$ are both expressed in terms of exponentials, but U is given in terms of *hypergeometric* functions. The hypergeometric function $F(a, b; c; z)$ is given by the following infinite series [AS72], where Γ is the Gamma function,

$$F(a, b; c; z) = \frac{\Gamma(c)}{\Gamma(a)\,\Gamma(b)} \sum_{n=0}^{\infty} \frac{\Gamma(a+n)\,\Gamma(b+n)}{\Gamma(c+n)} \frac{z^n}{n!}.$$

4.3.2 A Shell Problem, Not a Shell Game

Insurance. An ingenious modern game of chance in which the player is permitted to enjoy the comfortable conviction that he is beating the man who keeps the table.
Ambrose Bierce, American author, *The Devil's Dictionary (1881–1906)*

In creating physics exams at the freshman level, I often play a bit of a shell game, presenting problems similar to those that the students have solved for homework, but in new guises and combinations. The hope is that they really understand the underlying principles and methods and haven't merely memorized the solutions to the homework problems. At the senior level, I rely less on "disguise " and more on having students explore challenging problems, even "standard" ones, in some depth. A computer algebra approach is encouraged as an auxiliary tool. The following recipe, submitted by Ms. I. M. Curious, is based on a standard problem appearing on an exam given to my senior electromagnetic theory class.

A very long circular cylindrical shell of dielectric constant ϵ and inner and outer radii a and b, respectively, is placed in a previously uniform electric field $\vec{E}_0$ with the cylinder axis perpendicular to the field. The medium inside ($r < a$) and outside ($r > b$) the cylindrical shell has a dielectric constant of unity.

- **(a)** Determine the potential and electric field in the three regions.
- **(b)** Taking $\epsilon = 3$, $a = 1$, $b = 2$, and $E_0 = 1$, plot the equipotentials and electric field vectors in all three regions in a single figure. Discuss the results and explore the effect of changing the parameter values.

In addition to the plots library package, I. M. loads the plottools and VectorCalculus packages. Plottools contains the `circle` command which she will use for drawing the inner and outer radii of the cylindrical shell. The VectorCalculus package is needed for the `Laplacian` and `Gradient` commands.

```
> restart: with(plots): with(plottools): with(VectorCalculus):
```

Neglecting end effects, the cylindrical shell is taken to be infinitely long in the z direction, thus reducing the problem to 2 dimensions in the x-y plane. Noting the circular symmetry, I. M. introduces the polar coordinates (r, θ) with $x = r\cos\theta$ and $y = r\sin\theta$, r being measured from the cylinder axis and θ from the x-axis. Laplace's equation is entered in polar coordinates and expanded.

```
> pde:=expand(Laplacian(phi(r,theta),'polar'[r,theta]))=0;
```

$$pde := \frac{\frac{\partial}{\partial r}\phi(r,\theta)}{r} + \left(\frac{\partial^2}{\partial r^2}\phi(r,\theta)\right) + \frac{\frac{\partial^2}{\partial \theta^2}\phi(r,\theta)}{r^2} = 0$$

Using the separation of variables method, a general solution is sought of the form $\phi(r,\theta) = R(r)\,\Theta(\theta)$. For convenience, I. M. replaces the separation constant $\sqrt{\_c_1}$ that appears on the rhs of *sol* with the symbol k.

```
> sol:=pdsolve(pde,HINT=R(r)*Theta(theta),INTEGRATE,build);
> sol2:=subs(sqrt(_c[1])=k,rhs(sol));
```

$$sol2 := \_C3\sin(k\,\theta)\,\_C1\,r^k + \frac{\_C3\sin(k\,\theta)\,\_C2}{r^k} + \_C4\cos(k\,\theta)\,\_C1\,r^k + \frac{\_C4\cos(k\,\theta)\,\_C2}{r^k}$$

Taking the electric field to be in the x direction, I. M. notes that the solution must have reflection symmetry (is unchanged if $\theta \to -\theta$) around the x axis. So she removes the sine terms, which are odd functions of θ, from *sol2*. As $r \to \infty$, the electric field must remain uniform and is given by $\vec{E}_0 = E_0\,\hat{e}_x = -(\partial\phi/\partial x)\,\hat{e}_x$, so the asymptotic potential is $\phi = -E_0\,x = -E_0\,r\cos\theta$, the arbitrary constant in the potential being set equal to zero. This immediately implies that $k = 1$, which must hold in every region to satisfy the boundary conditions. I. M. substitutes $k = 1$ and, without loss of generality, also sets the redundant coefficient $\_C4$ equal to 1 as well.

```
> sol3:=subs({_C4=1,k=1},remove(has,sol2,sin));
```

$$sol3 := \cos(\theta)\,\_C1\,r + \frac{\cos(\theta)\,\_C2}{r}$$

I. M. labels ϕ in the regions $r < a$, $a < r < b$, and $r > b$ as $\phi1$, $\phi2$, and $\phi3$. An operator `P` is formed for relabeling the coefficients $\_C1$ and $\_C2$ for each ϕ.

```
> P:=(u,v)->subs({_C1=u,_C2=v},sol3):
```

For $r < a$, the $1/r$ term must be removed from $\phi1$ for it to remain finite at $r = 0$. So I. M. forms $\phi1$ by setting $v = 0$ in `P` and $u = A_1$. For $\phi2$, she chooses $u = A_2$, $v = B_2$. For $\phi3$, she takes $u = -E0$, in order to match the asymptotic boundary condition as $r \to \infty$, and $v = B_3$.

```
> phi1:=P(A[1],0); phi2:=P(A[2],B[2]); phi3:=P(-E0,B[3]);
```

$$\phi 1 := \cos(\theta)\, A_1\, r$$

$$\phi 2 := \cos(\theta)\, A_2\, r + \frac{\cos(\theta)\, B_2}{r}$$

$$\phi 3 := -\cos(\theta)\, E0\, r + \frac{\cos(\theta)\, B_3}{r}$$

With A_1, A_2, B_2, and A_3 unknown, 4 boundary conditions are required. The potentials $\phi 1 = \phi 2$ at $r = a$ and $\phi 2 = \phi 3$ at b for arbitrary θ. The following operator `F` is created to match the potentials u and v at a radius $r = R$.

```
> F:=(u,v,R)-> expand(eval((u=v)/cos(theta),r=R)):
```

Using `F`, the above boundary conditions are applied in *eq1* and *eq2*.

```
> eq1:=F(phi1,phi2,a); eq2:=F(phi2,phi3,b);
```

$$eq1 := A_1\, a = A_2\, a + \frac{B_2}{a}$$

$$eq2 := A_2\, b + \frac{B_2}{b} = -E0\, b + \frac{B_3}{b}$$

The radial component of the *displacement vector* $\vec{D} = \epsilon\, \vec{E}$ is continuous at the boundaries, so $\partial\phi 1/\partial r = \epsilon\,(\partial\phi 2/\partial r)$ at $r = a$ and $\epsilon\,(\partial\phi 2/\partial r) = \partial\phi 3/\partial r$ at $r = b$. An operator `G` is created for equating the radial derivative of u and v at a radius R. Using `G`, the two boundary conditions are applied in *eq3* and *eq4*.

```
> G:=(u,v,R)->expand(eval(diff(u=v,r),r=R)/cos(theta)):
> eq3:=G(phi1,epsilon*phi2,a); eq4:=G(epsilon*phi2,phi3,b);
```

$$eq3 := A_1 = \varepsilon\, A_2 - \frac{\varepsilon\, B_2}{a^2}$$

$$eq4 := \varepsilon\, A_2 - \frac{\varepsilon\, B_2}{b^2} = -E0 - \frac{B_3}{b^2}$$

The four equations are solved for the four coefficients and *sol4* assigned.

```
> sol4:=solve({eq1,eq2,eq3,eq4},{A[1],A[2],B[2],B[3]});
  assign(sol4):
```

The potentials $\phi 1$, $\phi 2$, and $\phi 3$ are now determined, the coefficients being automatically substituted.

```
> phi1:=phi1; phi2:=simplify(phi2); phi3:=phi3;
```

$$\phi 1 := -\frac{4\cos(\theta)\,\varepsilon\, E0\, b^2\, r}{2\,\varepsilon\, a^2 - a^2 + 2\,\varepsilon\, b^2 + b^2 - \varepsilon^2\, a^2 + b^2\, \varepsilon^2}$$

$$\phi 2 := -\frac{2\cos(\theta)\, E0\, b^2\,(r^2\,\varepsilon + r^2 + \varepsilon\, a^2 - a^2)}{(2\,\varepsilon\, a^2 - a^2 + 2\,\varepsilon\, b^2 + b^2 - \varepsilon^2\, a^2 + b^2\,\varepsilon^2)\, r}$$

$$\phi 3 := -\cos(\theta)\, E0\, r + \frac{\cos(\theta)\, E0\, b^2\,(-\varepsilon^2\, a^2 + b^2\,\varepsilon^2 + a^2 - b^2)}{(2\,\varepsilon\, a^2 - a^2 + 2\,\varepsilon\, b^2 + b^2 - \varepsilon^2\, a^2 + b^2\,\varepsilon^2)\, r}$$

An operator `EF` is formed for calculating the electric field in polar coordinates, given some potential f.

```
> EF:=f->-Gradient(f,'polar'[r,theta]):
```

The electric field is then explicitly calculated in each region, but only the field *EF1* in region 1 is displayed here.

```
> EF1:=EF(phi1); EF2:=EF(phi2); EF3:=EF(phi3);
```

$$EF1 := \frac{4\cos(\theta)\,\varepsilon\,E0\,b^2}{2\,\varepsilon\,a^2 - a^2 + 2\,\varepsilon\,b^2 + b^2 - \varepsilon^2\,a^2 + \varepsilon^2\,b^2}\,\overline{e}_r - \frac{4\sin(\theta)\,\varepsilon\,E0\,b^2}{2\,\varepsilon\,a^2 - a^2 + 2\,\varepsilon\,b^2 + b^2 - \varepsilon^2\,a^2 + \varepsilon^2\,b^2}\,\overline{e}_\theta$$

The parameter values $\epsilon=3$, $a=1$, $b=2$, and $E_0=1$ are now entered.

```
> epsilon:=3: a:=1: b:=2: E0:=1:
```

For plotting purposes, I. M. changes to Cartesian coordinates by entering $r=\sqrt{x^2+y^2}$, $\cos\theta=x/r$ and $\sin\theta=y/r$.

```
> r:=sqrt(x^2+y^2): cos(theta):=x/r: sin(theta):=y/r:
```

The potential V for all three regions is formed with the `piecewise` command and the radical expressions simplified with the `radsimp` command. The electric field *Ef* is then calculated from V using the `Gradient` command and again the radicals are simplified.

```
> V:=radsimp(piecewise(r<a,phi1,r<b,phi2,phi3));
  Ef:=-radsimp(Gradient(V,[x,y]));
```

$$V := \begin{cases} -\dfrac{4\,x}{5} & \sqrt{x^2+y^2} < 1 \\ -\dfrac{4\,(2\,x^2+2\,y^2+1)\,x}{15\,(x^2+y^2)} & \sqrt{x^2+y^2} < 2 \\ -\dfrac{(5\,x^2+5\,y^2-8)\,x}{5\,(x^2+y^2)} & \mathit{otherwise} \end{cases}$$

The equipotentials of V are produced in `gr1` over the range $x=-4..4$, $y=-4..4$ with the `contourplot` command, 25 contours being requested. I. M. colors the plot by including `filled=true` as an option.

```
> gr1:=contourplot(V,x=-4..4,y=-4..4,contours=25,filled=true):
```

An operator `C` is formed to plot a thick red circle of radius r, centered at the origin. Then `C` is used in `gr2` and `gr3` to produce circles of radius a and b, representing the inner and outer radii of the cylindrical shell.

```
> C:=r->circle([0,0],r,color=red,thickness=3):
  gr2:=C(a): gr3:=C(b):
```

The `fieldplot` command is used in `gr4` to plot the electric field vectors as medium sized blue arrows. The density of the arrows is controlled.

```
> gr4:=fieldplot([Ef[1],Ef[2]],x=-4..4,y=-4..4,color=blue,
        arrows=MEDIUM,grid=[20,20]):
```

The four graphs are superimposed with the `display` command, the scaling being constrained. The resulting picture is shown in Figure 4.12.

```
> display({gr1,gr2,gr3,gr4},scaling=constrained);
```



Figure 4.12: The electric field arrows and equipotentials for a cylindrical dielectric shell inserted in a previously uniform (horizontal) electric field.

Referring to the figure, I. M. notes that the electric field for $r < a$ is completely horizontal and therefore parallel to the asymptotic field. The arrows are slightly shorter however. Quantitatively, the ratio of the electric field for $r < a$ to the asymptotic field is $A_1/E_0 = 4/5$. Unlike the situation for a conductor, the surfaces of the dielectric shell are not equipotentials so the electric field vectors are not perpendicular to the inner and outer surfaces of the shell.

I. M. leaves it to you, the reader, to explore her recipe. For example, you might take ϵ to be much larger, or alter b with a fixed. She reminds you that in interpreting any result to remember that the arrows are not field lines.

4.3.3 The Little Drummer Boy

Shall I play for you! pa rum pum pum on my drum.
From the Christmas carol, *Little Drummer Boy*

Little Daniel loves to bang on large pots and pans with a wooden spoon, creating his own version of music, which to untrained adult ears such as mine sounds a lot like noise. Here's a quieter version of "drum playing".

A large circular drumhead of radius $r = a = 1$ m has its perimeter fixed. If the drumhead has an initial shape $U(r, \theta, 0) = r\,(1 - r^2/a^2)\,\sin(2\,\theta)/20$ and is released from rest, determine the shape of the drumhead at arbitrary time

$t > 0$. Take the wave speed to be $c = 1$ m/s. Then, animate the motion of the drumhead in time steps of 0.1 s over the interval $t = 0$ to 2 s.

After loading the plots and VectorCalculus packages, the wave equation *pde* is entered in polar coordinates (r, θ), making use of the `Laplacian` command.

```
> restart: with(plots): with(VectorCalculus):
> pde:=expand(Laplacian(U(r,theta,t),'polar'[r,theta]))
      =(1/c^2)*diff(U(r,theta,t),t,t);
```

$$pde := \frac{\frac{\partial}{\partial r} U(r,\, \theta,\, t)}{r} + \left(\frac{\partial^2}{\partial r^2} U(r,\, \theta,\, t)\right) + \frac{\frac{\partial^2}{\partial \theta^2} U(r,\, \theta,\, t)}{r^2} = \frac{\frac{\partial^2}{\partial t^2} U(r,\, \theta,\, t)}{c^2}$$

Then *pde* is analytically solved, assuming that $U(r, \theta, t) = R(r)\,\Theta(\theta)\,T(t)$.

```
> sol:=pdsolve(pde,HINT=R(r)*Theta(theta)*T(t),INTEGRATE,
        build);
```

$$sol := U(r,\, \theta,\, t) = e^{(\sqrt{\_c_3}\, t)}\, e^{(\sqrt{\_c_2}\, \theta)}\, \_C5\, \_C3\, \_C1\, \mathrm{BesselJ}(\sqrt{-\_c_2},\, \frac{\sqrt{-\_c_3}\, r}{c})$$
$$+ e^{(\sqrt{\_c_3}\, t)}\, e^{(\sqrt{\_c_2}\, \theta)}\, \_C5\, \_C3\, \_C2\, \mathrm{BesselY}(\sqrt{-\_c_2},\, \frac{\sqrt{-\_c_3}\, r}{c}) + \cdots$$

The answer involves exponentials and Bessel functions of the first and second kinds. The separation constants $\_c_2$ and $\_c_3$ on the rhs of *sol* are replaced with $-p^2$ and $-c^2\, k^2$, and the result simplified with the `symbolic` option.

```
> U1:=simplify(subs({_c[2]=-p^2,_c[3]=-c^2*k^2},rhs(sol)),
      symbolic);
```

$$U1 := \_C5\, \_C3\, \_C1\, \mathrm{BesselJ}(p,\, k\, r)\, e^{((c\, k\, t + p\, \theta)\, I)}$$
$$+ \_C5\, \_C3\, \_C2\, \mathrm{BesselY}(p,\, k\, r)\, e^{((c\, k\, t + p\, \theta)\, I)} + \cdots$$

The Bessel functions $Y_p(k\, r)$ of the second kind diverge at $r = 0$, so are removed from *U1*. Then *U2* is converted to trig form and expanded in *U3*.

```
> U2:=remove(has,U1,BesselY);
> U3:=expand(convert(U2,trig));
```

$$U3 := \_C5\, \_C3\, \_C1\, \mathrm{BesselJ}(p,\, k\, r)\cos(c\, k\, t)\cos(p\, \theta)$$
$$- \_C5\, \_C3\, \_C1\, \mathrm{BesselJ}(p,\, k\, r)\sin(c\, k\, t)\sin(p\, \theta)$$
$$+ \_C5\, \_C4\, \_C1\, \mathrm{BesselJ}(p,\, k\, r)\sin(c\, k\, t)\cos(p\, \theta)\, I$$
$$+ \_C5\, \_C3\, \_C1\, \mathrm{BesselJ}(p,\, k\, r)\cos(c\, k\, t)\sin(p\, \theta)\, I + \cdots$$

The initial transverse velocity of the drumhead is zero, so the $\sin(c\, k\, t)$ terms must be removed from *U3*. Since the initial shape involves a sine function only, there can be no cosine terms present in the solution. The $\cos(p\, \theta)$ terms are therefore also removed in *U4* and the result then factored.

```
> U4:=factor(remove(has,U3,{sin(c*k*t),cos(p*theta)}));
```

$$U4 := \_C1\, (\_C3 - \_C4)\, (\_C5 + \_C6)\sin(p\, \theta)\cos(c\, k\, t)\, \mathrm{BesselJ}(p,\, k\, r) I$$

The solution must vanish at the perimeter, so the Bessel functions $J_p(k\,r)$ must be zero at $r=a$. Thus, $k\,a$ must equal the mth zero of J_p, with $m=1, 2, \ldots$. The allowed k values are now entered.

```
> k:=BesselJZeros(p,m)/a:
```

The parameter values $a=1$ and $c=1$ are given. To match the initial angular dependence, we must have $p=2$, i.e., only the Bessel function J_2 will be present in the solution. The radial portion of the initial shape is entered in f. The form of the Bessel functions is given in g.

```
> a:=1: c:=1: p:=2: f:=r*(1-r^2/a^2)/20; g:=BesselJ(p,k*r);
```

$$f := \frac{r\,(-r^2+1)}{20} \qquad g := \text{BesselJ}(2,\, \text{BesselJZeros}(2,\, m)\, r)$$

The messy constants are removed from $U4$ with the following `select` command.

```
> U[2,m]:=select(has,U4,{cos,sin,BesselJ});
```

$$U_{2,\,m} := \cos(\text{BesselJZeros}(2,\, m)\, t)\sin(2\,\theta)\,\text{BesselJ}(2,\, \text{BesselJZeros}(2,\, m)\, r)$$

Making use of orthogonality and noting that the weight function for the Bessel functions is r, the mth coefficient is given by $A_{2,m} = \int_0^a f\,r\,g\,dr / \int_0^a r\,g^2\,dr$.

```
> A[2,m]:=int(f*r*g,r=0..a)/int(r*g^2,r=0..a):
```

The first 4 terms of the series solution U are now displayed in decimal form.

```
> U:=evalf(sum(A[2,m]*U[2,m],m=1..4));
```

$$\begin{aligned} U := {} & 0.04223579864\,\cos(5.135622302\,t)\sin(2.\,\theta)\,\text{BesselJ}(2.,\, 5.135622302\,r) \\ & + 0.002159293112\cos(8.417244140\,t)\sin(2.\,\theta)\,\text{BesselJ}(2.,\, 8.417244140\,r) \\ & + 0.005912588156\cos(11.61984117\,t)\sin(2.\,\theta)\text{BesselJ}(2.,\, 11.61984117\,r) \\ & + 0.001008506062\cos(14.79595178\,t)\sin(2.\,\theta)\,\text{BesselJ}(2.,\, 14.79595178\,r) \end{aligned}$$

For animation purposes, we convert from polar coordinates to Cartesian coordinates, setting $r=\sqrt{x^2+y^2}$ and using the trig identity $\sin(2\,\theta)=2\,\sin\theta\,\cos\theta = 2\,x\,y/r^2$. Note that a floating point evaluation is used in entering the latter so that the substitution will actually occur. This is necessary because a floating point evaluation was used in expressing U.

```
> r:=sqrt(x^2+y^2): sin(evalf(2*theta)):=2*x*y/r^2:
```

The solution is then expressed as the piecewise function $UU=U$ for $r<a$ and 0 for $r>a$.

```
> UU:=evalf(piecewise(r<a,U,r>a,0)):
```

To animate UU, a functional operator `gr` is created to make a 3-dimensional plot of the drumhead shape on the ith time step, the stepsize being 0.1 s.

```
> gr:=i->plot3d(eval(UU,t=0.1*i),x=-a..a,y=-a..a,
  style=patchcontour,shading=zhue):
```

Then using the sequence command, the profiles on 20 consecutive time steps are displayed. The `insequence=true` option is included to produce the animation.

```
> display([seq(gr(i),i=1..20)],insequence=true,axes=framed);
```

If you wish to see the drumhead animation, execute the recipe on your computer, then click on the computer plot and on the start arrow in the tool bar.

4.3.4 The Cannon Ball

The sound of a kiss is not so loud as that of a cannon, but its echo lasts a great deal longer.
Oliver Wendell Holmes Sr., American writer, physician, (1809–94)

In the re-enactment of a Civil war battle, a cannon is fired and a hot spherical iron cannon ball plunges into an icy lake whose temperature is very close to freezing (0°C). If the cannon ball has a radius $R = 20$ cm and is initially 100°C throughout on entering the lake, determine the temperature distribution inside the cooling cannon ball as a function of time. Plot the temperature distribution in 1 minute intervals up to 15 minutes. What is the temperature at the center of the cannon ball 15 minutes after plunging into the lake? For iron, $K/(\rho C) = 0.185$ in cgs units, where K is the thermal conductivity, ρ the density, and C the specific heat.

After loading the plots and VectorCalculus packages,

```
> restart: with(plots): with(VectorCalculus):
```

the heat diffusion equation $\nabla^2 T = (1/a^2)\,(\partial T/\partial t)$, with T the temperature and $a^2 \equiv K/(\rho\, C)$, is entered in spherical polar coordinates $(r,\, \theta,\, \phi)$. r is the radial distance from the center of the cannon ball, θ is the angle that the radial vector makes with the z-axis, and ϕ is the angle that the projection of the radial vector into the x-y plane makes with the x-axis.

```
> pde:=expand(Laplacian(T(r,theta,phi,t),'spherical')
      [r,theta,phi]=diff(T(r,theta,phi,t),t)/a^2;
```

$$pde := \frac{2\,(\frac{\partial}{\partial r}\,T(r,\theta,\phi,t))}{r} + (\frac{\partial^2}{\partial r^2}\,T(r,\theta,\phi,t)) + \frac{\cos(\theta)\,(\frac{\partial}{\partial \theta}\,T(r,\theta,\phi,t))}{r^2\sin(\theta)} + \frac{\frac{\partial^2}{\partial \theta^2}\,T(r,\theta,\phi,t)}{r^2} + \frac{\frac{\partial^2}{\partial \phi^2}\,T(r,\theta,\phi,t)}{r^2\sin(\theta)^2} = \frac{\frac{\partial}{\partial t}\,T(r,\theta,\phi,t)}{a^2}$$

Since the initial temperature of the cannon ball is uniform throughout, the solution will have no angular dependence. Assuming that $T(r,\theta,\phi,t) = R(r)\,F(t)$, the heat flow equation *pde* is analytically solved using the `pdsolve` command.

```
> sol:=pdsolve(pde,HINT=R(r)*F(t),INTEGRATE,build);
```

$$sol := T(r,\,\theta,\,\phi,\,t) = \frac{\_C3\, e^{(a^2\,\_c_1\,t)}\,\_C1\,\sinh(\sqrt{\_c_1}\,r)}{r} + \frac{\_C3\, e^{(a^2\,\_c_1\,t)}\,\_C2\,\cosh(\sqrt{\_c_1}\,r)}{r}$$

The hyperbolic sine (sinh) term remains finite at the origin $(r = 0)$, but the cosh term diverges to ∞ and must be removed from the rhs of *sol*.

```
> T1:=remove(has,rhs(sol),cosh);
```

$$T1 := \frac{\_C3\, e^{(a^2\, \_c_1\, t)}\, \_C1 \sinh(\sqrt{\_c_1}\, r)}{r}$$

The separation constant $\_c_1$ is replaced with $-k^2$ in $T1$ and the result simplified with the `symbolic` option.

```
> T2:=simplify(subs(_c[1]=-k^2,T1),symbolic);
```

$$T2 := \frac{\_C3\, e^{(-a^2\, k^2\, t)}\, \_C1 \sin(k\, r)\, I}{r}$$

It should be noted that the term $\sin(k\, r)/r$ is ([AS72]) just the zeroth order *spherical Bessel function* of the first kind.[3] Making use of the `select` command, the coefficient combination $\_C3\, \_C1\, I$ in $T2$ is replaced with the symbol A.

```
> T3:=A*select(has,T2,{exp,sin,r});
```

$$T3 := \frac{A\, e^{(-a^2\, k^2\, t)} \sin(k\, r)}{r}$$

Taking the radius of the cannon ball to be R, the surface of the ball is held at 0 degrees, so $\sin(k\, R) = 0$, and therefore $k = n\, \pi/R$, with $n = 1$, 2, ... Entering this result, the nth normal mode of the temperature is displayed in $T4$.

```
> k:=n*Pi/R: T4:=T3;
```

$$T4 := \frac{A\, e^{\left(-\frac{a^2\, n^2\, \pi^2\, t}{R^2}\right)} \sin\left(\frac{n\, \pi\, r}{R}\right)}{r}$$

The initial temperature is 100° throughout the cannon ball. Noting that the weight function ([AS72]) for the spherical Bessel functions is r^2, orthogonality leads to the following expression for the coefficients:

$$A = \int_0^R r^2\, 100\, (\sin(k\, r)/r)\, dr \Big/ \int_0^R r^2\, (\sin(k\, r)/r)^2\, dr.$$

A is now calculated, assuming that n is an integer.

```
> A:=int(r^2*100*sin(k*r)/r,r=0..R)/int(r^2*(sin(k*r)/r)^2,
    r=0..R) assuming n::integer;
```

$$A := \frac{200\, (-1)^{(1+n)}\, R}{n\, \pi}$$

The formal series representation of the temperature distribution T inside the cannon ball is now completely determined. Retaining 200 terms in the series, T has the following structure.

```
> T:=Sum(T4,n=1..200);
```

$$T := \sum_{n=1}^{200} \left(\frac{200\, (-1)^{(1+n)}\, R\, e^{\left(-\frac{a^2\, n^2\, \pi^2\, t}{R^2}\right)} \sin\left(\frac{n\, \pi\, r}{R}\right)}{n\, \pi\, r} \right)$$

[3]The spherical Bessel functions j_n of the first kind are related to the "ordinary" Bessel functions by the relation $j_n(x) \equiv \sqrt{(\pi/2x)}\, \mathrm{J}_{n+1/2}(x)$.

Taking $R=20$ and $a=\sqrt{0.185}$, T is now evaluated, but not displayed.

```
>  T:=eval(value(T),{R=20,a=sqrt(0.185)}):
```

An arrow operator `gr` is formed to plot T at 60 s (1 min) intervals.

```
>  gr:=i->plot(eval(T,t=i*60),r=0..20,numpoints=1000):
```

Using `gr` and the sequence command, the temperature T is displayed as a function of radius r in Figure 4.13 at $t=0$ (top curve), $t=1$ min (next lowest curve), etc, to $t=15$ minutes (bottom curve).

```
>  display(seq(gr(i),i=0..15),labels=["r","T"]);
```



Figure 4.13: Time evolution of the temperature T inside the cannon ball.

The "ringing" in the initial temperature distribution is, of course, the Gibb's phenomenon due to the initial step function temperature profile. Taking the limit of T as $r \to 0$, the temperature at the center of the sphere at 15 minutes, or 900 seconds, is calculated and found to be about 3.3°C.

```
>  Tcenter:=eval(limit(T,r=0),t=900);
```

$$Tcenter := 3.287377363$$

4.3.5 Variation on a Split-sphere Potential

Would you convey my compliments to the purist who reads your proofs and tell him or her ... that when I split an infinitive, God damn it, I split it so it will stay split.
Raymond Chandler, American writer of detective fiction, (1888–1959)

The following recipe is based on a simple variation of a standard problem in electrostatics. The objective is to completely determine the potential V inside and outside a hollow sphere of unit radius with a specified piecewise potential on the spherical surface. With θ measured from the positive z-axis (pointing vertically upwards), the spherical surface between $\theta = 0$ and $45°$ has the constant potential $V0$, the intermediate section between 45° and 135° has a variable potential given by $\sqrt{2}\cos(\theta)\,V0$, and the lower portion between 135° and 180° held at $-V0$. Keeping the first 6 non-zero terms in V and taking $V0=1$, plot the equipotentials corresponding to $V=0.8$, 0.6, ..., -0.6, -0.8.

After loading the necessary library packages, Laplace's equation is entered in spherical coordinates, the origin taken at the center of the sphere. By symmetry, V must be independent of the azimuthal[4] angle ϕ, i.e., $V = V(r,\theta)$.

```
> restart: with(plots): with(VectorCalculus):
> pde:=expand(Laplacian(V(r,theta),'spherical'
      [r,theta,phi]))=0;
```

$$pde := \frac{2\,(\frac{\partial}{\partial r}\mathrm{V}(r,\,\theta))}{r} + (\frac{\partial^2}{\partial r^2}\mathrm{V}(r,\,\theta)) + \frac{\cos(\theta)\,(\frac{\partial}{\partial \theta}\mathrm{V}(r,\,\theta))}{r^2\sin(\theta)} + \frac{\frac{\partial^2}{\partial \theta^2}\mathrm{V}(r,\,\theta)}{r^2} = 0$$

Then *pde* is analytically solved, assuming that $V(r,\theta) = R(r)\,\Theta(\theta)$, the result involving Legendre functions.

```
> V:=rhs(pdsolve(pde,HINT=R(r)*Theta(theta),INTEGRATE,build));
```

$$V := \frac{\_C3\,\mathrm{LegendreP}(-\frac{1}{2}+\frac{1}{2}\sqrt{1+4\,\_c_1},\cos(\theta))\,\_C1\,r^{(1/2\,\sqrt{1+4\,\_c_1})}}{\sqrt{r}} + \cdots$$

The separation constant $\_c_1$ is replaced in V with $-1/4+(n+1/2)^2$ and the result simplified with the `symbolic` option and then expanded.

```
> V:=expand(simplify(subs(_c[1]=-1/4+(n+1/2)^2,V),symbolic));
```

$$V := \_C3\,\mathrm{LegendreP}(n,\,\cos(\theta))\,\_C1\,r^n + \frac{\_C3\,\mathrm{LegendreP}(n,\,\cos(\theta))\,\_C2}{r\,r^n}$$
$$+\,\_C4\,\mathrm{LegendreQ}(n,\,\cos(\theta))\,\_C1\,r^n + \frac{\_C4\,\mathrm{LegendreQ}(n,\,\cos(\theta))\,\_C2}{r\,r^n}$$

V is expressed in terms of the Legendre functions of the first ($P_n(\cos\theta)$) and second ($Q_n(\cos\theta)$) kinds. The Q_n diverge at the end points of the θ range and must be rejected. The redundant constant $\_C3$ in the P_n terms is set equal to 1 and $\_C1$ and $\_C2$ are replaced with the symbols A and B.

```
> V:=subs({LegendreQ(n,cos(theta))=0,_C3=1,_C1=A,_C2=B},V);
```

$$V := \mathrm{LegendreP}(n,\,\cos(\theta))\,A\,r^n + \frac{\mathrm{LegendreP}(n,\,\cos(\theta))\,B}{r\,r^n}$$

For the inside ($r<1$) solution Vin, we set $B=0$ in V so that Vin doesn't diverge at the origin. For $r>1$, we take $A=0$ so that $Vout\to 0$ as $r\to\infty$.

```
> Vin:=subs(B=0,V); Vout:=subs(A=0,V);
```

$$Vin := \mathrm{LegendreP}(n,\,\cos(\theta))\,A\,r^n$$

[4]The angle between the projection of the radius vector into the x-y (horizontal) plane and the x-axis.

$$Vout := \frac{\text{LegendreP}(n,\, \cos(\theta))\, B}{r\, r^n}$$

Setting $u = \cos\theta$, the angular distributions in each region are entered.

```
> f1:=-V0: f2:=sqrt(2)*u*V0: f3:=V0:
```

Making use of orthogonality of the $P_n(u)$, the coefficients in the series representation of the solution are evaluated using $A_n = ((2\,n+1)/2)\,\int_{-1}^{1} f\, P_n(u)\, du$. An operator `AA` is introduced to evaluate the coefficients for a given n value.

```
> AA:=n->((2*n+1)/2)*(int(f1*LegendreP(n,u),u=-1..-1/sqrt(2))
        +int(f2*LegendreP(n,u),u=-1/sqrt(2)..1/sqrt(2))
        +int(f3*LegendreP(n,u),u=1/sqrt(2)..1)):
```

Then, employing `AA(n)`, the inside and outside solutions are determined, the series being terminated at $n = 12$.

```
> VIN:=sum(eval(Vin,A=AA(n)),n=0..12);
```

$$\begin{aligned} VIN := {} & \frac{5}{4}\cos(\theta) V0\, r - \frac{7}{32}\,\text{LegendreP}(3, \cos(\theta)) V0\, r^3 \\ & - \frac{11}{128}\,\text{LegendreP}(5, \cos(\theta)) V0\, r^5 + \frac{85}{2048}\,\text{LegendreP}(7, \cos(\theta)) V0\, r^7 \\ & + \frac{323}{8192}\,\text{LegendreP}(9, \cos(\theta)) V0\, r^9 - \frac{1219}{65536}\,\text{LegendreP}(11, \cos(\theta)) V0\, r^{11} \end{aligned}$$

```
> VOUT:=sum(eval(Vout,B=AA(n)),n=0..12);
```

$$VOUT := \frac{5}{4}\,\frac{\cos(\theta) V0}{r^2} - \frac{7}{32}\,\frac{\text{LegendreP}(3, \cos(\theta)) V0}{r^4} + \cdots$$

The equipotentials will now be plotted in the x-z plane, by setting $r = \sqrt{x^2 + z^2}$ and $\cos\theta = z/r$. We also set $V0 = 1$.

```
> r:=sqrt(x^2+z^2): cos(theta):=z/r: V0:=1.0:
```

A piecewise potential function *VPW* is formed with $V = VIN$ for $r < 1$ and $VOUT$ for $r > 1$.

```
> VPW:=piecewise(r<1,VIN,r>1,VOUT):
```

Loading the plottools package, a blue circle of radius 1 centered on the origin is produced in `c` to represent the spherical surface (a circle in 2 dimensions).

```
> with(plottools): c:=circle([0,0],1,color=blue,thickness=2):
```

The `contourplot` command is used in `cp` to plot the requested equipotentials.

```
> cp:=contourplot(VPW,x=-3..3,z=-4..4,contours=
  [seq(0.8-0.2*i,i=0..8)],grid=[60,60],thickness=2,color=red):
```

The graphs `c` and `cp` are now displayed together with constrained scaling,

```
> display({c,cp},scaling=constrained);
```

the resulting picture being shown in Figure 4.14.



Figure 4.14: Equipotentials for the split-sphere in the x-z plane.

The 3-dimensional equipotential surfaces are obtained by mentally rotating the picture about the z-axis. The recipe is easily adjusted to handle other variations on the angular distribution and can even be modified to handle cylindrical geometry.

4.3.6 Another Poisson Recipe

Every man is a potential genius until he does something.
Sir Herbert Beerbohm Tree, English actor-manager, (1853–1917)

In magnetostatics, the vector potential $\vec{A}(\vec{R})$ at a point $\vec{R}$, associated with a current density $\vec{J}$ in free space, satisfies the *vector Poisson equation*, [Gri99]

$$\nabla^2 \vec{A} = -\mu_0 \, \vec{J}, \tag{4.9}$$

where μ_0 is the permeability of free space. Equation (4.9) is three scalar Poisson equations, one for each Cartesian component, e.g., $\nabla^2 A_x = -\mu_0 \, J_x$. In any other curvilinear coordinate system, the unit vectors are functions of position. Thus, e.g., in spherical polar coordinates it is not true that $\nabla^2 A_r = -\mu_0 \, J_r$.
Assuming that $\vec{J} \to 0$ at infinity, the solution of Eq. (4.9) is given by,

$$\vec{A}(\vec{R}\,) = \frac{\mu_0}{4\,\pi} \int \frac{\vec{J}(\vec{R}_1)}{|\vec{R} - \vec{R}_1|} \, dv_1, \tag{4.10}$$

again representing three 3-dimensional integrals, e.g.,

$$A_x(x, y, z) = \frac{\mu_0}{4\,\pi} \int\!\!\int\!\!\int \frac{J_x(x_1, y_1, z_1)\, dx_1\, dy_1\, dz_1}{\sqrt{(x - x_1)^2 + (y - y_1)^2 + (z - z_1)^2}}.$$

If you want to calculate the integrals in Equation (4.10) in other curvilinear coordinates, you must first express $\vec{J}$ in terms of its Cartesian components. Once $\vec{A}$ is determined, the magnetic field is then given by $\vec{B} = \nabla \times \vec{A}$. As a representative example, consider the following magnetostatic problem.

A uniformly charged solid sphere of radius a carries a total charge Q and is spinning with angular velocity ω about the vertical z-axis. Determine the magnetic vector potential $\vec{A}$ inside and outside the sphere and then calculate the magnetic field $\vec{B}$ in both regions. Plot $\vec{B}/(\mu_0 Q\omega)$ for $a=1$.

Taking the origin at the sphere's center, we let $\vec{R}$ $(r,\, \theta,\, \phi)$ be the location of the "observation" point P inside or outside the sphere and $\vec{R}1$ $(r1,\, \theta1,\, \phi1)$ be the location of a current density "source" point P_1 inside the sphere. The polar angles θ and $\theta1$ are measured from the z-axis and the azimuthal angles ϕ and $\phi1$ from the projection of the radius vector into the (horizontal) x-y plane with the x-axis. To ensure a later simplification, it is assumed that $r > 0$.

```
> restart: with(plots): with(VectorCalculus): assume(r>0):
```

Since the charge is uniformly distributed in the sphere, the charge density ρ is equal to the total charge Q divided by the volume $4\,\pi\, a^3/3$ of the sphere. At a radial distance $r1$ and angle $\theta1$ with the z-axis, the source point P_1 has a linear velocity $r1\, \sin(\theta1)\,\omega$. So the current density magnitude J is equal to ρ times the linear velocity, the entered form of ρ being automatically substituted. The corresponding vector $\vec{J}$ points in the $\hat{\phi}$ direction for every P_1.

```
> rho:=Q/((4*Pi*a^3)/3); J:=rho*r1*sin(theta1)*omega;
```

$$\rho := \frac{3\,Q}{4\,\pi\, a^3} \qquad J := \frac{3}{4}\,\frac{Q\, r1\, \sin(\theta1)\,\omega}{\pi\, a^3}$$

Without loss of generality, let's take the observation point P to be in the x-z plane, so that $\phi=0$. The position vector $\vec{R}$ of P is now entered, being expressed in terms of the Cartesian unit vectors along the x and z axes.

```
> R:=<r*sin(theta),0,r*cos(theta)>;
```

$$R := r\sin(\theta)\,\mathrm{e_x} + r\cos(\theta)\,\mathrm{e_z}$$

The position vector $\vec{R}1$ of the source point P_1 is also entered.

```
> R1:=<r1*sin(theta1)*cos(phi1),r1*sin(theta1)*sin(phi1),
     r1*cos(theta1)>;
```

$$R1 := r1\, \sin(\theta1)\cos(\phi1)\,\mathrm{e_x} + r1\, \sin(\theta1)\sin(\phi1)\,\mathrm{e_y} + r1\, \cos(\theta1)\,\mathrm{e_z}$$

To calculate $\vec{A}$, let's first evaluate $1/(|\vec{R}-\vec{R}1|)=1/\sqrt{(\vec{R}-\vec{R}1)\cdot(\vec{R}-\vec{R}1)}$. This is done in f using the `DotProduct` command, the result then being simplified with the `symbolic` option.

```
> f:=simplify(1/sqrt(DotProduct(R-R1,R-R1)),symbolic);
```

$$f := \frac{1}{\sqrt{r^2 - 2\,r\sin(\theta)\, r1\, \sin(\theta1)\cos(\phi1) + r1^2 - 2\,r\cos(\theta)\, r1\, \cos(\theta1)}}$$

We could attempt to evaluate the integral in $\vec{A}$ directly, but it is more instructive to Taylor expand f about a specified value of $r1$, out to some given order, and see what the various orders contribute to the overall answer. This approach is equivalent to the *multipole expansion* discussed in standard electromagnetic texts such as Griffiths [Gri99]. So, a functional operator `T` is formed to Taylor expand f to order n about a specified point $r1 = d$. The `convert( ,polynom)` command is included to remove the order of term which would otherwise appear.

```
> T:=(n,d)->convert(taylor(f,r1=d,n),polynom):
```

The current density vector can be resolved into the Cartesian components $J_x = J\,\sin(\phi 1)$, $J_y = J\,\cos(\phi 1)$, and $J_z = 0$. At the observation point (chosen in the x-z plane), the J_x contribution to $\vec{A}$ will add up to zero, but the J_y contribution will not. But since, our choice of observation point in the x-z plane was arbitrary and there is complete rotational symmetry about the z-axis, the resultant component J_y will yield the ϕ component of $\vec{A}$. A functional operator `A` is created to perform the volume integration in (4.10) using J_y and the Taylor expansion of f and taking spherical polar coordinates. The volume element is $r1^2 \sin(\theta 1)\, d\theta 1\, d\phi 1\, dr1$. The angular coordinate $\theta 1$ ranges from 0 to π, while $\phi 1$ varies from 0 to 2π. The order n, the radial distance d about which Taylor expansion is taking place, and the lower and upper limits, $d1$ and $d2$, of the $r1$ integration must be specified. Again the result is simplified.

```
> A:=(n,d,d1,d2)->simplify((mu[0]/(4*Pi))*int(int(int(T(n,d)
     *J*cos(phi1)*r1^2*sin(theta1),theta1=0..Pi),r1=d1..d2),
     phi1=0..2*Pi),symbolic):
```

First, let's take the observation point P to be outside the sphere, i.e., $r > a$. Since $r1 \leq a$, then $r1/r < 1$ and we can Taylor expand f about $r1 = d = 0$. The limits of the $r1$ integration are $d1 = 0$ and $d2 = a$. Making uses of the operator `A` with the above arguments, the vector potential is evaluated in *Out* to order $n = 1$, 2, and 3 and the result assigned.

```
> Out:=seq(A||n=A(n,0,0,a),n=1..3); assign(Out):
```

$$Out := A1 = 0, \quad A2 = \frac{1}{20}\,\frac{\mu_0 \sin(\theta)\, a^2\, Q\, \omega}{\pi\, r^2}, \quad A3 = \frac{1}{20}\,\frac{\mu_0 \sin(\theta)\, a^2\, Q\, \omega}{\pi\, r^2}$$

For $n = 1$, the so-called *monopole* contribution $A1$ to the vector potential is 0, a well-known general result. For $n = 2$, there is a non-zero *dipole* contribution $A2$. For $n = 3$ (and higher), there is no additional contribution to the vector potential, indicating that outside the sphere the vector potential (and hence, the magnetic field) is that of a "pure" magnetic dipole. The vector potential *Aout* outside the sphere is now expressed in spherical polar coordinates, having only a ϕ component.

```
> Aout:=VectorField(<0,0,A2>,'spherical'[r,theta,phi]);
```

$$Aout := \frac{1}{20}\,\frac{\mu_0 \sin(\theta)\, a^2\, Q\, \omega}{\pi\, r^2}\,\bar{e}_\phi$$

Now, consider P to be inside the sphere, i.e., $r < a$. For $r1 < r$, we can again Taylor expand f about $r1 = 0$, the integration being from $r1 = 0$ to r. But for $r1 > r$, f is Taylor expanded about $r1 = \infty$, the $r1$ integration being from r to a. Adding the two contributions, $\vec{A}$ inside the sphere is calculated for $n = 1$, 2, and 3. No further contribution occurs for higher n values.

```
> In:=seq(AA||n=A(n,0,0,r)+A(n,infinity,r,a),n=1..3);
    assign(In):
```

$$In := AA1 = 0, \quad AA2 = \frac{1}{20}\,\frac{\mu_0 \sin(\theta)\,Q\,\omega\,r^3}{\pi\,a^3},$$

$$AA3 = \frac{1}{20}\,\frac{\mu_0 \sin(\theta)\,Q\,\omega\,r^3}{\pi\,a^3} - \frac{1}{8}\,\frac{\mu_0\,Q\,\omega\,r \sin(\theta)\,(-a^2+r^2)}{\pi\,a^3}$$

In terms of spherical polar coordinates, the vector potential *Ain* inside the sphere takes the following form.

```
> Ain:=VectorField(<0,0,AA||3>,'spherical'[r,theta,phi]);
```

$$Ain := \left(\frac{1}{20}\,\frac{\mu_0 \sin(\theta)\,Q\,\omega\,r^3}{\pi\,a^3} - \frac{1}{8}\,\frac{\mu_0\,Q\,\omega\,r \sin(\theta)\,(-a^2+r^2)}{\pi\,a^3}\right)\bar{\mathrm{e}}_\phi$$

Using `Curl`, the magnetic field is calculated outside and inside the sphere.

```
> Bout:=Curl(Aout); Bin:=simplify(Curl(Ain));
```

$$Bout := \frac{1}{10}\,\frac{\mu_0\,a^2\,Q\,\omega \cos(\theta)}{r^3\,\pi}\,\bar{\mathrm{e}}_{\mathrm{r}} + \frac{1}{20}\,\frac{\sin(\theta)\,\mu_0\,a^2\,Q\,\omega}{r^3\,\pi}\,\bar{\mathrm{e}}_\theta$$

$$Bin := -\frac{1}{20}\,\frac{\cos(\theta)\,\mu_0\,Q\,\omega\,(3\,r^2-5\,a^2)}{\pi\,a^3}\,\bar{\mathrm{e}}_{\mathrm{r}} + \frac{1}{20}\,\frac{\sin(\theta)\,\mu_0\,Q\,\omega\,(6\,r^2-5\,a^2)}{\pi\,a^3}\,\bar{\mathrm{e}}_\theta$$

To plot $\vec{B}/(\mu_0 Q\omega)$, the `MapToBasis` command is used to convert the normalized magnetic field to Cartesian coordinates.

```
> F:=u->MapToBasis(u/(mu[0]*Q*omega),'cartesian'[x,y,z]):
```

Taking $a=1$, we will plot the magnetic field in the x-z plane. To accomplish this, let's set $r=\sqrt{x^2+z^2}$ and form an operator `G` to evaluate the magnetic field for $y=0$.

```
> a:=1: r:=sqrt(x^2+z^2): G:=B->simplify(eval(F(B),y=0)):
```

The magnetic field inside and outside the sphere then takes the following forms, expressed in Cartesian coordinates.

```
> Bin2:=G(Bin); Bout2:=G(Bout);
```

$$Bin2 := \frac{3\,x\,z}{20\,\pi}\,\mathrm{e_x} - \frac{6\,x^2+3\,z^2-5}{20\,\pi}\,\mathrm{e_z}$$

$$Bout2 := \frac{3\,z\,x}{20\,(x^2+z^2)^{(5/2)}\,\pi}\,\mathrm{e_x} - \frac{-2\,z^2+x^2}{20\,(x^2+z^2)^{(5/2)}\,\pi}\,\mathrm{e_z}$$

The complete magnetic field $\vec{B}$ is formed with the piecewise operator `PW`, taking the nth component of *Bin2* and *Bout2* for $r<a$ and $r>a$, respectively.

```
> PW:=n->piecewise(r<a,Bin2[n],r>a,Bout2[n]):
```

The x and z components of $\vec{B}$ are obtained by taking $n=1$ and 3 in `PW`.

```
> B[1]:=PW(1): B[3]:=PW(3):
```

In `c`, a blue circle of radius a is plotted to represent the spherical surface.

```
> c:=plot(a,theta=0..2*Pi,coords=polar,color=blue,thickness=2):
```

The `fieldplot` command is used in `fp` to plot the magnetic field as thick red arrows, the grid density being taken to be 10 × 10.

```
> fp:=fieldplot([B[1],B[3]],x=-1.1..1.1,z=-1.1..1.1,
      arrows=THICK,grid=[10,10],color=red):
```

The two graphs are superimposed with the `display` command to produce Figure 4.15. The picture should be mentally rotated around the z-axis to obtain the 3-dimensional magnetic field of the rotating uniformly charged sphere.

```
> display({c,fp});
```



Figure 4.15: Magnetic field of rotating charged sphere in x-z plane.

Along the spin axis, the magnetic field points vertically upwards, having maximum strength at the center of the sphere. The magnetic field rapidly drops in strength outside the sphere. The arrows form curved paths which leave the top of the sphere and loop back on the outside to re-enter the sphere at the bottom. The magnetic field behavior is characteristic of a dipole field.

4.4 Supplementary Recipes

04-S01: General Solutions

Consider a second order linear PDE of the general form (a, b, c are constants and the subscripts denote derivatives),

$$a\,\psi_{xx} + b\,\psi_{xy} + c\,\psi_{yy} = 0.$$

(a) By assuming a solution of the form $\psi = f(x + r\,y)$ where f is arbitrary, show that the general solution of the PDE is $\psi = f(x + r_1\,y) + g(x + r_2\,y)$ where

r_1 and r_2 are distinct roots of the quadratic equation $c\,r^2 + b\,r + a = 0$. Determine the two roots. If $r_2 = r_1$, show that the general solution is $\psi = f(x + r_1\,y) + y\,g(x + r_1\,y)$, where f and g are arbitrary functions.

(b) Making use of **(a)**, find the general solutions of the following PDEs. In each case, confirm the solution by solving the PDE directly with `pdsolve`.

(i) $\psi_{xy} - \psi_{yy} = 0$; **(ii)** $\psi_{xx} + \psi_{xy} - 2\,\psi_{yy} = 0$; **(iii)** $\psi_{xx} - 2\,\psi_{xy} + \psi_{yy} = 0$.

04-S02: Balalaika Blues

While trying to create music with a small balalaika, Justine plucks one of the strings which is initially at rest. If the string is fixed at $x = 0$ and L and has an initial transverse profile $\psi(x, 0) = h\,x^3\,\sin(3\,\pi\,x/2L)\,(L - x)^2/L^5$, determine the subsequent displacement $\psi(x, t)$ of the string using a Fourier series approach. Taking $L = 20$ cm, $h = 30$ cm, and wave speed $c = 1$ cm/s, animate the motion of the string. Leave the scaling unconstrained for better viewing of the vibrations.

04-S03: Damped Oscillations

If damping (damping coefficient R) is included, the transverse vibrations of a light, homogeneous, string under tension are governed by

$$\frac{\partial^2\psi}{\partial x^2} = \frac{1}{c^2}\frac{\partial^2\psi}{\partial t^2} + \frac{2R}{c^2}\frac{\partial\psi}{\partial t}.$$

Consider a horizontal string of length L fixed at $x = 0$ and L. If it is initially at rest and is given the initial shape $f(x) = h\,x\,(x - L)$, use the separation of variables method to determine $\psi(x, t)$ for $t > 0$. Taking $L = 2$ m, $h = 0.1$ m^{-1}, speed $c = 5$ m/s, and $R = 1$ s^{-1}, animate the motion of the string over the time interval $t = 0$ to 10 s. You should observe *underdamped* oscillations. Show that *overdamping* occurs for $R = 10$ s^{-1}.

04-S04: Kids Will Be Kids

Consider a very long uniform clothesline under sufficient tension T to keep it horizontal, with Justine's recently washed soccer jersey (mass m) attached with a single clothes peg to the middle of the line at $x = 0$. In an unsuccessful attempt to dump her sister's jersey onto the ground, Gabrielle shakes the clothesline in such a way that a plane wave with amplitude A and phase velocity $c = \omega/k$ is incident from $x < 0$. Show that reflection and transmission occur at $x = 0$ and that the energy reflection coefficient $R = \sin^2(\theta)$ and transmission coefficient $Tr = \cos^2(\theta)$, where $\theta = \arctan(m\omega^2/2kT)$. Determine the phase angle changes for the reflected and transmitted waves in terms of θ.

04-S05: Energy of a Vibrating String

For a finite horizontal string, with linear density $\epsilon(x)$ and under tension T, stretched between $x = x_1$ and x_2, the kinetic energy for small transverse vibrations is $\mathrm{KE} = (1/2)\int_{x_1}^{x_2} \epsilon(x)\,(\partial\psi/\partial t)^2\,dx$. If the ends of the string are either fixed ($\psi = 0$) or free ($\psi' = 0$), the potential energy is $\mathrm{PE} = (1/2)\int_{x_1}^{x_2} T\,(\partial\psi/\partial x)^2\,dx$.

Consider a horizontal string of constant density ϵ and under tension T fixed at $x=0$ and L. If at time $t=0$, the string is given the initial triangular profile $f(x)=2\,h\,x/L$ for $0 \le x \le L$, $f(x)=2\,h\,(L-x)/L$ for $L/2 \le x \le L$, and is released from rest, calculate explicit analytic expressions for the kinetic energy, potential energy, and total energy. What fraction of the total energy is in the fundamental (lowest) frequency? first harmonic (next highest frequency)? second harmonic? Plot the kinetic and potential energies divided by the total energy as a function of time for $L=1$ m and wave velocity $c=\sqrt{T/\epsilon}=1$ m/s.

04-S06: Vibrations of a Tapered String
A horizontal tapered string fixed at $x=0$ and L has a linear density $\epsilon=a(1+b\,x)$. Determine the normal modes for transverse oscillations of the string. Identify the functions which occur in the analytic solution. If $L=1$ m, $a=1/1000$ kg/m, $b=2$ m^{-1}, and the tension $T=1$ N, determine the three lowest eigenfrequencies. Animate the fundamental mode over one period, taking 50 frames.

04-S07: Green Function for Forced Vibrations
A stretched (tension T) string fixed at $x=0$ and L is subjected to forced vibrations by an external force per unit length $f(x,t)=F\,\delta(x-\zeta)\,e^{-i\,\omega\,t}$ with the force amplitude $F=T$ and $0<\zeta<L$. Solve this Green function problem and show that the Green function (spatial part) is

$$G=\frac{\sin[k(L-\zeta)]\,\sin(k\,x)}{k\,\sin(k\,L)},\quad x\le\zeta,\quad G=\frac{\sin[k(L-x)]\,\sin(k\,\zeta)}{k\,\sin(k\,L)},\quad x\ge\zeta,$$

where $k=\omega/v$, v being the wave velocity. G has singularities at certain values of k. Determine these values. Explain, in physical terms, the origin of these singularities and how in real life, they would be "removed".

04-S08: Plane-wave Propagation in a 5-Piece String
Consider an infinitely long string which has a linear density $\epsilon_1=1$ in region 1 ($x<0$), density $\epsilon_2=4$ in region 2 ($0<x<L$), density $\epsilon_3=1$ in region 3 ($L<x<2L$), density $\epsilon_4=4$ in region 4 ($2L<x<3L$), and density $\epsilon_5=1$ in region 5 ($x>3L$). For a plane wave of incident amplitude 1, frequency ω, and wave number K, coming from $x=-\infty$:

(a) Derive the energy transmission coefficient T into region 5 and simplify the result as much as possible.

(b) Show that $T+R=1$, where R is the energy reflection coefficient.

(c) Plot R and T in the same graph as a function of $K\,L$ up to $K\,L=6$. At what values of $K\,L$ is there 100% transmission? Discuss your answer.

04-S09: Transverse Vibrations of a Whirling String
Derive the wave equation for transverse vibrations of a light string of length L pivoted at one end and whirling in a horizontal plane about that pivot with an angular velocity ν. Determine the normal modes of oscillation of the whirling string and the five lowest allowed frequencies. Animate the mode with the second lowest frequency, taking $L=1$ m and $\nu=1$ s^{-1}.

04-S10: Newton Would Think That This Recipe Is Cool

Newton's law of cooling says that the heat flux (amount of heat crossing a unit area per unit time) across a surface is proportional to the temperature difference $T_S - T_0$ between the surface (temperature T_S) and the surrounding medium (temperature T_0). Explicitly this gives the boundary condition $\hat{n} \cdot \nabla T|_S = -h(T_S - T_0)$, where $\hat{n}$ is the outward unit normal to the surface and h is the heat exchange coefficient. Consider an infinitely long uniform bar of rectangular cross-section with axis along the z direction. The two opposite faces at $y=0$ and $y=b$ are held at the temperatures $T=0$ and $T=A$, while the other two faces at $x=\pm a$ radiate heat according to Newton's cooling law into the surrounding medium which is held at $T_0=0$. Derive the solution which takes the form

$$T(x, y) = \sum_{n=1}^{\infty} C_n \sinh(g_n\, y/a) \cos(g_n\, x/a),$$

where the coefficients C_n remain to be identified and g_n are the positive roots of the transcendental equation $g \tan(g) = a\, h$. Taking $a=1$, $b=0.5$, $h=1$, $A=100$, solve the transcendental equation and plot the isotherms inside the bar.

04-S11: Locomotive on a Bridge

As a model of the motion of a locomotive across a railway bridge, consider a periodically vibrating point load $A \sin(\omega\, t)$ moving with constant velocity V along a horizontal, uniform, rectangular steel beam of density ρ, cross-sectional area S, and length L. The beam is fixed at $x=0$ and L and it is assumed that at time $t=0$ the beam is at rest and the locomotive is at $x=0$. The equation of motion for transverse vibrations (amplitude ψ) of the beam is given by

$$a^4 \frac{\partial^4 \psi}{\partial t^4} + \frac{\partial^2 \psi}{\partial t^2} = \frac{q(x, t)}{\epsilon},$$

with $q(x, t) = A \sin(\omega\, t)\, \delta(x - V\, t)$. Here $\epsilon = \rho\, S$ is the linear density of the beam and $a = (\kappa^2\, Y/\rho)^{1/4}$ with κ the radius of gyration of the beam and Y its Young's modulus. Assuming a Fourier series of the form $\psi(x, t) = \sum_{n=1}^{\infty} b_n(t) \sin(n\, \pi\, x/L)$, determine the transverse vibrations of the beam. Taking $\rho=7800$ kg/m^3, $Y=2.1 \times 10^{11}$ N/m^2, $S=1$ m^2, $\kappa=0.5$ m, $A=5 \times 10^4$ N, $\omega=1$ s^{-1}, $L=1000$ m, and $V=1$ m/s, animate the motion of the beam with the motion of the locomotive along the beam superimposed. Keep 20 terms in the series solution and use 100 frames in the animation. To see the oscillations, unconstrained scaling should be used.

04-S12: The Temperature Switch

The temperature at the ends $x = 0$ and $x = 100$ of a rod (insulated on its sides) 100 cm long is held at $0\,°$ and $100\,°$, respectively, until steady-state is achieved. Then, at the instant $t = 0$, the temperature of the two ends is interchanged. Determine the resultant temperature distribution $T(x, t)$. If $a = \sqrt{K/\rho C} = 10/\pi$ cm/s$^{1/2}$, where K is the thermal conductivity of the rod, ρ is its density, and C the specific heat, animate the temperature distribution. What is the temperature at $x = 20$ cm after 25 seconds?

04-S13: Telegraph Equation Revisited

For an audio-frequency submarine cable [SR66], the telegraph equation applies with the leakage constant $G=0$ and the self inductance (per unit length) $L=0$. As shown in recipe **04-2-1**, in this case the telegraph equation reduces to a 1-dimensional diffusion equation for the potential V with a diffusion constant $d=1/(RC)$. Consider a submarine cable $\ell=1000$ km in length, and let the voltage at the source (at $x=0$), under steady-state conditions be 1200 volts and at the receiving end (at $x=\ell$) be 1100 volts. At time $t=0$, the receiving end is grounded, so that its voltage is reduced to zero, but the potential at the source is maintained at its constant value of 1200 volts. If $R=2$ ohms/km and $C=3\times10^{-7}$ farad/km, determine the current and voltage in the line after the grounding of the receiving end and animate the results.

04-S14: Another "Trampoline" Example

A "trampoline" consists of a uniform, light, horizontal, stretched, rectangular membrane with edges of length a and $2a$. The edges at $x=0$, a are fixed, while those at $y=0$, $2a$ are free. Using the separation of variables method, determine the trampoline's subsequent motion if it has the initial shape $f=2\,x\,h/a$ for $0\le x\le a/2$ and $f=2\,h\,(a-x)/a$ for $a/2\le x\le a$ and is released from rest. Taking $h=1/5$ m, $a=2$ m, and $c=2$ m/s, animate the motion of the membrane over the time interval $t=0$ to 5 seconds taking 100 frames.

04-S15: An Electrostatic Poisson Problem

The faces of a rectangular box, $0\le x\le a$, $0\le y\le b$, $0\le z\le c$ are held at the potential $\phi=0$ and the interior is filled with charge with a charge density $\rho=A\,\sin(\pi\,x/a)\,\sin(\pi\,z/c)\,[y\,(y-b)]$ Coulombs/meter^3. Using the electrostatic Poisson equation $\nabla^2\phi=-\rho/\epsilon_0$, where ϵ_0 is the permittivity of free space, determine the potential at an arbitrary point inside the box. Taking $A=10^{-9}$ C/m^5, $a=b=c=1$ m, and $\epsilon_0=8.85\times10^{-12}$ Farads/meter, determine the potential Φ (in volts) at the center of the box. Plot the equipotentials $\Phi/10$, $2\,\Phi/10$,..., $9\,\Phi/10$ in the mid-plane $x=a/2$.

04-S16: SHE Does Not Want to Separate

Demonstrate that the scalar Helmholtz equation (SHE) does not separate in bispherical coordinates even with the modified assumption that was successful for Laplace's equation in recipe **04-3-1**.

04-S17: WE Can Separate

A 2-dimensional curvilinear coordinate system $(u,\,v)$ can be defined through the equations $x=\sqrt{u\,v}$, $y=(u-v)/2$, where the range of u, v is from 0 to ∞.

(a) Plot the contours in the x-y plane corresponding to holding u and v constant. Show that this curvilinear coordinate system is orthogonal.

(b) Calculate the scale factors and the wave equation (WE) for this system.

(c) Show that WE is separable. Identify the separated ODEs and solutions.

04-S18: The Stark Effect

A hydrogenic atom consists of an electron of charge $-e$ and mass m moving in the attractive Coulomb field of a nucleus (atomic number Z) of charge $Z\,e$ and mass M. If $Z=1$, one has the hydrogen atom, $Z=2$ corresponds to the He^{+} ion, $Z=3$ to the Li^{++} ion, and so on. The time-independent Schrödinger equation for the wave function ψ then takes the form $\nabla^2\psi + (2\,m/\hbar^2)[E + Z\,e^2/r]\,\psi = 0$, where $\hbar$ is Planck's constant divided by $2\,\pi$, E is the total energy, and r the radial distance of the electron from the nucleus. When a hydrogenic atom is placed in an external electric field, the energy levels are found to shift. This phenomenon is referred to as the *Stark effect*. If the electric field (magnitude E_0) is oriented in the positive z direction, a potential energy term $-e\,E_0\,z$ must be added to the hydrogenic atom problem. Show that the time-independent Schrödinger equation is still separable in *parabolic* (or *paraboloidal*) coordinates (ζ, η, ϕ) which are related to Cartesian coordinates by the relations $x=\zeta\,\eta\,\cos\phi$, $y=\zeta\,\eta\,\sin\phi$, and $z=(\eta^2-\zeta^2)/2$, with $0 \le \zeta < \infty$, $0 \le \eta < \infty$, $0 \le \phi \le 2\,\pi$.

04-S19: Annular Temperature Distribution

An annular region, of inner radius $r = 10$ cm and outer radius 20 cm, has its inner boundary maintained at the temperature (in degrees Celsius) $T = 20\,\cos\theta$ and the outer boundary held at $T = 30\,\sin\theta$. Determine the steady-state temperature distribution in the annular region and plot the isotherms corresponding to -30, -20, -15,..., 0, 5, 10,..., 30 degrees.

04-S20: Split-boundary Temperature Problem

A thin circular plate of radius 1 m, whose two faces are insulated, has half of its circular boundary kept at the constant temperature $T1$ and the other half at the constant temperature $T2$. Find the steady-state temperature distribution in the plate. Taking $T1 = 300$ degrees Celsius and $T2 = 200$ degrees Celsius, plot the isotherms in 5 degree increments.

04-S21: Fluid Flow Around a Sphere

A solid sphere of radius a is placed in a fluid which was flowing uniformly with speed V_0 in the z direction. The velocity potential U for the fluid in the region outside the sphere satisfies Laplace's equation in spherical polar coordinates and the velocity field is given by $\vec{v} = -\nabla U$. If the sphere is assumed to be rigid, the normal component of $\vec{v}$ must vanish at the surface of the sphere. Determine the velocity field for the fluid outside the sphere and plot the velocity vectors. Take $a=1$ m and $V_0=1$ m/s.

04-S22: Sound of Music?

Some musically inclined people like to sing in the shower stall when taking their shower. In this problem, the shower stall is empty without the water running and consists of a completely enclosed hollow vertical metal cylinder of radius a and height h with (approximately) rigid walls. The speed of sound for the air inside the cylinder is c. By solving the scalar Helmholtz equation for the spatial part of the velocity potential, determine the allowed normal modes inside the cylinder. For rigid walls, the normal component of the fluid velocity (or the normal derivative of the potential) must vanish at each wall. Taking $a = 1.83$ m, $h = 3.04$ m, and $c = 344$ m/s, determine the three lowest eigenfrequencies. By either consulting a musically inclined friend or a music reference book, or going to the Internet, find the closest musical notes on the equal-tempered scale to these eigenfrequencies.